Finite Rod, Albedo Problem, Anisotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[148]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential Random Flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)
a - length of the rod
g - 'mean cosine' of scattering

Analytic solutions

Finite Rod Reflectance(R) and Transmittance(T)

In[149]:=
```
Clear[α, g];
finiterodalbedoanisoscatter`R[a_, α_, Σt_, g_] :=
 (1 - g) α / (2 - α - g α + 2 √((1 - α) (1 - g α)) Coth[a √((1 - α) (1 - g α)) Σt])
```

In[151]:=
```
finiterodalbedoanisoscatter`T[a_, α_, Σt_, g_] :=
 2 / (2 Cosh[a √((1 - α) (1 - g α)) Σt] + ((-2+α+g α) Sinh[-a √((1-α) (1-g α)) Σt]) / √((1-α) (1-g α)))
```

In[152]:=
```
finiterodalbedoanisoscatter`R[a_, α_, Σt_, g_, n_] := α^n (SeriesCoefficient[
      finiterodalbedoanisoscatter`R[a, A, Σt, g], {A, 0, n}] /. A → α);
finiterodalbedoanisoscatter`T[a_, α_, Σt_, g_, n_] := α^n (SeriesCoefficient[
     finiterodalbedoanisoscatter`T[a, A, Σt, g], {A, 0, n}] /. A → α)
```

'Radiance'

In[154]:=
```
finiterodalbedoanisoscatter`LR[x_, α_, Σt_, a_, g_] :=
 -(((e^(-(a+x) Sqrt[(-1+α) (-1+g α)] Σt) Sqrt[(-1+α)/(-1+g α)]
     (-2 Sqrt[-1+α] Sqrt[-1+g α] Cosh[(a-x) Sqrt[-1+α] Sqrt[-1+g α] Σt] +
       (-2+α+g α) Sinh[(a-x) Sqrt[-1+α] Sqrt[-1+g α] Σt])
     (Sqrt[-1+α] Sqrt[-1+g α] Cosh[(a+x) Sqrt[-1+α] Sqrt[-1+g α] Σt] +
       Sqrt[(-1+α) (-1+g α)] Sinh[(a+x) Sqrt[-1+α] Sqrt[-1+g α] Σt]))/
   ((-1+α) (2 (-1+α) Sqrt[(-1+g α)/(-1+α)] Cosh[a Sqrt[(-1+α) (-1+g α)] Σt] +
       (-2+α+g α) Sinh[a Sqrt[(-1+α) (-1+g α)] Σt])))
```

In[155]:=
```
finiterodalbedoanisoscatter`LL[x_, α_, Σt_, a_, g_] :=
 (e^(-(a+x) Sqrt[(-1+α) (-1+g α)] Σt) (e^(2 a Sqrt[(-1+α) (-1+g α)] Σt) - e^(2 x Sqrt[(-1+α) (-1+g α)] Σt))
   Sqrt[(-1+α)/(-1+g α)] (-Sqrt[(-1+g α)/(-1+α)] + g α Sqrt[(-1+g α)/(-1+α)] + Sqrt[(-1+α) (-1+g α)]))/
  (2 (2 (-1+α) Sqrt[(-1+g α)/(-1+α)] Cosh[a Sqrt[(-1+α) (-1+g α)] Σt] +
      (-2+α+g α) Sinh[a Sqrt[(-1+α) (-1+g α)] Σt]))
```

Fluence

In[156]:=
```
finiterodalbedoanisoscatter`ϕ[x_, α_, Σt_, a_, g_] :=
 finiterodalbedoanisoscatter`LR[x, α, Σt, a, g] +
  finiterodalbedoanisoscatter`LL[x, α, Σt, a, g]
```

In[157]:=
```
finiterodalbedoanisoscatter`ϕ[x_, α_, Σt_, a_, g_, n_] := α^n (SeriesCoefficient[
      finiterodalbedoanisoscatter`ϕ[x, A, Σt, a, g], {A, 0, n}] /. A → α);
```

load MC data

In[255]:=
```
finiterodalbedoanisoscatter`ppoints[xs_, dx_, maxx_, Σt_] :=
  Table[{dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]];
```

In[256]:=
```
finiterodalbedoanisoscatter`fs = FileNames[
   "code/rod/finiterod/albedoProblem/data/batch_finiterod_albedoproblem
     _anisotropicscatter_exp*"];
```

In[257]:=
```
finiterodalbedoanisoscatter`index[x_] := Module[{data, α, Σt, g, a},
   data = Import[x, "Table"];
   Σt = data[[1, 11]];
   α = data[[2, 3]];
   g = data[[1, -1]];
   a = data[[2, 5]];
   {α, Σt, g, a, data}];
finiterodalbedoanisoscatter`simulations =
  finiterodalbedoanisoscatter`index /@ finiterodalbedoanisoscatter`fs;
```

In[259]:=
```
finiterodalbedoanisoscatter`alphas =
 Union[#[[1]] & /@ finiterodalbedoanisoscatter`simulations]
```

Out[259]= {0.3, 0.7, 0.9}

In[260]:=
```
finiterodalbedoanisoscatter`muts =
 Union[#[[2]] & /@ finiterodalbedoanisoscatter`simulations]
```

Out[260]= {1, 3}

In[261]:=
```
finiterodalbedoanisoscatter`gs =
 Union[#[[3]] & /@ finiterodalbedoanisoscatter`simulations]
```

Out[261]= {-0.5, 0.3, 0.7}

In[262]:=
```
finiterodalbedoanisoscatter`as =
 Union[#[[4]] & /@ finiterodalbedoanisoscatter`simulations]
```

Out[262]= {0.3, 1, 3}

In[263]:=
```
finiterodalbedoanisoscatter`numcollorders =
 finiterodalbedoanisoscatter`simulations[[1]][[5]][[2, 11]]
```

Out[263]= 10

In[329]:=
```
finiterodalbedoanisoscatter`nummoments =
 finiterodalbedoanisoscatter`simulations[[1]][[5]][[2, 13]]
```

Out[329]= 10

Monte Carlo comparisons

In[167]:=
```
fsplot1 =
  FileNames["code/rod/finiterod/albedoProblem/data/finiterod_albedoproblem
     _anisotropicscatter_exp_c0.7_mut1_*"];
```

Reflectance/Transmittance

In[168]:=
```
MCR[f_] := Module[{data, a},
    data = Import[f, "Table"];
    a = data[[2, 5]];
    {a, data[[3, 4]]}
   ];
MCT[f_] := Module[{data, a},
   data = Import[f, "Table"];
   a = data[[2, 5]];
   {a, data[[4, 4]]}
  ]
```

In[170]:=
```
Rs = Table[MCR[f], {f, fsplot1}];
Ts = Table[MCT[f], {f, fsplot1}];
```

In[172]:=
```
vizfiniterodRTaniso = Show[
   ListLogLinearPlot[Rs, PlotRange → {-0.05, 1},
    PlotStyle → {Black, PointSize[Medium]}],
   ListLogLinearPlot[Ts, PlotRange → {-0.05, 1},
    PlotStyle → {Black, PointSize[Medium]}],
   LogLinearPlot[finiterodalbedoanisoscatter`R[a, 0.7, 1, 0.7],
    {a, 0.1, 100}, PlotStyle → Black],
   LogLinearPlot[finiterodalbedoanisoscatter`T[a, 0.7, 1, 0.7],
    {a, 0.1, 100}, PlotStyle → Dashed],
   ImageSize → 450,
   Frame → True, FrameLabel →
    {{"R(τ),T(τ)",}, {"Optical thickness: τ", "Total Reflectance/Transmittance
        (R/T): anisotropically-scattering finite rod (g = 0.7)"}}
  ]
```

Out[172]=



In[173]:=
```
fs2 =
   FileNames["code/rod/finiterod/albedoProblem/data/finiterod_albedoproblem
      _anisotropicscatter_exp_*length2.6*"];
```

In[174]:=
```
MCR2[f_] := Module[{data, a},
    data = Import[f, "Table"];
    a = data[[2, 3]];
    {a, data[[3, 4]]}
   ];
MCT2[f_] := Module[{data, a},
   data = Import[f, "Table"];
   a = data[[2, 3]];
   {a, data[[4, 4]]}
  ]
```

In[176]:=
```
Rs2 = Table[MCR2[f], {f, fs2}];
Ts2 = Table[MCT2[f], {f, fs2}];
```

In[178]:=
```
a = 2.6;
mut = 1;
finiterodvarycplot = Show[
   ListPlot[Rs2, PlotRange → {0, 0.8}, PlotStyle → {Black, PointSize[Medium]}],
   ListPlot[Ts2, PlotStyle → {Black, PointSize[Medium]}],
   Plot[{finiterodalbedoanisoscatter`R[a, c, mut, 0.7],
     finiterodalbedoanisoscatter`T[a, c, mut, 0.7]},
    {c, 0.01, .99}, PlotRange → {0, 1}, PlotStyle → {Black, Dashed}],
   ImageSize → 450,
   Frame → True, FrameLabel → {{"R(α),T(α)",},
     {"Single scattering albedo: α", "Total Reflectance/Transmittance
        (R/T): anisotropically-scattering (g = 0.7) finite rod"}}
  ]
```

Out[180]=



In[181]:=
```
fs3 =
   FileNames["code/rod/finiterod/albedoProblem/data/finiterod_albedoproblem
      _anisotropicscatter_exp_*length1.2*"];
```

In[182]:=
```
MCR3[f_] := Module[{data, g},
    data = Import[f, "Table"];
    g = data[[1, -1]];
    {g, data[[3, 4]]}
   ];
MCT3[f_] := Module[{data, g},
   data = Import[f, "Table"];
   g = data[[1, -1]];
   {g, data[[4, 4]]}
  ]
```

In[184]:=
```
Rs3 = Table[MCR3[f], {f, fs3}];
Ts3 = Table[MCT3[f], {f, fs3}];
```

In[186]:=
```
a = 1.2;
mut = 1;
c = 0.8;
finiterodvarygplot = Show[
  ListPlot[Rs3, PlotRange → {0, 0.8}, PlotStyle → {Black, PointSize[Medium]}],
  ListPlot[Ts3, PlotStyle → {Black, PointSize[Medium]}],
  Plot[{finiterodalbedoanisoscatter`R[a, c, mut, g],
    finiterodalbedoanisoscatter`T[a, c, mut, g]},
   {g, -0.9, 0.9}, PlotRange → {0, 1}, PlotStyle → {Black, Dashed}],
  ImageSize → 450,
  Frame → True, FrameLabel → {{"R(g),T(g)",},
    {"Anisotropy parameter: g", "Total Reflectance/Transmittance (R/T):
       anisotropically-scattering finite rod (c = 0.8, a = 1.2)"}}
 ]
```

Out[189]=



n-th collided Reflectance/Transmittance

In[278]:=
```
Manipulate[
 If[Length[finiterodalbedoanisoscatter`simulations] > 0,
  Module[{data, numcollorders, Rs, ns, analyticR, analyticT, Ts, j},
   data = SelectFirst[finiterodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g && #[[4]] == a &][[5]];
   numcollorders = data[[2, 11]];
   Rs = N[{data[[6]]}];
   Ts = N[{data[[8]]}];
   ns = Table[n, {n, 0, numcollorders - 1}];
   analyticR = Table[finiterodalbedoanisoscatter`R[a, α, Σt, g, n], {n, ns}];
   analyticT = Table[finiterodalbedoanisoscatter`T[a, α, Σt, g, n], {n, ns}];
   j = Join[{ns}, {analyticR}, Rs, {analyticT}, Ts];
   TableForm[
    Join[
     {{"n", "R (analytic)", "R (MC)", "T (analytic)", "T (MC)"}}, Transpose[j]]
   ]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, finiterodalbedoanisoscatter`alphas},
 {Σt, finiterodalbedoanisoscatter`muts},
 {g, finiterodalbedoanisoscatter`gs}, {a, finiterodalbedoanisoscatter`as}]
```

Out[278]=

α 0.3 | 0.7 | 0.9

Σt 1 | 3

g −0.5 | 0.3 | 0.7

a 0.3 | 1 | 3

n	R (analytic)	R (MC)	T (analytic)	T (MC)
0	0.	0.	0.00012341	0.0001279
1	0.1225	0.122614	0.000505363	0.0004986
2	0.0557375	0.0557034	0.00106621	0.0010617
3	0.0271988	0.0271757	0.00155481	0.0015657
4	0.0140481	0.0140483	0.00177532	0.0017924
5	0.00758836	0.0076386	0.00170579	0.0017036
6	0.00424506	0.0042538	0.00144745	0.0014361
7	0.00244069	0.0024229	0.00112336	0.0011204
8	0.00143382	0.0014357	0.000818442	0.0008188
9	0.000856773	0.0008555	0.000570684	0.0005821

Internal Distributions

In[322]:=
```
Manipulate[
 If[Length[finiterodalbedoanisoscatter`simulations] > 0,
  Module[{data, numcollorders, nummoments, dx, pointsCL, plotpointsCL,
    pointsCR, plotpointsCR, plotpointsϕ, plotϕ, plotLL, plotLR},
   data = SelectFirst[finiterodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g && #[[4]] == a &][[5]];
   numcollorders = data[[2, 11]];
   nummoments = data[[2, 13]];
   dx = data[[2, 7]];
```

```
  pointsCL = data[[10]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = finiterodalbedoanisoscatter`ppoints[pointsCL, dx, a, Σt];
  pointsCR = data[[12]];
  plotpointsCR = finiterodalbedoanisoscatter`ppoints[pointsCR, dx, a, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ =
   finiterodalbedoanisoscatter`ppoints[pointsCL + pointsCR, dx, a, Σt];

  plotϕ = Show[
    ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[
     finiterodalbedoanisoscatter`ϕ[x, α, Σt, a, g], {x, 0, a}, PlotRange → All]
    , Frame → True,
    FrameLabel ->
     {{ϕ[x],}, {x, "Finite rod, albedo problem, anisotropic scattering,
          fluence ϕ[x]\n α = " <> ToString[α] <> ", Σt = " <>
       ToString[Σt] <> ", a = " <> ToString[a] <> ", g = " <> ToString[g]}}
   ];

  plotLL = Show[
    ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[
     finiterodalbedoanisoscatter`LL[x, α, Σt, a, g], {x, 0, a}, PlotRange → All],
    Plot[finiterodalbedoanisoscatter`ϕ[x, α, Σt, a, g],
     {x, 0, a}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L-[x],},
      {x, "Finite rod, albedo problem, anisotropic scattering, L-[x]\n α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt] <> ", a = " <>
        ToString[a] <> ", g = " <> ToString[g]}}, PlotRange → All
   ];

  plotLR = Show[
    ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[
     finiterodalbedoanisoscatter`LR[x, α, Σt, a, g], {x, 0, a}, PlotRange → All],
    Plot[finiterodalbedoanisoscatter`ϕ[x, α, Σt, a, g],
     {x, 0, a}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L+[x],},
      {x, "Finite rod, albedo problem, anisotropic scattering, L+[x]\n α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt] <> ", a = " <>
        ToString[a] <> ", g = " <> ToString[g]}}, PlotRange → All
   ];
  GraphicsRow[{Show[plotϕ, ImageSize → 400], plotLL, plotLR}]
 ]
 ,
 Text["Uh oh!  Couldn't find MC data.  Try to evaluate
    this entire notebook and ensure the data path is setup correctly."]
]
, {α, finiterodalbedoanisoscatter`alphas},
{Σt, finiterodalbedoanisoscatter`muts},
{g, finiterodalbedoanisoscatter`gs}, {a, finiterodalbedoanisoscatter`as}]
```

Out[322]=

α 0.3 0.7 0.9

Σt 1 3

g −0.5 0.3 0.7

a 0.3 1 3





Nth order fluence/density

In[345]:=
```
Manipulate[
 If[Length[finiterodalbedoanisoscatter`simulations] > 0,
  Module[{data, numcollorders, nummoments, dx, nthϕ, plotpointsϕ, plotϕ},
   data = SelectFirst[finiterodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g && #[[4]] == a &][[5]];
   numcollorders = data[[2, 11]];
   nummoments = data[[2, 13]];
   dx = data[[2, 7]];

   nthϕ = data[[16 + numcollorders + 1 ;; 16 + 2 numcollorders]];
   plotpointsϕ = finiterodalbedoanisoscatter`ppoints[nthϕ[[n + 2]], dx, a, Σt];

   plotϕ = Show[
     ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
     Plot[
      finiterodalbedoanisoscatter`ϕ[x, α, Σt, a, g, n], {x, 0, a}, PlotRange → All]
     , Frame → True,
     FrameLabel ->
      {{ϕ[x | n],}, {x, "Finite rod, albedo problem, anisotropic scattering,
           fluence ϕ[x|n]\n α = " <> ToString[α] <> ", Σt = " <>
        ToString[Σt] <> ", a = " <> ToString[a] <> ", g = " <> ToString[g]}}
    ];


   plotϕ
  ]
  ,
  Text["Uh oh!  Couldn't find MC data.  Try to evaluate
     this entire notebook and ensure the data path is setup correctly."]
 ]
 , {α, finiterodalbedoanisoscatter`alphas},
 {Σt, finiterodalbedoanisoscatter`muts},
 {g, finiterodalbedoanisoscatter`gs}, {a, finiterodalbedoanisoscatter`as},
 {{n, 4}, Range[If[NumberQ[finiterodalbedoanisoscatter`nummoments],
    finiterodalbedoanisoscatter`nummoments, 1]]}
]
```

Out[345]=

α 0.3 0.7 0.9

Σt 1 3

g −0.5 0.3 0.7

a 0.3 1 3

n 3