Infinite Rod, Isotropic Point Source, Isotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[78]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential random flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)

Analytic solutions

'Radiance'

In[79]:=
```
infrodisopointisoscatter`LR[x_, α_, Σt_] := e^(-Abs[x] Sqrt[1-α] Σt) (1 + Sqrt[1 - α] Sign[x]) / (4 Sqrt[1 - α])
```

In[80]:=
```
infrodisopointisoscatter`LL[x_, α_, Σt_] := e^(-Abs[x] Sqrt[1-α] Σt) (1 - Sqrt[1 - α] Sign[x]) / (4 Sqrt[1 - α])
```

Fluence

In[81]:=
```
infrodisopointisoscatter`ϕ[x_, α_, Σt_] := 1/(2 Sqrt[1 - α]) E^(-Σt Sqrt[1-α] Abs[x])
```

n-th collided fluence

In[82]:=
```
infrodisopointisoscatter`ϕ[x_, n_, α_, Σt_] :=
 2^(-n-1/2) α^n (Abs[x] Σt)^(n+1/2) BesselK[n + 1/2, Σt Abs[x]] / (Sqrt[Pi] Gamma[n + 1])
```

moments

In[83]:=
```
infrodisopointisoscatter`ϕm[α_, Σt_, k_] := 1/2 (1 + (-1)^k) Σt^(-1-k) (1 - α)^(-1-k/2) k!
```

In[84]:=
```
infrodisopointisoscatter`ϕm[α_, Σt_, k_, n_] :=
 α^n (1/2 (-1)^n (1 + (-1)^k) Σt^(-1-k) Binomial[-1 - k/2, n] k!)
```

load MC data

In[85]:=
```
infrodisopointisoscatter`ppoints[xs_, dx_, maxx_, Σt_] :=
 Table[{-maxx + dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[2 ;; -2]]
```

In[86]:=
```
infrodisopointisoscatter`fs = FileNames[
   "code/rod/infiniterod/Isotropicpointsource/data/infrod_isotropicpoint
     _isotropicscatter_exp*"];
```

In[87]:=
```
infrodisopointisoscatter`index[x_] := Module[{data, α, Σt},
   data = Import[x, "Table"];
   Σt = data[[1, 12]];
   α = data[[2, 3]];
   {α, Σt, data}];
infrodisopointisoscatter`simulations =
  infrodisopointisoscatter`index /@ infrodisopointisoscatter`fs;
```

In[89]:=
```
infrodisopointisoscatter`alphas =
 Union[#[[1]] & /@ infrodisopointisoscatter`simulations]
```

Out[89]= {0.1, 0.3, 0.5, 0.7, 0.9}

In[90]:=
```
infrodisopointisoscatter`muts =
 Union[#[[2]] & /@ infrodisopointisoscatter`simulations]
```

Out[90]= {1, 3}

In[91]:=
```
infrodisopointisoscatter`numcollorders =
  infrodisopointisoscatter`simulations[[1]][[3]][[2, 11]];
```

Compare Deterministic and MC

Internal distributions

In[99]:=
```
Clear[α, Σt];
Manipulate[
```

```
If[Length[infrodisopointisoscatter`simulations] > 0,
 Module[{data, maxx, dx, numcollorders, nummoments, densmom, pointsCL,
   plotpointsCL, pointsCR, plotpointsCR, plotpointsϕ, plotϕ, plotLL, plotLR},
  data = SelectFirst[infrodisopointisoscatter`simulations,
     #[[1]] == α && #[[2]] == Σt &][[3]];
  maxx = data[[2, 5]];
  dx = data[[2, 7]];
  numcollorders = data[[2, 11]];
  nummoments = data[[2, 13]];

  pointsCL = data[[4]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = infrodisopointisoscatter`ppoints[pointsCL, dx, maxx, Σt];
  pointsCR = data[[6]];
  plotpointsCR = infrodisopointisoscatter`ppoints[pointsCR, dx, maxx, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ =
   infrodisopointisoscatter`ppoints[pointsCL + pointsCR, dx, maxx, Σt];

  plotϕ = Show[
    ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[infrodisopointisoscatter`ϕ[x, α, Σt],
     {x, -maxx, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{ϕ[x],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, fluence
          ϕ[x], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}
   ];

  plotLL = Show[
    ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[infrodisopointisoscatter`LL[x, α, Σt],
     {x, -maxx, maxx}, PlotRange → All],
    Plot[infrodisopointisoscatter`ϕ[x, α, Σt], {x, -maxx, maxx},
     PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L-[x],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, L-[x], α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];

  plotLR = Show[
    ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[infrodisopointisoscatter`LR[x, α, Σt],
     {x, -maxx, maxx}, PlotRange → All],
    Plot[infrodisopointisoscatter`ϕ[x, α, Σt], {x, -maxx, maxx},
     PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L+[x],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, L+[x], α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];
  Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
 ]
 ,
 Text[
  "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
    ensure the data path is setup correctly."]
]
, {{α, 0.7}, infrodisopointisoscatter`alphas},
{Σt, infrodisopointisoscatter`muts}]
```

Out[100]=

α 0.1 0.3 0.5 0.7 0.9

Σt 1 3









Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

In[101]:=
```
Manipulate[
 If[Length[infrodisopointisoscatter`simulations] > 0,
  Module[{data, ϕmoments, ks, analytic, j, nummoments},
   data = SelectFirst[
      infrodisopointisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 13]];
   ϕmoments = {data[[8]]}/Σt;
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic = Table[infrodisopointisoscatter`ϕm[α, Σt, k], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.5}, infrodisopointisoscatter`alphas},
 {{Σt, 3}, infrodisopointisoscatter`muts}]
```

Out[101]=

α 0.1 0.3 0.5 0.7 0.9

Σt 1 3

k	analytic	MC
0	0.666667	0.666633
1	0.	0.000123903
2	0.296296	0.296473
3	0.	0.00088346
4	0.790123	0.79103
5	0.	0.0130178
6	5.26749	5.2798
7	0.	0.376367
8	65.551	66.264
9	0.	18.4986

nth-collided moments of ϕ

In[103]:=
```
Manipulate[
 If[Length[infrodisopointisoscatter`simulations] > 0,
  Module[{data, ϕmoments, ks, analytic, j, nummoments},
   data = SelectFirst[
      infrodisopointisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[10 + n]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic =
    Table[Quiet[N[infrodisopointisoscatter`ϕm[α, Σt, k, n]]], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.7}, infrodisopointisoscatter`alphas},
 {{Σt, 3}, infrodisopointisoscatter`muts},
 {{n, 8}, Range[If[NumberQ[infrodisopointisoscatter`numcollorders],
    infrodisopointisoscatter`numcollorders, 1]]}]
```

Out[103]=



k	analytic	MC
0	0.019216	0.0191944
1	0.	0.0000504033
2	0.038432	0.0383507
3	0.	0.00034342
4	0.256213	0.255702
5	0.	0.0067798
6	3.1315	3.12896
7	0.	0.150773
8	58.4546	58.4373
9	0.	3.1651

N-th order Radiance/Angular flux

In[104]:=
```
Manipulate[
 If[Length[infrodisopointisoscatter`simulations] > 0,
  Module[{data, nthL, nthR, numcollorders, LnR, maxx, dx},
   data = SelectFirst[
      infrodisopointisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   numcollorders = data[[2, 11]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   nthL = data[[10 + numcollorders + 1 ;; 10 + 2 numcollorders]];
   nthR = data[[10 + 2 numcollorders + 2 ;; -1]];

   Clear[c, x];
   LnR = SeriesCoefficient[infrodisopointisoscatter`LR[x, c, Σt], {c, 0, n}] α^n;
   Show[
    ListPlot[infrodisopointisoscatter`ppoints[nthR[[n + 1]], dx, maxx, Σt],
     PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LnR, {x, -maxx, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{L+["x|2"],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, L+[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <> ", Σt = " <>
        ToString[Σt]}}, PlotRange → All, ImageSize → 500
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, infrodisopointisoscatter`alphas},
 {{Σt, 1}, infrodisopointisoscatter`muts},
 {{n, 5}, Range[If[NumberQ[infrodisopointisoscatter`numcollorders],
    infrodisopointisoscatter`numcollorders, 1]]}]
```

Out[104]=

α 0.1 0.3 0.5 0.7 0.9

Σt 1 3

n 5



N-th order Fluence / scalar flux

In[105]:=
```
Manipulate[
 If[Length[infrodisopointisoscatter`simulations] > 0,
  Module[{data, nthL, nthR, numcollorders, maxx, dx},
   data = SelectFirst[
      infrodisopointisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   numcollorders = data[[2, 11]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   nthL = data[[10 + numcollorders + 1 ;; 10 + 2 numcollorders]];
   nthR = data[[10 + 2 numcollorders + 2 ;; -1]];

   Show[
    ListPlot[infrodisopointisoscatter`ppoints[nthR[[n + 1]] + nthL[[n + 1]],
      dx, maxx, Σt], PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[infrodisopointisoscatter`ϕ[x, n, α, Σt],
     {x, -maxx, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{ϕ["x|7"],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, ϕ[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <> ", Σₜ = " <>
        ToString[Σt]}}, PlotRange → All, ImageSize → 500
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.7}, infrodisopointisoscatter`alphas},
 {Σt, infrodisopointisoscatter`muts},
 {{n, 2}, Range[If[NumberQ[infrodisopointisoscatter`numcollorders],
    infrodisopointisoscatter`numcollorders, 1]]}]
```

Out[105]=

α 0.1 0.3 0.5 0.7 0.9

Σt 1 3

n 2