Infinite 3D medium, Isotropic Plane Source, Isotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[50]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



α - single-scattering albedo
Σt - extinction coefficient
z - scalar position coordinate in medium (distance from plane source at origin)
u = cos θ - direction cosine

Analytic solutions

Caseology quantities

In[51]:=
```
CaseN0[c_, v0_] := 1/2 c v0^3 (c/(v0^2 - 1) - 1/v0^2)
```

In[52]:=
```
Casev0[c_?NumericQ] :=
 FindRoot[c v ArcTanh[1/v] - 1 == 0, {v, 1.00000000001, 10^10}, Method -> "Brent"][[1]][[
  2]]
```

In[53]:=
```
CaseN[c_, v_] := v (Caseλ[v, c]^2 + (π c v/2)^2)
```

In[54]:=
```
Caseλ[v_, c_] := 1 - c v ArcTanh[v]
```

In[55]:=
```
Caseψ0[u_, v0_, c_, z_] := c/2 v0/(v0 - Sign[z] u)
```

Rigorous diffusion approximation

In[56]:=
```
inf3Disoplaneisoscatter`ϕrigourousDiffusion[z_, Σt_, α_] :=
 1/2 E^(-Abs[z] Σt/#)/CaseN0[α, #] & [Casev0[α]]
```

Fluence: exact solution

Fourier Transform:

In[57]:=
```
inf3Disoplaneisoscatter`ϕunscattered[z_, Σt_] := Gamma[0, Abs[z Σt]]/2
```

In[58]:=
```
inf3Disoplaneisoscatter`ϕexact1[z_, Σt_, α_] :=
 inf3Disoplaneisoscatter`ϕunscattered[z, Σt] +
  NIntegrate[α ArcTan[k]^2 Cos[k z Σt]/(π (k^2 - α k ArcTan[k])), {k, 0, Infinity}]
```

[Case 1960, Case and Zwiefel 1967]

In[59]:=
```
inf3Disoplaneisoscatter`ϕexact2[z_, Σt_, α_] :=
 inf3Disoplaneisoscatter`ϕrigourousDiffusion[z, Σt, α] +
  1/2 NIntegrate[ℯ^(-Σt Abs[z]/v)/CaseN[α, v], {v, 0, 1}]
```

Nth-scattered fluence

In[60]:=
```
inf3Disoplaneisoscatter`ϕexact1[x_, Σt_, α_, n_] :=
 (α Σt)^n/π NIntegrate[ArcTan[z/Σt]^(n+1) Cos[x z]/z^(n+1),
  {z, 0, Infinity}, Method -> "ExtrapolatingOscillatory"]
```

In[61]:=
$$\text{inf3Disoplaneisoscatter`}\phi\text{exact2}[x\_, \Sigma t\_, \alpha\_, n\_] := \frac{\alpha^n}{\pi}\,\text{Chop}\left[\text{NIntegrate}\left[i\, 2^{-2-n}\, e^{-\text{Abs}[x]\, z\, \Sigma t}\left(\left(\frac{-i\,\pi + \text{Log}\left[\frac{z+1}{z-1}\right]}{z}\right)^{1+n} - \left(\frac{i\,\pi + \text{Log}\left[\frac{z+1}{z-1}\right]}{z}\right)^{1+n}\right), \{z, 1, \text{Infinity}\}\right]\right]$$

In[62]:=
$$\text{inf3Disoplaneisoscatter`}\phi\text{Gaussian}[x\_, \Sigma t\_, \alpha\_, n\_] := \frac{e^{-\frac{3\,x^2\,\Sigma t^2}{4\,(1+n)}}\sqrt{\frac{3}{\pi}}\,\alpha^n}{2\sqrt{1+n}}$$

Classical Diffusion Approximation

In[63]:=
$$\text{inf3Disoplaneisoscatter`}\phi\text{Diffusion}[x\_, \Sigma t\_, \alpha\_] := \frac{3\, e^{-\text{Abs}[x]\sqrt{3-3\,\alpha}\,\Sigma t}}{2\sqrt{3-3\,\alpha}}$$

Grosjean Modified Diffusion Approximation

In[64]:=
$$\text{inf3Disoplaneisoscatter`}\phi\text{Grosjean}[x\_, \Sigma t\_, \alpha\_] := -\frac{1}{2}\,\text{ExpIntegralEi}[-\text{Abs}[x]\,\Sigma t] + \frac{e^{-\text{Abs}[x]\sqrt{3+\frac{3}{-2+\alpha}}\,\Sigma t}\sqrt{3+\frac{3}{-2+\alpha}}\,\alpha}{2\,(1-\alpha)}$$

Moments

In[121]:=
$$\text{inf3Disoplaneisoscatter`}\phi\text{m}[c\_, \Sigma t\_, m\_?\text{IntegerQ}, n\_] := \frac{1}{2\,\frac{m}{2}+1}\,\text{Limit}\left[\text{Simplify}\left[(-1)^{m/2}\left(\frac{2\,\text{Gamma}\left[\frac{3+m}{2}\right]}{\text{Gamma}\left[\frac{1+m}{2}\right]}\,\text{D}\left[\frac{\left(\frac{c\,\Sigma t\,\text{ArcTan}\left[\frac{z}{\Sigma t}\right]}{z}\right)^{1+n}}{c\,\Sigma t}, \{z, m\}\right]\right)\right], z \to 0\right]$$

In[122]:=
$$\text{TableForm}[\text{Table}[\text{inf3Disoplaneisoscatter`}\phi\text{m}[\alpha, \Sigma t, m, n], \{m, 0, 6, 2\}]]$$

Out[122]//TableForm=

$$\frac{\alpha^n}{\Sigma t}$$
$$\frac{2\,(1+n)\,\alpha^n}{3\,\Sigma t^3}$$
$$\frac{4\,(1+n)\,(18+5\,n)\,\alpha^n}{15\,\Sigma t^5}$$
$$\frac{8\,(1+n)\,(810+343\,n+35\,n^2)\,\alpha^n}{63\,\Sigma t^7}$$

In[123]:=
$$\text{inf3Disoplaneisoscatter`}\phi\text{m}[c\_, \Sigma t\_, m\_?\text{IntegerQ}] := \frac{1}{2\,\frac{m}{2}+1}\,\text{Limit}\left[\text{Simplify}\left[(-1)^{m/2}\left(\frac{2\,\text{Gamma}\left[\frac{3+m}{2}\right]}{\text{Gamma}\left[\frac{1+m}{2}\right]}\,\text{D}\left[\frac{\text{ArcTan}\left[\frac{z}{\Sigma t}\right]}{z - c\,\Sigma t\,\text{ArcTan}\left[\frac{z}{\Sigma t}\right]}, \{z, m\}\right]\right)\right], z \to 0\right]$$

In[124]:=
$$\text{TableForm}[\text{Table}[\text{inf3Disoplaneisoscatter`}\phi\text{m}[\alpha, \Sigma t, m], \{m, 0, 6, 2\}]]$$

Out[124]//TableForm=

$$\frac{1}{\Sigma t - \alpha\,\Sigma t}$$
$$\frac{2}{3\,(-1+\alpha)^2\,\Sigma t^3}$$
$$\frac{8\,(-9+4\,\alpha)}{15\,(-1+\alpha)^3\,\Sigma t^5}$$
$$\frac{16\,(135-144\,\alpha+44\,\alpha^2)}{21\,(-1+\alpha)^4\,\Sigma t^7}$$

Radiance (Angular Flux)

Caseology - asymptotic solution:

In[65]:=
```
inf3Disoplaneisoscatter`LrigourousDiffusion[z_, u_, Σt_, α_] :=
 1/(4 Pi) Caseψ0[u, #, α, z] E^(-Abs[z] Σt/#)/CaseN0[α, #] &[Casev0[α]]
```

Caseology - exact solution:

In[66]:=
```
inf3Disoplaneisoscatter`Lexact[z_, u_, Σt_, α_] :=
 inf3Disoplaneisoscatter`LrigourousDiffusion[z, u, Σt, α] +
  1/(4 Pi) (Caseλ[u, α] E^(-Abs[z] Σt/u)/CaseN[α, u] HeavisideTheta[1 - u] HeavisideTheta[u]
    + NIntegrate[E^(-Abs[z] Σt/v)/CaseN[α, v] α/2 v/(v - u)
     , {v, 0, u, 1}, Method → "PrincipalValue", PrecisionGoal → 5])
```

Fourier transform exact solution [Beach et al. 1959]:

In[67]:=
```
inf3Disoplaneisoscatter`FourierLunscattered[z_, u_, Σt_] :=
 E^(-z Σt/u)/(4 Pi u) Sign[z] HeavisideTheta[u Sign[z]]
inf3Disoplaneisoscatter`FourierLscattered[z_, c_, u_, Σt_] :=
 1/(4 Pi) c/Pi NIntegrate[ArcTan[k] (Cos[k z Σt] + k u Sin[k z Σt])/((1 + k^2 u^2) (k - c ArcTan[k])), {k, 0, Infinity}]
```

In[69]:=
```
inf3Disoplaneisoscatter`LexactFourier[z_, u_, Σt_, α_] :=
 inf3Disoplaneisoscatter`FourierLunscattered[z, u, Σt] +
  inf3Disoplaneisoscatter`FourierLscattered[z, α, u, Σt]
```

In[70]:=
```
inf3Disoplaneisoscatter`LexactFourier2[z_, u_, Σt_, c_] :=
 1/(4 Pi) 1/Pi NIntegrate[(Cos[k z Σt] + k u Sin[k z Σt])/(1 + k^2 u^2) 1/(1 - c ArcTan[k]/k), {k, 0, Infinity}]
```

load MC data

In[71]:=
```
inf3Disoplaneisoscatter`ppoints[zs_, dz_, maxz_, Σt_] :=
 Table[{dz (i) - 0.5 dz - maxz, zs[[i]]/Σt}, {i, 1, Length[zs]}][[2 ;; -2]]
```

In[72]:=
```
inf3Disoplaneisoscatter`ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]]/(2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[73]:=
```
inf3Disoplaneisoscatter`fs =
  FileNames["code/3D_medium/infinite3Dmedium/Isotropicplanesource/data/
     inf3D_isotropicplane_isotropicscatter*"];
```

In[74]:=
```
inf3Disoplaneisoscatter`index[x_] := Module[{data, α, Σt},
    data = Import[x, "Table"];
    Σt = data[[1, 13]];
    α = data[[2, 3]];
    {α, Σt, data}];
inf3Disoplaneisoscatter`simulations =
  inf3Disoplaneisoscatter`index /@ inf3Disoplaneisoscatter`fs;
inf3Disoplaneisoscatter`alphas =
 Union[#[[1]] & /@ inf3Disoplaneisoscatter`simulations]
```

Out[76]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[77]:=
```
inf3Disoplaneisoscatter`muts =
 Union[#[[2]] & /@ inf3Disoplaneisoscatter`simulations]
```

Out[77]=
```
{1, 3}
```

In[78]:=
```
inf3Disoplaneisoscatter`numcollorders =
  inf3Disoplaneisoscatter`simulations[[1]][[3]][[2, 13]];
inf3Disoplaneisoscatter`maxz =
 inf3Disoplaneisoscatter`simulations[[1]][[3]][[2, 5]];
inf3Disoplaneisoscatter`dz =
 inf3Disoplaneisoscatter`simulations[[1]][[3]][[2, 7]];
inf3Disoplaneisoscatter`numz =
  Floor[2 inf3Disoplaneisoscatter`maxz / inf3Disoplaneisoscatter`dz];
```

Compare Deterministic and MC

Fluence - Exact solution 1 (Fourier Transform) comparison to MC

In[81]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ, plotpointsϕ, logplotϕ, plotϕ, exact1points},
   data = SelectFirst[
      inf3Disoplaneisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];

   pointsϕ = data[[4]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disoplaneisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   exact1points =
    Quiet[{#[[1]], inf3Disoplaneisoscatter`ϕexact[#[[1]], Σt, α]} & /@
     plotpointsϕ;

   numpoints = Length[plotpointsϕ];
   skip = Floor[numpoints 6/7 1/2];

   plotϕ = Quiet[Show[
      Plot[inf3Disoplaneisoscatter`ϕexact1[z, Σt, α], {z, -maxz/7, maxz/7}],
      Plot[inf3Disoplaneisoscatter`ϕunscattered[z, Σt],
       {z, -maxz, maxz}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListPlot[plotpointsϕ[[skip ;; -skip]], PlotRange → All,
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ ϕ[z],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[exact1points, PlotRange → All, Joined → True],
      ListLogPlot[plotpointsϕ[[1 ;; -1 ;; 3]],
       PlotRange → All, PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ ϕ[z],}, {z,}}
     ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800], PlotLabel ->
     "Exact solution 1 (thin), Unscattered component (dashed)\nInfinite
        3D, isotropic plane source, isotropic scattering, fluence
        ϕ[z], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.5}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 1}, inf3Disoplaneisoscatter`muts}]
```

Out[81]=

α 0.5

Σt 1 3



Fluence - Exact solution 2 (Caseology) comparison to MC

In[82]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ, plotpointsϕ, logplotϕ, plotϕ, exact1points},
   data = SelectFirst[
      inf3Disoplaneisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];

   pointsϕ = data[[4]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disoplaneisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   exact1points =
    Quiet[{#[[1]], inf3Disoplaneisoscatter`ϕexact2[#[[1]], Σt, α]}] & /@
     plotpointsϕ;

   numpoints = Length[plotpointsϕ];
   skip = Floor[numpoints 6/7 1/2];

   plotϕ = Quiet[Show[
      (*ListPlot[exact1points[[skip;;-skip]],PlotRange→{0,6},Joined→True],*)
      Plot[inf3Disoplaneisoscatter`ϕexact2[z, Σt, α],
       {z, -maxz/7, maxz/7}, PlotRange → {0, 6}],
      Plot[inf3Disoplaneisoscatter`ϕrigourousDiffusion[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListPlot[plotpointsϕ[[skip ;; -skip]], PlotRange → All,
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ϕ[z],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[exact1points, PlotRange → All, Joined → True],
      LogPlot[inf3Disoplaneisoscatter`ϕrigourousDiffusion[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListLogPlot[plotpointsϕ[[1 ;; -1 ;; 3]], PlotRange → All,
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ϕ[z],}, {z,}}
     ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800],
    PlotLabel -> "Exact solution 2 (thin), Rigorous Diffusion
        component (dashed)\nInfinite 3D, isotropic plane
        source, isotropic scattering, fluence ϕ[z], α = " <>
      ToString[α] <> ", Σt = " <> ToString[Σt]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.95}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 1}, inf3Disoplaneisoscatter`muts}]
```

Out[82]=

α 0.95

Σt 1 3



Fluence - Diffusion approximations (Classical and Grosjean) comparison to MC

In[83]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ,
    plotpointsϕ, logplotϕ, plotϕ, exact1points, numpoints, skip},
   data = SelectFirst[inf3Disoplaneisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];

   pointsϕ = data[[4]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disoplaneisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   numpoints = Length[plotpointsϕ];
   skip = Floor[numpoints 6/7 1/2];

   plotϕ = Quiet[Show[
      ListPlot[plotpointsϕ[[skip ;; -skip]],
       PlotRange → All, PlotStyle → {PointSize[.01], Black}],
      Plot[inf3Disoplaneisoscatter`ϕGrosjean[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All],
      Plot[inf3Disoplaneisoscatter`ϕDiffusion[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListPlot[plotpointsϕ[[skip ;; -skip]], PlotRange → All,
       PlotStyle → {PointSize[.01], Black}],
      Frame → True,
      FrameLabel -> {{ ϕ[z],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[plotpointsϕ,
       PlotRange → All, PlotStyle → {PointSize[.01], Black}],
      LogPlot[inf3Disoplaneisoscatter`ϕGrosjean[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All],
      LogPlot[inf3Disoplaneisoscatter`ϕDiffusion[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListLogPlot[plotpointsϕ, PlotRange → All,
       PlotStyle → {PointSize[.01], Black}],
      Frame → True,
      FrameLabel -> {{ ϕ[z],}, {z,}}
     ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800],
    PlotLabel -> "Classical (dashed) and Grosjean Modified (thin)
        Diffusion Approximation\nInfinite 3D, isotropic plane
        source, isotropic scattering, fluence ϕ[z], α = " <>
      ToString[α] <> ", Σt = " <> ToString[Σt]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.5}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 1}, inf3Disoplaneisoscatter`muts}]
```

Out[83]=

α 0.5

Σt 1 3



N-th order fluence / scalar flux

N-th collided Fluence - Exact solution (1) comparison to MC

In[84]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ,
    plotpointsϕ, logplotϕ, plotϕ, exact1points, numorders},
   data = SelectFirst[inf3Disoplaneisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];

   numorders = data[[2, 13]];

   pointsϕ = data[[9 + numorders + n + 1]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disoplaneisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   exact1points =
    Quiet[{#[[1]], inf3Disoplaneisoscatter`ϕexact1[#[[1]], Σt, α, n]} & /@
     plotpointsϕ;

   plotϕ = Quiet[Show[
      ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      ListPlot[exact1points, PlotRange → All, Joined → True],
      Frame → True,
      FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      ListLogPlot[exact1points, PlotRange → All, Joined → True],
      Frame → True,
      FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
     ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800],
    PlotLabel -> "Exact solution (1)\nInfinite 3D, isotropic plane source,
       isotropic scattering, n-th scattered fluence ϕ[z|n], α = " <>
      ToString[α] <> ", Σt = " <> ToString[Σt]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.99}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 3}, inf3Disoplaneisoscatter`muts},
 {{n, 13}, Range[If[NumberQ[inf3Disoplaneisoscatter`numcollorders],
    inf3Disoplaneisoscatter`numcollorders, 1]]}

]
```

Out[84]=



N-th collided Fluence - Exact solution (2) comparison to MC

In[85]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ,
    plotpointsϕ, logplotϕ, plotϕ, exact1points, numorders},
   data = SelectFirst[inf3Disoplaneisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];

   numorders = data[[2, 13]];

   pointsϕ = data[[9 + numorders + n + 1]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disoplaneisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   exact1points =
    Quiet[{#[[1]], inf3Disoplaneisoscatter`ϕexact2[#[[1]], Σt, α, n]}] & /@
     plotpointsϕ;

   plotϕ = Quiet[Show[
      ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      ListPlot[exact1points, PlotRange → All, Joined → True],
      Frame → True,
      FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      ListLogPlot[exact1points, PlotRange → All, Joined → True],
      Frame → True,
      FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
     ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800],
    PlotLabel -> "Exact solution (2)\nInfinite 3D, isotropic plane source,
       isotropic scattering, n-th scattered fluence ϕ[z|n], α = " <>
      ToString[α] <> ", Σt = " <> ToString[Σt]]
 ]
 ,
 Text[
  "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
    ensure the data path is setup correctly."]
]
, {{α, 0.99}, inf3Disoplaneisoscatter`alphas},
{{Σt, 3}, inf3Disoplaneisoscatter`muts},
{{n, 13}, Range[If[NumberQ[inf3Disoplaneisoscatter`numcollorders],
   inf3Disoplaneisoscatter`numcollorders, 1]]}

]
```

Out[85]=



N-th collided Fluence - Approximations

In[86]:=

```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ,
    plotpointsϕ, logplotϕ, plotϕ, exact1points, numorders},
   data = SelectFirst[inf3Disoplaneisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];
   numorders = data[[2, 13]];

   pointsϕ = data[[9 + numorders + n + 1]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disoplaneisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   seriesclassical = α^n SeriesCoefficient[
      inf3Disoplaneisoscatter`ϕDiffusion[z, Σt, C], {C, 0, n}];
   seriesG = α^n SeriesCoefficient[inf3Disoplaneisoscatter`ϕGrosjean[z, Σt, C],
      {C, 0, n}];

   numpoints = Length[plotpointsϕ];
   skip = Floor[numpoints 6/7 1/2];

   plotϕ = Quiet[Show[
      Plot[inf3Disoplaneisoscatter`ϕGaussian[z, Σt, α, n],
       {z, -maxz/7, maxz/7}, PlotRange → All],
      Plot[seriesclassical, {z, -maxz/7, maxz/7}, PlotRange → All,
       PlotStyle → {Red, Dashed}],
      Plot[seriesG, {z, -maxz/7, maxz/7}, PlotRange → All,
       PlotStyle → {DotDashed, Purple}],
      ListPlot[plotpointsϕ[[skip ;; -skip]], PlotRange → All,
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ϕ["z|" <> ToString[n]],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      LogPlot[inf3Disoplaneisoscatter`ϕGaussian[z, Σt, α, n],
       {z, -maxz, maxz}, PlotRange → {10^-8, 1}],
```

```
      LogPlot[seriesclassical, {z, -maxz, maxz},
       PlotRange → {10^-8, 1}, PlotStyle → {Red, Dashed}],
      LogPlot[seriesG, {z, -maxz, maxz}, PlotRange → {10^-8, 1},
       PlotStyle → {DotDashed, Purple}],
      ListLogPlot[plotpointsϕ, PlotRange → {10^-8, 1},
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
     ]];
  Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 1000],
   PlotLabel -> "Approximate solutions, Gaussian (thin) Classical
       Diffusion (Dashed) Grosjean (Dot-Dashed)\nInfinite
       3D medium, isotropic planesource, isotropic
       scattering, n-th scattered fluence ϕ[z|" <>
     ToString[n] <> "], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]]
 ]
 ,
 Text[
  "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
    ensure the data path is setup correctly."]
]
, {{α, 0.9}, inf3Disoplaneisoscatter`alphas},
{Σt, inf3Disoplaneisoscatter`muts},
{{n, 3}, Range[If[NumberQ[inf3Disoplaneisoscatter`numcollorders],
   inf3Disoplaneisoscatter`numcollorders, 1]]}]
```

Out[86]=

α 0.9

Σt 1 3

n 3

Approximate solutions, Gaussian (thin) Classical Diffusion (Dashed)
Infinite 3D medium, isotropic planesource, isotropic scattering, n−th scatter





Fluence Moments

Moments of the total fluence

In[128]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[
      inf3Disoplaneisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 15]];
   ϕmoments = N[{data[[6]]/Σt}];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic =
    {Table[Re[inf3Disoplaneisoscatter`ϕm[α, Σt, m]], {m, 0, nummoments - 1}]};
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.95}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 3}, inf3Disoplaneisoscatter`muts}]
```

Out[128]=

α 0.95

Σt 1 3

k	analytic	MC
0	6.66667	6.6763
1	0.	−0.00273606
2	9.87654	9.91067
3	0	0.0239086
4	91.3032	91.6733

Moments of the n-th collided fluence

In[130]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[
      inf3Disoplaneisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 15]];
   ϕmoments = N[{data[[9 + n]]}/Σt];
   ks = Table[k, {k, 0, nummoments - 1}];
   analytic = Table[Re[inf3Disoplaneisoscatter`ϕm[α, Σt, k, n]], {k, ks}];
   j = Join[{ks}, {analytic}, ϕmoments];

   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.95}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 3}, inf3Disoplaneisoscatter`muts},
 {{n, 11}, Range[If[NumberQ[inf3Disoplaneisoscatter`numcollorders],
    inf3Disoplaneisoscatter`numcollorders, 1]]}
]
```

Out[130]=

α 0.999

Σt 1 3

n 19

k	analytic	MC
0	0.327057	0.327087
1	0.	−0.000693697
2	0.484528	0.486047
3	0.	−0.00758363
4	2.43341	2.42987

Angular Distributions

Exact solution (Caseology)

In[87]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, numorders, pointsu, plotpointsu, du, r, dz, maxz, zsim},
   data = SelectFirst[
      inf3Disoplaneisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   numorders = data[[2, 13]];
   du = data[[2, 9]];
   dz = data[[2, 7]];
   maxz = data[[2, 5]];

   pointsu  = data[[9 + 2 numorders + n]];

   zsim  = dz * n - 0.5 dz - maxz;

   plotpointsu = inf3Disoplaneisoscatter`ppointsu[pointsu, du, Σt];
   pp = Show[
     ListPlot[plotpointsu, PlotRange → All, PlotStyle → Black,
      Frame → True,
      FrameLabel -> {{ Pi L[zsim, u],}, {u,}}],
     Plot[Pi inf3Disoplaneisoscatter`LrigourousDiffusion[zsim, u, Σt, α],
      {u, -1, 1}, PlotStyle → {Red, Dashed}
     ],
     Plot[Pi inf3Disoplaneisoscatter`Lexact[zsim, u, Σt, α], {u, -1, 1}
     ],
     PlotLabel -> "Infinite 3D Medium, isotropic plane
          source, isotropic scattering\nAngular distribution:
          rigorous diffusion approximation\nπ L[z,u], z = " <>
       ToString[zsim] <> ", α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]
    ]

  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 1}, inf3Disoplaneisoscatter`muts}, {{n, 127}, Range[If[
    NumberQ[inf3Disoplaneisoscatter`numz], inf3Disoplaneisoscatter`numz, 1]]}]
```

Out[87]=



Exact solution (Fourier Transform)

In[88]:=
```
Manipulate[
 If[Length[inf3Disoplaneisoscatter`simulations] > 0,
  Module[{data, numorders, pointsu, plotpointsu, du, r, dz, maxz, zsim},
   data = SelectFirst[
      inf3Disoplaneisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   numorders = data[[2, 13]];
   du = data[[2, 9]];
   dz = data[[2, 7]];
   maxz = data[[2, 5]];

   pointsu  = data[[9 + 2 numorders + n]];

   zsim = dz * n - 0.5 dz - maxz;

   plotpointsu = inf3Disoplaneisoscatter`ppointsu[pointsu, du, Σt];
   Show[
    ListPlot[plotpointsu, PlotRange → All, PlotStyle → Black,
     Frame → True,
     FrameLabel -> {{ Pi L[zsim, u],}, {u,}}],
    Plot[Pi inf3Disoplaneisoscatter`LexactFourier[zsim, u, Σt, α], {u, -1, 1}],
    PlotLabel -> "Infinite 3D Medium, isotropic plane
        source, isotropic scattering\nAngular distribution:
        Exact Fourier Transform solution\nπ L[z,u], z = " <>
      ToString[zsim] <> ", α = " <> ToString[α] <> ", Σₜ = " <> ToString[Σt]
   ]

  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.5}, inf3Disoplaneisoscatter`alphas},
 {{Σt, 3}, inf3Disoplaneisoscatter`muts}, {{n, 122}, Range[If[
    NumberQ[inf3Disoplaneisoscatter`numz], inf3Disoplaneisoscatter`numz, 1]]}]
```

Out[88]=

α 0.5

Σt 1 3

n 122