Semi-Infinite Rod, Albedo Problem, Anisotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[141]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential Random Flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)
g = 'mean cosine' of scattering

Analytic solutions

Half rod reflectance/albedo (R)

In[143]:=
```
Clear[α, g]; halfrodalbedoanisoscatter`R[α_, g_] := α (1 - g) / (-α - α g + 2 √((1 - α) (1 - α g)) + 2)
```

In[144]:=
```
Series[halfrodalbedoanisoscatter`R[α, g], {α, 0, 3}]
```

Out[144]=
$$\left(\frac{1}{4}-\frac{g}{4}\right)\alpha+\left(\frac{1}{8}-\frac{g^2}{8}\right)\alpha^2+\frac{1}{64}\left(5+g-g^2-5\,g^3\right)\alpha^3+O[\alpha]^4$$

In[145]:=
```
halfrodalbedoanisoscatter`R[α_, g_, n_] :=
 α^n (SeriesCoefficient[halfrodalbedoanisoscatter`R[A, g], {A, 0, n}] /. A → α)
```

'Radiance'

In[146]:=
```
halfrodalbedoanisoscatter`LR[x_, α_, Σt_, g_] := e^(- Σt x √((1-α) (1-α g)))
```

In[147]:=
```
halfrodalbedoanisoscatter`LL[x_, α_, Σt_, g_] :=
  α (1 - g) E^(-Σt x √((1-α) (1-α g))) / (-α (2 √((1-α g)/(1-α)) + g + 1) + 2 (√((1-α g)/(1-α)) + 1))
```

Fluence

In[148]:=
```
halfrodalbedoanisoscatter`ϕ[x_, α_, Σt_, g_] :=
 halfrodalbedoanisoscatter`LR[x, α, Σt, g] +
  halfrodalbedoanisoscatter`LL[x, α, Σt, g]
```

n-th collided fluence

In[149]:=
```
halfrodalbedoanisoscatter`ϕ[x_, α_, Σt_, g_, n_] := α^n
  (SeriesCoefficient[halfrodalbedoanisoscatter`ϕ[x, A, Σt, g], {A, 0, n}] /. A → α)
```

moments

In[150]:=
```
halfrodalbedoanisoscatter`ϕm[α_, Σt_, k_, g_] := Σt^(-1-k)
  ((-1 + α) (-1 + g α))^(1/2 (-1-k)) (1 + (-1 + g) α / (-2 (1 + √((-1+g α)/(-1+α))) + α (1 + g + 2 √((-1+g α)/(-1+α))))) Gamma[1 + k]
```

In[151]:=
```
halfrodalbedoanisoscatter`ϕm[α_, Σt_, k_, g_, n_] := α^n (SeriesCoefficient[
    halfrodalbedoanisoscatter`ϕm[A, Σt, k, g], {A, 0, n}] /. A → α)
```

load MC data

In[152]:=
```
halfrodalbedoanisoscatter`ppoints[xs_, dx_, maxx_, Σt_] :=
  Table[{dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]];
```

In[153]:=
```
halfrodalbedoanisoscatter`fs =
  FileNames["code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem
     _anisotropicscatter_exp*"];
```

In[154]:=
```
halfrodalbedoanisoscatter`index[x_] := Module[{data, α, Σt, g},
   data = Import[x, "Table"];
   Σt = data[[1, 11]];
   α = data[[2, 3]];
   g = data[[1, -1]];
   {α, Σt, g, data}];
halfrodalbedoanisoscatter`simulations =
  halfrodalbedoanisoscatter`index /@ halfrodalbedoanisoscatter`fs;
```

In[156]:=
```
halfrodalbedoanisoscatter`alphas =
 Union[#[[1]] & /@ halfrodalbedoanisoscatter`simulations]
```

Out[156]=
```
{0.1, 0.3, 0.5, 0.7, 0.9, 0.95, 0.98, 0.99, 0.999}
```

In[157]:=
```
halfrodalbedoanisoscatter`muts =
 Union[#[[2]] & /@ halfrodalbedoanisoscatter`simulations]
```
Out[157]= {1, 3}

In[158]:=
```
halfrodalbedoanisoscatter`gs =
 Union[#[[3]] & /@ halfrodalbedoanisoscatter`simulations]
```
Out[158]= {-0.9, -0.7, -0.5, -0.3, -0.1, 0.1, 0.3, 0.5, 0.7, 0.9}

In[159]:=
```
halfrodalbedoanisoscatter`numcollorders =
 halfrodalbedoanisoscatter`simulations[[1]][[4]][[2, 11]]
```
Out[159]= 20

Monte Carlo vs Analytic

Albedo (reflectance) variation in g

In[160]:=
```
fsgvariation =
  FileNames["code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem
     _anisotropicscatter_exp_c0.95_mut1.0_*"];
```

In[161]:=
```
RMC[filename_] := Module[{data},
  data = Import[filename, "Table"];
  {data[[1, -1]], data[[3, 3]]}
 ]
```

In[162]:=
```
gvariationpoints = Table[RMC[f], {f, fsgvariation}];
```

In[163]:=
```
Clear[g]; vizgvariation = Show[
  Plot[halfrodalbedoanisoscatter`R[0.95, g], {g, -0.9, 0.9}],
  ListPlot[gvariationpoints, PlotStyle → {PointSize[Medium], Black}],
  Frame → True, FrameLabel → {{"R"[0.95, g],}, {"g",
     "Total Reflectance/Albedo R(α,g): anisotropically-scattering half rod"}}
 ]
```
Out[163]=



n-th collided albedo

In[171]:=
```
Manipulate[
 If[Length[halfrodalbedoanisoscatter`simulations] > 0,
  Module[{data, numcollorders, Rs, ns, analytic, j},
   data = SelectFirst[halfrodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
   numcollorders = data[[2, 11]];
   Rs = N[{data[[5]]}];
   ns = Table[n, {n, 0, numcollorders - 1}];
   analytic = Table[halfrodalbedoanisoscatter`R[α, g, n], {n, ns}];
   j = Join[{ns}, {analytic}, Rs];
   TableForm[
    Join[{{"n", "analytic", "MC"}}, Transpose[j]]
   ]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, halfrodalbedoanisoscatter`alphas},
 {Σt, halfrodalbedoanisoscatter`muts}, {{g, 0.3}, halfrodalbedoanisoscatter`gs}]
```

Out[171]=

α 0.9 ∨

Σt 1 3

g 0.3 ∨

n	analytic	MC
0	0.	0.
1	0.1575	0.157528
2	0.0921375	0.0919856
3	0.0578074	0.0577948
4	0.0383885	0.0384093
5	0.0266623	0.0265883
6	0.0191798	0.0192289
7	0.0141834	0.0141904
8	0.0107217	0.0107707
9	0.00825015	0.008224
10	0.00644158	0.006426
11	0.00509098	0.0051127
12	0.00406508	0.0040554
13	0.00327455	0.0032918
14	0.00265785	0.002671
15	0.00217163	0.0021557
16	0.00178473	0.001781
17	0.00147437	0.0014765
18	0.00122362	0.0012167
19	0.00101973	0.0010099

Internal distributions

In[174]:=
```
Clear[alpha, Σt, g];
Manipulate[
 If[Length[halfrodalbedoanisoscatter`simulations] > 0,
  Module[{data, maxx, dx, numcollorders, nummoments, pointsCL, plotpointsCL,
```

```
   pointsCR, plotpointsCR, plotpointsϕ, plotLL, plotLR, plotϕ},
  data = SelectFirst[halfrodalbedoanisoscatter`simulations,
     #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  maxx = data[[2, 5]];
  dx = data[[2, 7]];
  numcollorders = data[[2, 11]];
  nummoments = data[[2, 13]];

  pointsCL = data[[7]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = halfrodalbedoanisoscatter`ppoints[pointsCL, dx, maxx, Σt];
  pointsCR = data[[9]];
  plotpointsCR = halfrodalbedoanisoscatter`ppoints[pointsCR, dx, maxx, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ =
   halfrodalbedoanisoscatter`ppoints[pointsCL + pointsCR, dx, maxx, Σt];

  plotϕ = Show[
    ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoanisoscatter`ϕ[x, α, Σt, g], {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel ->
     {{ϕ[x],}, {x, "Semi-infinite rod, albedo problem, anisotropic
          scattering, fluence ϕ[x], α = " <> ToString[α] <>
        ", Σt = " <> ToString[Σt] <> ", g = " <> ToString[g]}}
   ];

  plotLL = Show[
    ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoanisoscatter`LL[x, α, Σt, g],
     {x, 0, maxx}, PlotRange → All],
    Plot[halfrodalbedoanisoscatter`ϕ[x, α, Σt, g], {x, 0, maxx},
     PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel ->
     {{L-[x],}, {x, "Semi-infinite rod, albedo problem, anisotropic
          scattering, L-[x], α = " <> ToString[α] <> ", Σt = " <>
        ToString[Σt] <> ", g = " <> ToString[g]}}, PlotRange → All
   ];

  plotLR = Show[
    ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoanisoscatter`LR[x, α, Σt, g],
     {x, 0, maxx}, PlotRange → All],
    Plot[halfrodalbedoanisoscatter`ϕ[x, α, Σt, g], {x, 0, maxx},
     PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel ->
     {{L+[x],}, {x, "Semi-infinite rod, albedo problem, anisotropic
          scattering, L+[x], α = " <> ToString[α] <> ", Σt = " <>
        ToString[Σt] <> ", g = " <> ToString[g]}}, PlotRange → All
   ];
  Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
 ]
 ,
 Text[
  "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
    ensure the data path is setup correctly."]
]
, {{α, 0.99}, halfrodalbedoanisoscatter`alphas},
{{Σt, 3}, halfrodalbedoanisoscatter`muts},
{{g, -0.5}, halfrodalbedoanisoscatter`gs}]
```

Out[175]=

α 0.99

Σt 1 3

g −0.5









n-th collided radiance/angular flux

In[176]:=
```
Manipulate[
 If[Length[halfrodalbedoanisoscatter`simulations] > 0,
  Module[{data, nthL, nthR, maxx, dx, numcollorders, nummoments, LnR},
   data = SelectFirst[halfrodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   numcollorders = data[[2, 11]];
   nummoments = data[[2, 13]];
   nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
   nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

   Clear[c];
   LnR =
    SeriesCoefficient[halfrodalbedoanisoscatter`LR[x, c, Σt, g], {c, 0, n}] α^n;
   Show[
    ListPlot[halfrodalbedoanisoscatter`ppoints[nthR[[n + 1]], dx, maxx, Σt],
     PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LnR, {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{L+["x|" <> ToString[n]],},
      {x, "Semi-Infinite Rod, albedo problem, anisotropic scattering, L+[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt] <>
        ", g = " <> ToString[g]}}, PlotRange → All, ImageSize → 500
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
]
, {{α, 0.9}, halfrodalbedoanisoscatter`alphas},
{Σt, halfrodalbedoanisoscatter`muts}, {g, halfrodalbedoanisoscatter`gs},
{{n, 6}, Range[If[NumberQ[halfrodalbedoanisoscatter`numcollorders],
   halfrodalbedoanisoscatter`numcollorders, 1]]}]
```

Out[176]=

α 0.9

Σt 1 3

g −0.9

n 6



N-th order Fluence / scalar flux

In[177]:=
```
Manipulate[
 If[Length[halfrodalbedoanisoscatter`simulations] > 0,
  Module[{data, maxx, dx, numcollorders, nummoments, nthL, nthR},
   data = SelectFirst[halfrodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   numcollorders = data[[2, 11]];
   nummoments = data[[2, 13]];
   nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
   nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

   Show[
    ListPlot[halfrodalbedoanisoscatter`ppoints[nthR[[n + 1]] + nthL[[n + 1]],
      dx, maxx, Σt], PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoanisoscatter`ϕ[x, α, Σt, g, n],
     {x, 0, maxx}, PlotRange → All]
    , Frame → True, ImageSize → 500,
    FrameLabel -> {{ϕ["x|7"],},
      {x, "Semi-Infinite Rod, albedo problem, anisotropic scattering, ϕ[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <> ", Σₜ = " <>
        ToString[Σt] <> ", g = " <> ToString[g]}}, PlotRange → All
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, halfrodalbedoanisoscatter`alphas},
 {Σt, halfrodalbedoanisoscatter`muts}, {g, halfrodalbedoanisoscatter`gs},
 {{n, 4}, Range[If[NumberQ[halfrodalbedoanisoscatter`numcollorders],
    halfrodalbedoanisoscatter`numcollorders, 1]]}]
```

Out[177]=

α 0.9

Σt 1 3

g −0.9

n 4



Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

In[169]:=
```
Manipulate[
 If[Length[halfrodalbedoanisoscatter`simulations] > 0,
  Module[{data, ϕmoments, ks, analytic, j, nummoments},
   data = SelectFirst[halfrodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[11]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic = Table[halfrodalbedoanisoscatter`ϕm[α, Σt, k, g], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoanisoscatter`alphas},
 {Σt, halfrodalbedoanisoscatter`muts}, {g, halfrodalbedoanisoscatter`gs}]
```



Out[169]=

k	analytic	MC
0	1.05795	1.05794
1	1.06814	1.06807
2	2.15687	2.15666
3	6.53298	6.53162
4	26.3838	26.3768
5	133.19	133.191
6	806.843	807.646
7	5702.34	5721.85
8	46 058.3	46 460.1
9	418 520.	426 764.

n-th collided moments of ϕ

In[170]:=
```
Manipulate[
 If[Length[halfrodalbedoanisoscatter`simulations] > 0,
  Module[{data, ϕmoments, ks, analytic, j, nummoments},
   data = SelectFirst[halfrodalbedoanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[13 + n]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic =
    Table[Quiet[N[halfrodalbedoanisoscatter`ϕm[α, Σt, k, g, n]]], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoanisoscatter`alphas},
 {Σt, halfrodalbedoanisoscatter`muts}, {g, halfrodalbedoanisoscatter`gs},
 {n, Range[If[NumberQ[halfrodalbedoanisoscatter`numcollorders],
    halfrodalbedoanisoscatter`numcollorders, 1]]}]
```

Out[170]=

α 0.1

Σt 1 3

g −0.9

n 1

k	analytic	MC
0	0.0525	0.0524846
1	0.0575	0.0574576
2	0.125	0.12483
3	0.405	0.40423
4	1.74	1.73758
5	9.3	9.30941
6	59.4	59.7446
7	441.	446.754
8	3729.6	3813.62
9	35 380.8	36 579.4