Half Rod, Albedo Problem, Isotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)

Analytic solutions

Half rod reflectance/albedo (R)

```
Clear[α, g];
```

```
halfrodalbedoisoscatter`R[α_] := 2 (-α/2 - Sqrt[1 - α] + 1)/α
```

```
Series[halfrodalbedoisoscatter`R[α], {α, 0, 5}]
```

$$\frac{\alpha}{4}+\frac{\alpha^2}{8}+\frac{5\,\alpha^3}{64}+\frac{7\,\alpha^4}{128}+\frac{21\,\alpha^5}{512}+O[\alpha]^6$$

```
halfrodalbedoisoscatter`R[α_, n_] := α^n 2 (-1)^n Binomial[1 / 2, n + 1]
```

Internal distribution, 'radiance'

```
halfrodalbedoisoscatter`LR[x_, α_, Σt_] := e^(-√(1-α) Σt x)

halfrodalbedoisoscatter`LL[x_, α_, Σt_] := (α Exp[-Σt √(1-α) x]) / (-α + 2 √(1-α) + 2)
```

Fluence

```
halfrodalbedoisoscatter`ϕ[x_, α_, Σt_] :=
 halfrodalbedoisoscatter`LR[x, α, Σt] + halfrodalbedoisoscatter`LL[x, α, Σt]
```

n-th collided fluence

```
halfrodalbedoisoscatter`ϕ[x_, α_, Σt_, n_] :=
 α^n (SeriesCoefficient[halfrodalbedoisoscatter`ϕ[x, A, Σt], {A, 0, n}] /. A → α)
```

Moments

```
halfrodalbedoisoscatter`ϕm[α_, Σt_, k_] :=
 (2 (1 - α)^(-1-k/2) (-1 + √(1-α) + α) Σt^(-1-k) Gamma[1 + k]) / α
```

Only accurate for n even

```
halfrodalbedoisoscatter`ϕm[α_, Σt_, k_, n_] :=
 α^n (2 (-1)^n Σt^(-1-k) Gamma[1 + k] (-(2 + n) Gamma[(1 - k)/2] -
      ((k + n) Binomial[-k/2, 1 + n] + 2 (2 + n) Binomial[-k/2, 2 + n])
       Gamma[-1/2 - k/2 - n] Gamma[2 + n])) / (Gamma[-1/2 - k/2 - n] Gamma[3 + n])
```

load MC data

```
halfrodalbedoisoscatter`ppoints[xs_, dx_, maxx_, Σt_] :=
 Table[{dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]

halfrodalbedoisoscatter`fs = FileNames[
   "code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem_isotropicscatter
     _exp*"];

halfrodalbedoisoscatter`index[x_] := Module[{data, α, Σt},
   data = Import[x, "Table"];
   Σt = data[[1, 11]];
   α = data[[2, 3]];
   {α, Σt, data}];
halfrodalbedoisoscatter`simulations =
  halfrodalbedoisoscatter`index /@ halfrodalbedoisoscatter`fs;

halfrodalbedoisoscatter`alphas =
 Union[#[[1]] & /@ halfrodalbedoisoscatter`simulations]
```

```
{0.1, 0.3, 0.5, 0.7, 0.9, 0.95, 0.98, 0.99, 0.999}
```

```
halfrodalbedoisoscatter`muts =
 Union[#[[2]] & /@ halfrodalbedoisoscatter`simulations]
```

```
{1, 3}
```

```
halfrodalbedoisoscatter`numcollorders =
 halfrodalbedoisoscatter`simulations[[1]][[3]][[2, 11]]
```

```
20
```

Halfrod Albedo

```
MCalbedo[f_] := Module[{data, α},
  data = Import[f, "Table"];
  α = data[[2, 3]];
  {α, data[[3, 3]]}
 ]
```

```
MCalbedos = Table[MCalbedo[f], {f, halfrodalbedoisoscatter`fs}]
```

```
{{0.1, 0.0262874}, {0.1, 0.0262874}, {0.3, 0.0888128},
 {0.3, 0.0888128}, {0.5, 0.17156}, {0.5, 0.17156}, {0.7, 0.291991},
 {0.7, 0.291991}, {0.95, 0.633904}, {0.95, 0.633904}, {0.98, 0.751703},
 {0.98, 0.751703}, {0.999, 0.938793}, {0.999, 0.938793},
 {0.99, 0.818082}, {0.99, 0.818082}, {0.9, 0.519448}, {0.9, 0.519448}}
```

```
Clear[α]; vizrodalbedoiso = Show[
  Plot[halfrodalbedoisoscatter`R[c], {c, 0, 1}],
  ListPlot[MCalbedos]
  , Frame → True, ImageSize → 500,
  FrameLabel → {{"R[α]",}, {"Single scattering albedo: α",
     "Total Reflectance/Albedo R(α): isotropically-scattering half rod"}}
 ]
```



n-th collided albedo

```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, Rs, ns, analytic, j, numcollorders},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   numcollorders = data[[2, 11]];
   Rs = N[{data[[5]]}];
   ns = Table[n, {n, 0, numcollorders - 1}];
   analytic = Table[halfrodalbedoisoscatter`R[α, n], {n, ns}];
   j = Join[{ns}, {analytic}, Rs];
   TableForm[
    Join[{{"n", "analytic", "MC"}}, Transpose[j]]
   ]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.95}, halfrodalbedoisoscatter`alphas},
 {Σt, halfrodalbedoisoscatter`muts}]
```

α 0.95

Σt 1 3

n	analytic	MC
0	1.	0.
1	0.2375	0.236915
2	0.112813	0.112534
3	0.0669824	0.067129
4	0.0445433	0.044339
5	0.0317371	0.031289
6	0.0236895	0.023728
7	0.0182853	0.018173
8	0.0144759	0.014566
9	0.0116893	0.011592
10	0.00959052	0.009669
11	0.00797212	0.008212
12	0.00669964	0.006757
13	0.00568273	0.005733
14	0.00485874	0.004929
15	0.00418307	0.004272
16	0.00362328	0.003667
17	0.00315527	0.003168
18	0.00276086	0.002716
19	0.00242611	0.002433

Compare Deterministic and MC

Internal distribution

```
Clear[alpha, Σt];
Manipulate[
```

```
If[Length[halfrodalbedoisoscatter`simulations] > 0,
 Module[{data, maxx, dx, numcollorders, nummoments, pointsCL, plotpointsCL,
   pointsCR, plotpointsCR, plotpointsϕ, plotϕ, plotLL, plotLR},
  data = SelectFirst[halfrodalbedoisoscatter`simulations,
     #[[1]] == α && #[[2]] == Σt &][[3]];
  maxx = data[[2, 5]];
  dx = data[[2, 7]];
  numcollorders = data[[2, 11]];
  nummoments = data[[2, 13]];

  pointsCL = data[[7]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = halfrodalbedoisoscatter`ppoints[pointsCL, dx, maxx, Σt];
  pointsCR = data[[9]];
  plotpointsCR = halfrodalbedoisoscatter`ppoints[pointsCR, dx, maxx, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ =
   halfrodalbedoisoscatter`ppoints[pointsCL + pointsCR, dx, maxx, Σt];

  plotϕ = Show[
    ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoisoscatter`ϕ[x, α, Σt], {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{ϕ[x],},
      {"x", "Semi-infinite rod, albedo problem, isotropic scattering, fluence
          ϕ[x], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}
   ];

  plotLL = Show[
    ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoisoscatter`LL[x, α, Σt], {x, 0, maxx}, PlotRange → All],
    Plot[halfrodalbedoisoscatter`ϕ[x, α, Σt],
     {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel ->
     {{L-[x],}, {"x", "Semi-infinite rod, albedo problem, isotropic
          scattering, L-[x], α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];

  plotLR = Show[
    ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[halfrodalbedoisoscatter`LR[x, α, Σt], {x, 0, maxx}, PlotRange → All],
    Plot[halfrodalbedoisoscatter`ϕ[x, α, Σt],
     {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel ->
     {{L+[x],}, {"x", "Semi-infinite rod, albedo problem, isotropic
          scattering, L+[x], α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];
  Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
 ]
 ,
 Text[
  "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
    ensure the data path is setup correctly."]
]
, {{α, 0.7}, halfrodalbedoisoscatter`alphas},
{Σt, halfrodalbedoisoscatter`muts}]
```

α 0.7

Σt 1 3


Semi−infinite rod, albedo problem, isotropic scattering, fluence ϕ[x], α = 0.7, Σt = 1
ϕ(x)
x
0.0
0.2
0.4
0.6
0.8
1.0
1.2
0
2
4
6
8
10


Semi−infinite rod, albedo problem, isotropic scattering, L−[x], α = 0.7, Σt = 1
L−(x)
x
0.0
0.2
0.4
0.6
0.8
1.0
1.2
0
2
4
6
8
10


Semi−infinite rod, albedo problem, isotropic scattering, L+[x], α = 0.7, Σt = 1
L+(x)
0.0
0.2
0.4
0.6
0.8
1.0
1.2



N-th order Radiance/Angular flux

```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, nthL, nthR, maxx, dx, numcollorders, LnR},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   numcollorders = data[[2, 11]];
   nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
   nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

   Clear[c, x];
   LnR = SeriesCoefficient[halfrodalbedoisoscatter`LR[x, c, Σt], {c, 0, n}] α^n;
   Show[
    ListPlot[halfrodalbedoisoscatter`ppoints[nthR[[n + 1]], dx, maxx, Σt],
     PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LnR, {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{L+["x|" <> ToString[n]],},
      {x, "Semi-Infinite Rod, albedo problem, isotropic scattering, L+[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <>
        ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, halfrodalbedoisoscatter`alphas},
 {Σt, halfrodalbedoisoscatter`muts},
 {{n, 6}, Range[If[NumberQ[halfrodalbedoisoscatter`numcollorders],
    halfrodalbedoisoscatter`numcollorders, 1]]}]
```

α 0.9

Σt 1 3

n 6


Semi−Infinite Rod, albedo problem, isotropic scattering, $L_+[x|6]$, $\alpha = 0.9$, $\Sigma_t = 1$
$L_+(x|6)$
0.020
0.015
0.010
0.005
0.000
0
2
4
6
8
10
x

N-th order Fluence / scalar flux

```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, maxx, dx, numcollorders, nthL, nthR, ϕn},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   numcollorders = data[[2, 11]];
   nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
   nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

   Clear[c];
   ϕn = SeriesCoefficient[halfrodalbedoisoscatter`ϕ[x, c, Σt], {c, 0, n}] α^n;
   Show[
    ListPlot[halfrodalbedoisoscatter`ppoints[nthR[[n + 1]] + nthL[[n + 1]],
      dx, maxx, Σt], PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[ϕn, {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{L+["x|" <> ToString[n]],},
      {x, "Semi-Infinite Rod, albedo problem, isotropic scattering, ϕ[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <>
        ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, halfrodalbedoisoscatter`alphas},
 {Σt, halfrodalbedoisoscatter`muts},
 {{n, 6}, Range[If[NumberQ[halfrodalbedoisoscatter`numcollorders],
    halfrodalbedoisoscatter`numcollorders, 1]]}]
```



Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[11]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic = Table[halfrodalbedoisoscatter`ϕm[α, Σt, k], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts}]
```

α 0.1

Σt 1 3

k	analytic	MC
0	1.08185	1.08187
1	1.14037	1.13988
2	2.40411	2.40269
3	7.60247	7.59746
4	32.0548	32.0322
5	168.944	168.744
6	1068.49	1064.98
7	7884.05	7810.61
8	66 484.1	64 980.6
9	630 724.	600 952.

n-th collided moments of ϕ

```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, ϕmoments, ks, analytic, j, nummoments},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[13 + n]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic =
    Table[Quiet[N[halfrodalbedoisoscatter`ϕm[α, Σt, k, n]]], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.7}, halfrodalbedoisoscatter`alphas},
 {Σt, halfrodalbedoisoscatter`muts},
 {{n, 4}, Range[If[NumberQ[halfrodalbedoisoscatter`numcollorders],
    halfrodalbedoisoscatter`numcollorders, 1]]}]
```



k	analytic	MC
0	0.118174	0.11819
1	Indeterminate	0.362112
2	1.63943	1.64089
3	Indeterminate	9.50676
4	66.0425	66.2885
5	Indeterminate	537.415
6	4906.59	4950.81
7	Indeterminate	50 953.1
8	570 488.	577 954.
9	Indeterminate	7.14248×10^6