Half Rod, Albedo Problem, Isotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

© 2015 Eugene d'Eon

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[1]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential Random Flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)

Analytic solutions

Half rod reflectance/albedo (R)

In[2]:=
```
Clear[α, g]; R[α_] := 2 (-α/2 - Sqrt[1 - α] + 1)/α
```

In[3]:=
```
Series[R[α], {α, 0, 5}]
```

Out[3]=
$$\frac{\alpha}{4}+\frac{\alpha^2}{8}+\frac{5\,\alpha^3}{64}+\frac{7\,\alpha^4}{128}+\frac{21\,\alpha^5}{512}+O[\alpha]^6$$

In[4]:=
```
R[α_, n_] := α^n (SeriesCoefficient[R[A], {A, 0, n}] /. A → α)
```

Internal distribution, 'radiance'

In[5]:=
```
LR[x_, α_, Σt_] := e^(-Sqrt[1-α] Σt x)
```

In[6]:=
```
LL[x_, α_, Σt_] := α Exp[-Σt Sqrt[1 - α] x]/(-α + 2 Sqrt[1 - α] + 2)
```

Fluence

In[7]:=
```
ϕ[x_, α_, Σt_] := LR[x, α, Σt] + LL[x, α, Σt]
```

n-th collided fluence

In[8]:=
```
ϕ[x_, α_, Σt_, n_] := α^n (SeriesCoefficient[ϕ[x, A, Σt], {A, 0, n}] /. A → α)
```

Moments

In[9]:=
$$\phi m[\alpha\_, \Sigma t\_, k\_] := \frac{2\,(1-\alpha)^{-1-\frac{k}{2}}\left(-1+\sqrt{1-\alpha}+\alpha\right)\Sigma t^{-1-k}\,\text{Gamma}[1+k]}{\alpha}$$

Only accurate for n even

In[10]:=
$$\phi m[\alpha\_, \Sigma t\_, k\_, n\_] := \alpha^n \left(2\,(-1)^n\,\Sigma t^{-1-k}\,\text{Gamma}[1+k]\left(-(2+n)\,\text{Gamma}\left[\frac{1-k}{2}\right] - \left((k+n)\,\text{Binomial}\left[-\frac{k}{2}, 1+n\right] + 2\,(2+n)\,\text{Binomial}\left[-\frac{k}{2}, 2+n\right]\right)\text{Gamma}\left[-\frac{1}{2}-\frac{k}{2}-n\right]\text{Gamma}[2+n]\right)\Big/\left(\text{Gamma}\left[-\frac{1}{2}-\frac{k}{2}-n\right]\text{Gamma}[3+n]\right)$$

load MC data

In[11]:=
```
ppoints[xs_, dx_, maxx_, Σt_] :=
 Table[{dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[12]:=
```
fs = FileNames[
    "code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem_isotropicscatter
      _exp*"];
```

In[13]:=
```
index[x_] := Module[{data, α, Σt},
    data = Import[x, "Table"];
    Σt = data[[1, 11]];
    α = data[[2, 3]];
    {α, Σt, data}];
simulations = index /@ fs;
```

In[15]:=
```
alphas = Union[#[[1]] & /@ simulations]
```

Out[15]=
```
{0.1, 0.3, 0.5, 0.7, 0.9, 0.95, 0.98, 0.99, 0.999}
```

In[16]:=
```
muts = Union[#[[2]] & /@ simulations]
```

Out[16]=
```
{1, 3}
```

Halfrod Albedo

In[17]:=
```
MCalbedo[f_] := Module[{data, α},
   data = Import[f, "Table"];
   α = data[[2, 3]];
   {α, data[[3, 3]]}
  ]
```

In[18]:=
```
MCalbedos = Table[MCalbedo[f], {f, fs}]
```

Out[18]=
```
{{0.1, 0.0262874}, {0.1, 0.0262874}, {0.3, 0.0888128},
 {0.3, 0.0888128}, {0.5, 0.17156}, {0.5, 0.17156}, {0.7, 0.291991},
 {0.7, 0.291991}, {0.95, 0.633904}, {0.95, 0.633904}, {0.98, 0.751703},
 {0.98, 0.751703}, {0.999, 0.938793}, {0.999, 0.938793},
 {0.99, 0.818082}, {0.99, 0.818082}, {0.9, 0.519448}, {0.9, 0.519448}}
```

In[19]:=
```
Clear[α]; vizrodalbedoiso = Show[
  Plot[R[c], {c, 0, 1}],
  ListPlot[MCalbedos]
  , Frame → True, FrameLabel → {{"R[α]",}, {"Single scattering albedo: α",
     "Total Reflectance/Albedo R(α): isotropically-scattering half rod"}}
 ]
```

Out[19]=



Compare Deterministic and MC

Internal distributions

In[20]:=
```
Clear[alpha, Σt];
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  maxx = data[[2, 5]];
  dx = data[[2, 7]];
  numcollorders = data[[2, 11]];
  nummoments = data[[2, 13]];

  densmom = data[[11]];

  pointsCL = data[[7]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = ppoints[pointsCL, dx, maxx, Σt];
  pointsCR = data[[9]];
  plotpointsCR = ppoints[pointsCR, dx, maxx, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ = ppoints[pointsCL + pointsCR, dx, maxx, Σt];

  plotϕ = Show[
    ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[ϕ[x, α, Σt], {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{ϕ[x],},
      {x, "Semi-infinite rod, albedo problem, isotropic scattering, fluence
          ϕ[x], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}
   ];

  plotLL = Show[
    ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LL[x, α, Σt], {x, 0, maxx}, PlotRange → All],
    Plot[ϕ[x, α, Σt], {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L-[x],},
      {x, "Semi-infinite rod, albedo problem, isotropic scattering, L-[x], α =
          " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];

  plotLR = Show[
    ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LR[x, α, Σt], {x, 0, maxx}, PlotRange → All],
    Plot[ϕ[x, α, Σt], {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L+[x],},
      {x, "Semi-infinite rod, albedo problem, isotropic scattering, L+[x], α =
          " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];
  Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}]
```

α 0.98

Σt 1 3

Out[21]=









n-th collided albedo

In[22]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  Rs = N[{data[[5]]}];
  ns = Table[n, {n, 0, numcollorders - 1}];
  analytic = Table[R[α, n], {n, ns}];
  j = Join[{ns}, {analytic}, Rs];
  TableForm[
   Join[{{"n", "analytic", "MC"}}, Transpose[j]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}]
```

Out[22]=

α 0.98

Σt 1 3

n	analytic	MC
0	0.	0.
1	0.245	0.244497
2	0.12005	0.120324
3	0.0735306	0.074076
4	0.050442	0.050198
5	0.0370749	0.037007
6	0.0285477	0.028276
7	0.0227311	0.022734
8	0.0185637	0.018692
9	0.0154636	0.015586
10	0.0130878	0.013021
11	0.0112228	0.011227
12	0.0097293	0.00972
13	0.00851313	0.008508
14	0.00750858	0.007559
15	0.00666856	0.006809
16	0.00595856	0.005858
17	0.00535277	0.005461
18	0.00483158	0.004784
19	0.00437983	0.004334

Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

In[23]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  ϕmoments = N[{data[[11]]} / Σt];
  ks = {Table[k, {k, 0, nummoments - 1}]};
  analytic = Table[ϕm[α, Σt, k], {k, ks}];
  j = Join[ks, analytic, ϕmoments];
  TableForm[
   Join[{{"k", "analytic", "MC"}}, Transpose[j]]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}]
```

Out[23]=

α 0.95

Σt 1 3

k	analytic	MC
0	7.30976	7.32799
1	32.6902	32.7863
2	292.39	292.548
3	3922.83	3890.96
4	70 173.7	68 215.7
5	1.56913×10^{6}	1.46869×10^{6}
6	4.21042×10^{7}	3.70344×10^{7}
7	1.31807×10^{9}	1.05685×10^{9}
8	4.71567×10^{10}	3.32616×10^{10}
9	1.89802×10^{12}	1.13179×10^{12}

n-th collided moments of ϕ

In[24]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  ϕmoments = N[{data[[13 + n]]} / Σt];
  ks = {Table[k, {k, 0, nummoments - 1}]};
  analytic = Table[Quiet[N[ϕm[α, Σt, k, n]]], {k, ks}];
  j = Join[ks, analytic, ϕmoments];
  TableForm[
   Join[{{"k", "analytic", "MC"}}, Transpose[j]]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[24]=

α 0.7

Σt 1 3

n 4

k	analytic	MC
0	0.118174	0.11819
1	Indeterminate	0.362112
2	1.63943	1.64089
3	Indeterminate	9.50676
4	66.0425	66.2885
5	Indeterminate	537.415
6	4906.59	4950.81
7	Indeterminate	50 953.1
8	570 488.	577 954.
9	Indeterminate	7.14248×10^6

N-th order Radiance/Angular flux

In[25]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
  nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

  Clear[c];
  LnR = FullSimplify[SeriesCoefficient[LR[x, c, Σt], {c, 0, n}] α^n];
  Show[
   ListPlot[ppoints[nthR[[n + 1]], dx, maxx, Σt],
    PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[LnR, {x, 0, maxx}, PlotRange → All]
   , Frame → True,
   FrameLabel -> {{L+["x|" <> ToString[n]],},
     {x, "Semi-Infinite Rod, albedo problem, isotropic scattering, L+[x|" <>
       ToString[n] <> "], α = " <> ToString[α] <>
       ", Σt = " <> ToString[Σt]}}, PlotRange → All
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[25]=



N-th order Fluence / scalar flux

In[26]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
  nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

  Clear[c];
  ϕn = FullSimplify[SeriesCoefficient[ϕ[x, c, Σt], {c, 0, n}] α^n];
  Show[
   ListPlot[ppoints[nthR[[n + 1]] + nthL[[n + 1]], dx, maxx, Σt],
    PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[ϕn, {x, 0, maxx}, PlotRange → All]
   , Frame → True,
   FrameLabel -> {{L+["x|" <> ToString[n]],},
     {x, "Semi-Infinite Rod, albedo problem, isotropic scattering, ϕ[x|" <>
       ToString[n] <> "], α = " <> ToString[α] <>
       ", Σt = " <> ToString[Σt]}}, PlotRange → All
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[26]=