Semi-Infinite Rod, Albedo Problem, Anisotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
In[27]:= SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential Random Flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)
g = 'mean cosine' of scattering

Analytic solutions

Half rod reflectance/albedo (R)

In[28]:= `Clear[α, g]; R[α_, g_] :=` $\frac{\alpha\,(1-g)}{-\alpha-\alpha g+2\sqrt{(1-\alpha)\,(1-\alpha g)}+2}$

In[29]:= `Series[R[α, g], {α, 0, 3}]`

Out[29]= $\left(\frac{1}{4}-\frac{g}{4}\right)\alpha+\left(\frac{1}{8}-\frac{g^2}{8}\right)\alpha^2+\frac{1}{64}\left(5+g-g^2-5g^3\right)\alpha^3+O[\alpha]^4$

In[30]:= `R[α_, g_, n_] :=` α^n `(SeriesCoefficient[R[A, g], {A, 0, n}] /. A → α)`

'Radiance'

In[31]:= `LR[x_, α_, Σt_, g_] :=` $e^{-\Sigma t\, x\sqrt{(1-\alpha)\,(1-\alpha g)}}$

In[32]:= $LL[x\_, \alpha\_, \Sigma t\_, g\_] := \frac{\alpha (1-g) E^{-\Sigma t\, x \sqrt{(1-\alpha)(1-\alpha g)}}}{-\alpha \left(2 \sqrt{\frac{1-\alpha g}{1-\alpha}} + g + 1\right) + 2 \left(\sqrt{\frac{1-\alpha g}{1-\alpha}} + 1\right)}$

Fluence

In[33]:= $\phi[x\_, \alpha\_, \Sigma t\_, g\_] := LR[x, \alpha, \Sigma t, g] + LL[x, \alpha, \Sigma t, g]$

n-th collided fluence

In[34]:= $\phi[x\_, \alpha\_, \Sigma t\_, g\_, n\_] := \alpha^n \left(\text{SeriesCoefficient}[\phi[x, A, \Sigma t, g], \{A, 0, n\}] \text{ /. } A \to \alpha\right)$

moments

In[59]:= $\phi m[\alpha\_, \Sigma t\_, k\_, g\_] := \Sigma t^{-1-k} ((-1+\alpha)(-1+g\alpha))^{\frac{1}{2}(-1-k)} \left(1 + \frac{(-1+g)\alpha}{-2\left(1+\sqrt{\frac{-1+g\alpha}{-1+\alpha}}\right) + \alpha\left(1+g+2\sqrt{\frac{-1+g\alpha}{-1+\alpha}}\right)}\right) \text{Gamma}[1+k]$

In[61]:= $\phi m[\alpha\_, \Sigma t\_, k\_, g\_, n\_] := \alpha^n \left(\text{SeriesCoefficient}[\phi m[A, \Sigma t, k, g], \{A, 0, n\}] \text{ /. } A \to \alpha\right)$

load MC data

```
In[45]:= fs = FileNames["code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem
            _anisotropicscatter_exp*"];

In[46]:= index[x_] := Module[{data, α, Σt, g},
           data = Import[x, "Table"];
           Σt = data[[1, 11]];
           α = data[[2, 3]];
           g = data[[1, -1]];
           {α, Σt, g, data}];
         simulations = index /@ fs;

In[48]:= alphas = Union[#[[1]] & /@ simulations]

Out[48]= {0.1, 0.3, 0.5, 0.7, 0.9, 0.95, 0.98, 0.99, 0.999}

In[49]:= muts = Union[#[[2]] & /@ simulations]

Out[49]= {1, 3}

In[50]:= gs = Union[#[[3]] & /@ simulations]

Out[50]= {-0.9, -0.7, -0.5, -0.3, -0.1, 0.1, 0.3, 0.5, 0.7, 0.9}
```

Monte Carlo vs Analytic

Albedo (reflectance) variation in g

```
In[41]:= fsgvariation =
           FileNames["code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem
             _anisotropicscatter_exp_c0.95_mut1.0_*"];
```

In[42]:=
```
RMC[filename_] := Module[{data},
  data = Import[filename, "Table"];
  {data[[1, -1]], data[[3, 3]]}
 ]
```

In[43]:=
```
gvariationpoints = Table[RMC[f], {f, fsgvariation}];
```

In[44]:=
```
Clear[g]; vizgvariation = Show[
  Plot[R[0.95, g], {g, -0.9, 0.9}],
  ListPlot[gvariationpoints, PlotStyle → {PointSize[Medium], Black}],
  Frame → True, FrameLabel → {{"R"[0.95, g],}, {"g",
     "Total Reflectance/Albedo R(α,g): anisotropically-scattering half rod"}}
 ]
```

Out[44]=



n-th collided albedo

In[53]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  Rs = N[{data[[5]]}];
  ns = Table[n, {n, 0, numcollorders - 1}];
  analytic = Table[R[α, g, n], {n, ns}];
  j = Join[{ns}, {analytic}, Rs];
  TableForm[
   Join[{{"n", "analytic", "MC"}}, Transpose[j]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}, {g, gs}]
```

Out[53]=

α 0.9

Σt 1 3

g 0.5

n	analytic	MC
0	0.	0.
1	0.1125	0.112566
2	0.0759375	0.0759619
3	0.0526816	0.0526925
4	0.0374823	0.0375206
5	0.0272828	0.0272438
6	0.0202648	0.0202852
7	0.015322	0.0152574
8	0.0117657	0.0117876
9	0.00915738	0.0091485
10	0.00721114	0.0071977
11	0.0057366	0.0057307
12	0.00460428	0.0046037
13	0.00372433	0.0037391
14	0.00303328	0.0030306
15	0.00248549	0.0024809
16	0.00204766	0.0020653
17	0.00169512	0.0017075
18	0.00140939	0.0014139
19	0.00117643	0.0011714

Internal distributions

In[51]:=
```
Clear[alpha, Σt, g];
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  maxx = data[[2, 5]];
  dx = data[[2, 7]];
  numcollorders = data[[2, 11]];
  nummoments = data[[2, 13]];

  densmom = data[[11]];

  pointsCL = data[[7]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = ppoints[pointsCL, dx, maxx, Σt];
  pointsCR = data[[9]];
  plotpointsCR = ppoints[pointsCR, dx, maxx, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ = ppoints[pointsCL + pointsCR, dx, maxx, Σt];

  plotϕ = Show[
     ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
     Plot[ϕ[x, α, Σt, g], {x, 0, maxx}, PlotRange → All]
     , Frame → True,
     FrameLabel ->
      {{ϕ[x],}, {x, "Semi-infinite rod, albedo problem, anisotropic scattering,
           fluence ϕ[x], α = " <> ToString[α] <>
          ", Σt = " <> ToString[Σt] <> ", g = " <> ToString[g]}}
   ];

  plotLL = Show[
     ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
     Plot[LL[x, α, Σt, g], {x, 0, maxx}, PlotRange → All],
     Plot[ϕ[x, α, Σt, g], {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
     , Frame → True,
     FrameLabel ->
      {{L-[x],}, {x, "Semi-infinite rod, albedo problem, anisotropic
           scattering, L-[x], α = " <> ToString[α] <> ", Σt = " <>
          ToString[Σt] <> ", g = " <> ToString[g]}}, PlotRange → All
   ];

  plotLR = Show[
     ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
     Plot[LR[x, α, Σt, g], {x, 0, maxx}, PlotRange → All],
     Plot[ϕ[x, α, Σt, g], {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
     , Frame → True,
     FrameLabel ->
      {{L+[x],}, {x, "Semi-infinite rod, albedo problem, anisotropic
           scattering, L+[x], α = " <> ToString[α] <> ", Σt = " <>
          ToString[Σt] <> ", g = " <> ToString[g]}}, PlotRange → All
   ];
  Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}, {g, gs}]
```

α 0.95

Σt 1 3

g 0.5

Out[52]=









n-th collided radiance/angular flux

In[66]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
  nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

  Clear[c];
  LnR = FullSimplify[SeriesCoefficient[LR[x, c, Σt, g], {c, 0, n}] α^n];
  Show[
   ListPlot[ppoints[nthR[[n + 1]], dx, maxx, Σt],
    PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[LnR, {x, 0, maxx}, PlotRange → All]
   , Frame → True,
   FrameLabel -> {{L₊["x|" <> ToString[n]],},
     {x, "Semi-Infinite Rod, albedo problem, anisotropic scattering, L₊[x|" <>
       ToString[n] <> "], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt] <>
       ", g = " <> ToString[g]}}, PlotRange → All, ImageSize → 500
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}, {g, gs},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[66]=


α 0.1
Σt 1 3
g −0.9
n 1
Semi–Infinite Rod, albedo problem, anisotropic scattering, L₊[x|1], α = 0.1, Σₜ = 1, g = −0.9
L₊(x|1)
0.0015
0.0010
0.0005
0.0000
0
2
4
6
8
10
x

N-th order Fluence / scalar flux

In[67]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
  nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

  Show[
   ListPlot[ppoints[nthR[[n + 1]] + nthL[[n + 1]], dx, maxx, Σt],
    PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[ϕ[x, α, Σt, g, n], {x, 0, maxx}, PlotRange → All]
   , Frame → True, ImageSize → 500,
   FrameLabel -> {{ϕ["x|7"],},
     {x, "Semi-Infinite Rod, albedo problem, anisotropic scattering, ϕ[x|" <>
       ToString[n] <> "], α = " <> ToString[α] <> ", Σt = " <>
       ToString[Σt] <> ", g = " <> ToString[g]}}, PlotRange → All
  ],
  Text[
   "Uh oh! Couldn't find MC data. Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}, {g, gs},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[67]=



Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

In[55]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  ϕmoments = N[{data[[11]]}/Σt];
  ks = {Table[k, {k, 0, nummoments - 1}]};
  analytic = Table[ϕm[α, Σt, k, g], {k, ks}];
  j = Join[ks, analytic, ϕmoments];
  TableForm[
   Join[{{"k", "analytic", "MC"}}, Transpose[j]]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}, {g, gs}]
```

α 0.7

Σt 1 3

g 0.5

Out[55]=

k	analytic	MC
0	2.69692	2.6961
1	6.10731	6.10029
2	27.6607	27.5972
3	187.917	187.248
4	1702.19	1694.44
5	19 273.6	19 187.1
6	261 876.	261 277.
7	4.15123×10^{6}	4.16268×10^{6}
8	7.52055×10^{7}	7.59683×10^{7}
9	1.53276×10^{9}	1.55955×10^{9}

n-th collided moments of ϕ

In[63]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt && #[[3]] == g &][[4]];
  ϕmoments = N[{data[[13 + n]]} / Σt];
  ks = {Table[k, {k, 0, nummoments - 1}]};
  analytic = Table[Quiet[N[ϕm[α, Σt, k, g, n]]], {k, ks}];
  j = Join[ks, analytic, ϕmoments];
  TableForm[
   Join[{{"k", "analytic", "MC"}}, Transpose[j]]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}, {g, gs},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

α 0.7

Σt 1 3

g 0.3

n 3

Out[63]=

k	analytic	MC
0	0.21676	0.216829
1	0.647386	0.6475
2	2.82655	2.82672
3	15.6517	15.6511
4	103.863	103.854
5	799.736	799.75
6	6993.62	6997.56
7	68 385.2	68 544.5
8	738 827.	743 823.
9	8.73608×10^{6}	8.87589×10^{6}