Half Rod, Albedo Problem, Isotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[21]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential Random Flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)

Analytic solutions

Half rod reflectance/albedo (R)

In[56]:=
```
Clear[α, g];

halfrodalbedoisoscatter`R[α_] := 2 (-α/2 - Sqrt[1 - α] + 1)/α
```

In[58]:=
```
Series[halfrodalbedoisoscatter`R[α], {α, 0, 5}]
```

Out[58]=
$$\frac{\alpha}{4}+\frac{\alpha^2}{8}+\frac{5\,\alpha^3}{64}+\frac{7\,\alpha^4}{128}+\frac{21\,\alpha^5}{512}+O[\alpha]^6$$

In[66]:=
```
halfrodalbedoisoscatter`R[α_, n_] :=
 α^n (SeriesCoefficient[halfrodalbedoisoscatter`R[A], {A, 0, n}] /. A → α)
```

Internal distribution, 'radiance'

In[60]:=
```
halfrodalbedoisoscatter`LR[x_, α_, Σt_] := e^(-Sqrt[1-α] Σt x)
```

In[61]:=
$$\text{halfrodalbedoisoscatter`LL}[x\_, \alpha\_, \Sigma t\_] := \frac{\alpha\, \text{Exp}\left[-\Sigma t \sqrt{1-\alpha}\, x\right]}{-\alpha + 2\sqrt{1-\alpha} + 2}$$

Fluence

In[62]:=
```
halfrodalbedoisoscatter`ϕ[x_, α_, Σt_] :=
 halfrodalbedoisoscatter`LR[x, α, Σt] + halfrodalbedoisoscatter`LL[x, α, Σt]
```

n-th collided fluence

In[63]:=
```
halfrodalbedoisoscatter`ϕ[x_, α_, Σt_, n_] :=
 α^n (SeriesCoefficient[halfrodalbedoisoscatter`ϕ[x, A, Σt], {A, 0, n}] /. A → α)
```

Moments

In[64]:=
$$\text{halfrodalbedoisoscatter`}\phi\text{m}[\alpha\_, \Sigma t\_, k\_] := \frac{2\,(1-\alpha)^{-1-\frac{k}{2}}\left(-1+\sqrt{1-\alpha}+\alpha\right)\Sigma t^{-1-k}\,\text{Gamma}[1+k]}{\alpha}$$

Only accurate for n even

In[65]:=
$$\text{halfrodalbedoisoscatter`}\phi\text{m}[\alpha\_, \Sigma t\_, k\_, n\_] := \alpha^n \left(2\,(-1)^n \Sigma t^{-1-k}\,\text{Gamma}[1+k]\left(-(2+n)\,\text{Gamma}\left[\frac{1-k}{2}\right] - \left((k+n)\,\text{Binomial}\left[-\frac{k}{2}, 1+n\right] + 2\,(2+n)\,\text{Binomial}\left[-\frac{k}{2}, 2+n\right]\right)\text{Gamma}\left[-\frac{1}{2}-\frac{k}{2}-n\right]\text{Gamma}[2+n]\right)\Big/\left(\text{Gamma}\left[-\frac{1}{2}-\frac{k}{2}-n\right]\text{Gamma}[3+n]\right)$$

load MC data

In[32]:=
```
halfrodalbedoisoscatter`ppoints[xs_, dx_, maxx_, Σt_] :=
 Table[{dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[33]:=
```
halfrodalbedoisoscatter`fs = FileNames[
    "code/rod/halfrod/albedoProblem/data/halfrod_albedoproblem_isotropicscatter
      _exp*"];
```

In[34]:=
```
halfrodalbedoisoscatter`index[x_] := Module[{data, α, Σt},
    data = Import[x, "Table"];
    Σt = data[[1, 11]];
    α = data[[2, 3]];
    {α, Σt, data}];
halfrodalbedoisoscatter`simulations =
  halfrodalbedoisoscatter`index /@ halfrodalbedoisoscatter`fs;
```

In[36]:=
```
halfrodalbedoisoscatter`alphas =
 Union[#[[1]] & /@ halfrodalbedoisoscatter`simulations]
```

Out[36]= {0.1, 0.3, 0.5, 0.7, 0.9, 0.95, 0.98, 0.99, 0.999}

In[37]:=
```
halfrodalbedoisoscatter`muts =
 Union[#[[2]] & /@ halfrodalbedoisoscatter`simulations]
```

Out[37]= {1, 3}

In[38]:=
```
halfrodalbedoisoscatter`numcollorders =
 halfrodalbedoisoscatter`simulations[[1]][[3]][[2, 11]]
```

Out[38]= 20

Halfrod Albedo

In[49]:=
```
MCalbedo[f_] := Module[{data, α},
  data = Import[f, "Table"];
  α = data[[2, 3]];
  {α, data[[3, 3]]}
 ]
```

In[50]:=
```
MCalbedos = Table[MCalbedo[f], {f, halfrodalbedoisoscatter`fs}]
```

Out[50]=
```
{{0.1, 0.0262874}, {0.1, 0.0262874}, {0.3, 0.0888128},
 {0.3, 0.0888128}, {0.5, 0.17156}, {0.5, 0.17156}, {0.7, 0.291991},
 {0.7, 0.291991}, {0.95, 0.633904}, {0.95, 0.633904}, {0.98, 0.751703},
 {0.98, 0.751703}, {0.999, 0.938793}, {0.999, 0.938793},
 {0.99, 0.818082}, {0.99, 0.818082}, {0.9, 0.519448}, {0.9, 0.519448}}
```

In[52]:=
```
Clear[α]; vizrodalbedoiso = Show[
  Plot[halfrodalbedoisoscatter`R[c], {c, 0, 1}],
  ListPlot[MCalbedos]
  , Frame → True, ImageSize → 500,
  FrameLabel → {{"R[α]",}, {"Single scattering albedo: α",
     "Total Reflectance/Albedo R(α): isotropically-scattering half rod"}}
 ]
```

Out[52]=



Compare Deterministic and MC

Internal distributions

In[42]:=
```
Clear[alpha, Σt];
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, maxx, dx, numcollorders, nummoments, pointsCL, plotpointsCL,
```

```
  pointsCR, plotpointsCR, plotpointsϕ, plotϕ, plotLL, plotLR},
 data = SelectFirst[halfrodalbedoisoscatter`simulations,
    #[[1]] == α && #[[2]] == Σt &][[3]];
 maxx = data[[2, 5]];
 dx = data[[2, 7]];
 numcollorders = data[[2, 11]];
 nummoments = data[[2, 13]];

 pointsCL = data[[7]];
 (* divide by Σt to convert collision density into L *)
 plotpointsCL = halfrodalbedoisoscatter`ppoints[pointsCL, dx, maxx, Σt];
 pointsCR = data[[9]];
 plotpointsCR = halfrodalbedoisoscatter`ppoints[pointsCR, dx, maxx, Σt];
 (* divide by Σt to convert collision density into fluence *)
 plotpointsϕ =
  halfrodalbedoisoscatter`ppoints[pointsCL + pointsCR, dx, maxx, Σt];

 plotϕ = Show[
   ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[halfrodalbedoisoscatter`ϕ[x, α, Σt], {x, 0, maxx}, PlotRange → All]
   , Frame → True,
   FrameLabel -> {{ϕ[x],},
     {"x", "Semi-infinite rod, albedo problem, isotropic scattering, fluence
         ϕ[x], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}
  ];

 plotLL = Show[
   ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[halfrodalbedoisoscatter`LL[x, α, Σt], {x, 0, maxx}, PlotRange → All],
   Plot[halfrodalbedoisoscatter`ϕ[x, α, Σt],
    {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
   , Frame → True,
   FrameLabel ->
    {{L-[x],}, {"x", "Semi-infinite rod, albedo problem, isotropic
         scattering, L-[x], α = " <>
       ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
  ];

 plotLR = Show[
   ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[halfrodalbedoisoscatter`LR[x, α, Σt], {x, 0, maxx}, PlotRange → All],
   Plot[halfrodalbedoisoscatter`ϕ[x, α, Σt],
    {x, 0, maxx}, PlotRange → All, PlotStyle → Dashed]
   , Frame → True,
   FrameLabel ->
    {{L+[x],}, {"x", "Semi-infinite rod, albedo problem, isotropic
         scattering, L+[x], α = " <>
       ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
  ];
 Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
]
,
Text[
 "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
   ensure the data path is setup correctly."]
]
, {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts}]
```

Σt 1 3

Out[43]=


Semi−infinite rod, albedo problem, isotropic scattering, fluence ϕ[x], α = 0.98, Σt = 1
ϕ(x)
x
0.0
0.5
1.0
1.5
0
2
4
6
8
10


Semi−infinite rod, albedo problem, isotropic scattering, L−[x], α = 0.98, Σt = 1
L−(x)
x
0.0
0.5
1.0
1.5
0
2
4
6
8
10


Semi−infinite rod, albedo problem, isotropic scattering, L+[x], α = 0.98, Σt = 1
L+(x)
x
0.0
0.5
1.0
1.5
0
2
4
6
8
10

n-th collided albedo

In[54]:=
```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, Rs, ns, analytic, j, numcollorders},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   numcollorders = data[[2, 11]];
   Rs = N[{data[[5]]}];
   ns = Table[n, {n, 0, numcollorders - 1}];
   analytic = Table[halfrodalbedoisoscatter`R[α, n], {n, ns}];
   j = Join[{ns}, {analytic}, Rs];
   TableForm[
    Join[{{"n", "analytic", "MC"}}, Transpose[j]]
   ]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts}]
```

Out[54]=

α 0.99

Σt 1 3

n	analytic	MC
0	0.	0.
1	0.2475	0.247108
2	0.122513	0.122198
3	0.0758046	0.076075
4	0.0525326	0.052501
5	0.0390055	0.039366
6	0.0303407	0.030539
7	0.0244053	0.024466
8	0.0201344	0.020101
9	0.0169431	0.016946
10	0.0144863	0.01445
11	0.0125488	0.012497
12	0.0109898	0.010953
13	0.00971422	0.009698
14	0.00865537	0.008703
15	0.00776549	0.007911
16	0.0070095	0.00685
17	0.00636112	0.006439
18	0.00580034	0.005724
19	0.00531166	0.005367

Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

In[45]:=
```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[11]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic = Table[halfrodalbedoisoscatter`ϕm[α, Σt, k], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts}]
```

Out[45]=

α 0.7

Σt 1 3

k	analytic	MC
0	2.35926	2.3596
1	4.3074	4.30982
2	15.7284	15.7611
3	86.1481	86.5815
4	629.137	635.576
5	5743.21	5853.98
6	62 913.7	65 139.9
7	804 049.	855 879.
8	1.17439×10^7	1.31088×10^7
9	1.92972×10^8	2.32381×10^8

n-th collided moments of ϕ

In[46]:=
```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, ϕmoments, ks, analytic, j, nummoments},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 13]];
   ϕmoments = N[{data[[13 + n]]}/Σt];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic =
    Table[Quiet[N[halfrodalbedoisoscatter`ϕm[α, Σt, k, n]]], {k, ks}];
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts},
 {n, Range[If[NumberQ[halfrodalbedoisoscatter`numcollorders],
    halfrodalbedoisoscatter`numcollorders, 1]]}]
```

Out[46]=

α 0.9 ∨

Σt 1 3

n 7 ∨

k	analytic	MC
0	0.187856	0.187809
1	Indeterminate	0.767718
2	4.47393	4.46419
3	Indeterminate	32.4047
4	278.797	277.914
5	Indeterminate	2729.36
6	30 169.4	30 079.3
7	Indeterminate	366 824.
8	4.88998×10^6	4.90097×10^6
9	Indeterminate	7.12065×10^7

N-th order Radiance/Angular flux

In[47]:=
```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, nthL, nthR, maxx, dx, numcollorders, LnR},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   numcollorders = data[[2, 11]];
   nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
   nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

   Clear[c, x];
   LnR = SeriesCoefficient[halfrodalbedoisoscatter`LR[x, c, Σt], {c, 0, n}] α^n;
   Show[
    ListPlot[halfrodalbedoisoscatter`ppoints[nthR[[n + 1]], dx, maxx, Σt],
     PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LnR, {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{L+["x|" <> ToString[n]],},
      {x, "Semi-Infinite Rod, albedo problem, isotropic scattering, L+[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <>
        ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts},
 {n, Range[If[NumberQ[halfrodalbedoisoscatter`numcollorders],
    halfrodalbedoisoscatter`numcollorders, 1]]}]
```

Out[47]=

α 0.95

Σt 1 3

n 4



N-th order Fluence / scalar flux

In[48]:=
```
Manipulate[
 If[Length[halfrodalbedoisoscatter`simulations] > 0,
  Module[{data, maxx, dx, numcollorders, nthL, nthR, ϕn},
   data = SelectFirst[
      halfrodalbedoisoscatter`simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
   maxx = data[[2, 5]];
   dx = data[[2, 7]];
   numcollorders = data[[2, 11]];
   nthL = data[[13 + numcollorders + 1 ;; 13 + 2 numcollorders]];
   nthR = data[[13 + 2 numcollorders + 2 ;; -1]];

   Clear[c];
   ϕn = SeriesCoefficient[halfrodalbedoisoscatter`ϕ[x, c, Σt], {c, 0, n}] α^n;
   Show[
    ListPlot[halfrodalbedoisoscatter`ppoints[nthR[[n + 1]] + nthL[[n + 1]],
      dx, maxx, Σt], PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[ϕn, {x, 0, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{L+["x|" <> ToString[n]],},
      {x, "Semi-Infinite Rod, albedo problem, isotropic scattering, ϕ[x|" <>
        ToString[n] <> "], α = " <> ToString[α] <>
        ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, halfrodalbedoisoscatter`alphas}, {Σt, halfrodalbedoisoscatter`muts},
 {n, Range[If[NumberQ[halfrodalbedoisoscatter`numcollorders],
    halfrodalbedoisoscatter`numcollorders, 1]]}]
```

Out[48]=

α 0.999

Σt 1 3

n 13