Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Gitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Isotropic Scattering

```
pIsotropic[u_] := 1/(4 Pi)
```

Normalization condition

```
Integrate[2 Pi pIsotropic[u], {u, -1, 1}]
```

1

Mean-cosine

```
Integrate[2 Pi pIsotropic[u] u, {u, -1, 1}]
```

0

Legendre expansion coefficients

```
Integrate[
 2 Pi (2 k + 1) pIsotropic[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 0, {y, 0, Pi}]
```

1

```
Integrate[
 2 Pi (2 k + 1) pIsotropic[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 1, {y, 0, Pi}]
```

0

sampling

```
cdf = Integrate[2 Pi pIsotropic[u], {u, -1, x}]
```

$$\frac{1+x}{2}$$

```
Solve[cdf == e, x]
```

$\{\{x \to -1 + 2 e\}\}$

```
Clear[u]; Show[
 Plot[pIsotropic[u], {u, -1, 1}, PlotStyle → Thick]
 , Frame → True,
 FrameLabel → {{p[u],}, {"u = Cos[θ]", "Isotropic Scattering"}}]
```



Linearly-Anisotropic Scattering (Eddington)

```
pLinaniso[u_, b_] := 1/(4 Pi) (1 + b u)
```

```
Clear[u];
Show[
 Plot[pIsotropic[u], {u, -1, 1}, PlotStyle → Thick],
 Plot[pLinaniso[u, 1], {u, -1, 1}, PlotStyle → Dashed]
 , Frame → True,
 FrameLabel → {{p[u],}, {"u = Cos[θ]", "Linearly-Anisotropic Scattering"}}]
```



Normalization condition

```
Integrate[2 Pi pLinaniso[u, b], {u, -1, 1}, Assumptions → b > -1 && b < 1]
```

1

Mean cosine (g)

```
Integrate[2 Pi pLinaniso[u, b] u, {u, -1, 1}, Assumptions → b > -1 && b < 1]
```

$$\frac{b}{3}$$

Legendre expansion coefficients

```
Integrate[
 2 Pi (2 k + 1) pLinaniso[Cos[y], b] LegendreP[k, Cos[y]] Sin[y] /. k → 0, {y, 0, Pi}]
```

1

```
Integrate[
 2 Pi (2 k + 1) pLinaniso[Cos[y], b] LegendreP[k, Cos[y]] Sin[y] /. k → 1, {y, 0, Pi}]
```

b

sampling

```
cdf = Integrate[2 Pi pLinaniso[u, b], {u, -1, x}]
```

$$\frac{1}{2} - \frac{b}{4} + \frac{x}{2} + \frac{b\,x^2}{4}$$

```
Solve[cdf == e, x]
```

$$\left\{\left\{x \to \frac{-1 - \sqrt{1 - 2\,b + b^2 + 4\,b\,e}}{b}\right\}, \left\{x \to \frac{-1 + \sqrt{1 - 2\,b + b^2 + 4\,b\,e}}{b}\right\}\right\}$$

```
b = 0.7;
Show[
 Plot[2 Pi pLinaniso[u, b], {u, -1, 1}],
 Histogram[
  Map[(-1 + Sqrt[1 - 2 b + b^2 + 4 b #])/b &, Table[RandomReal[], {i, 1, 100000}]], 50, "PDF"]
]
Clear[b];
```



Rayleigh Scattering

General form:

In[8]:=
```
pRayleigh[u_, γ_] := 1/(4 Pi) 3/(4 (1 + 2 γ)) ((1 + 3 γ) + (1 - γ) u^2)
```

$$\text{pRayleigh}[u\_, \gamma\_] := \frac{1}{4\,\text{Pi}}\,\frac{3}{4\,(1+2\gamma)}\left((1+3\gamma)+(1-\gamma)\,u^2\right)$$

Common special case ($\gamma = 0$):

In[10]:=
$$\text{pRayleigh}[u\_] := \left(1+u^2\right)\frac{3}{16\,\text{Pi}}$$

Normalization condition

In[11]:=
```
Integrate[2 Pi pRayleigh[u], {u, -1, 1}]
```

Out[11]= 1

In[17]:=
```
Integrate[2 Pi pRayleigh[u, γ], {u, -1, 1}, Assumptions → γ > 0] // Simplify
```

Out[17]= 1

Mean cosine (g)

In[12]:=
```
Integrate[2 Pi pRayleigh[u] u, {u, -1, 1}]
```

Out[12]= 0

In[18]:=
```
Integrate[2 Pi pRayleigh[u, γ] u, {u, -1, 1}, Assumptions → γ > 0] // Simplify
```

Out[18]= 0

Legendre expansion coefficients

```
Integrate[
 2 Pi (2 k + 1) pRayleigh[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 0, {y, 0, Pi}]
```

1

```
Integrate[
 2 Pi (2 k + 1) pRayleigh[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 1, {y, 0, Pi}]
```

0

```
Integrate[
 2 Pi (2 k + 1) pRayleigh[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 2, {y, 0, Pi}]
```

$\frac{1}{2}$

sampling

```
Show[
 Plot[2 Pi pRayleigh[u], {u, -1, 1}],
 Histogram[Map[(1 - (2 - 4 # + Sqrt[5 + 16 (-1 + #) #])^(2/3)) / (2 - 4 # + Sqrt[5 + 16 (-1 + #) #])^(1/3) &,
   Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
]
Clear[b];
```



Henyey-greenstein Scattering

```
Clear[pHG]; pHG[dot_, g_] := 1/(4 Pi) (1 - g^2)/(1 + g^2 - 2 g dot)^(3/2)
```

```
pHGplot = Show[
  Plot[pHG[Cos[t], .8], {t, -Pi, Pi}, PlotRange → {0, 1}],
  Plot[pHG[Cos[t], .6], {t, -Pi, Pi}, PlotRange → All],
  Plot[pHG[Cos[t], .5], {t, -Pi, Pi}, PlotRange → All],
  Plot[pHG[Cos[t], .4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pHG[Cos[t], .3], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel → {{p[Cos[θ]],},
    {θ, "Henyey-Greenstein Scattering, g = 0.3, 0.4, 0.5, 0.6, 0.8"}}]
```



```
Show[
 ParametricPlot[{Sin[t], Cos[t]} (1),
  {t, -Pi, Pi}, PlotRange → All, PlotStyle → Dashed],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.75]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.68]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.6]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.5]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pHG[Cos[t], 0.3]),
  {t, -Pi, Pi}, PlotRange → All]
]
```



Normalization condition

```
Integrate[2 Pi pHG[u, g], {u, -1, 1}, Assumptions → g > -1 && g < 1]
```

```
1
```

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pHG[u, g] LegendreP[k, u] /. k → 0,
 {u, -1, 1}, Assumptions → g > -1 && g < 1]
```

```
1
```

```
Integrate[2 Pi (2 k + 1) pHG[u, g] LegendreP[k, u] /. k → 1,
 {u, -1, 1}, Assumptions → g > -1 && g < 1]
```

3 g

sampling

```
cdf = Integrate[2 Pi pHG[u, g], {u, -1, x}, Assumptions → g > -1 && g < 1 && x < 1]
```

$$\frac{(-1+g)\left(-1-g+\sqrt{1+g^2-2\,g\,x}\right)}{2\,g\,\sqrt{1+g^2-2\,g\,x}}$$

```
Solve[cdf == e, x]
```

$$\left\{\left\{x \to \frac{-1+2\,e+2\,g-2\,e\,g+2\,e^2\,g-g^2+2\,e\,g^2-2\,e\,g^3+2\,e^2\,g^3}{(1-g+2\,e\,g)^2}\right\}\right\}$$

```
FullSimplify[%]
```

$$\left\{\left\{x \to -\frac{(-1+g)^2+2\,e\,(-1+g)\left(1+g^2\right)-2\,e^2\left(g+g^3\right)}{(1+(-1+2\,e)\,g)^2}\right\}\right\}$$

```
g = 0.7;
Show[
 Plot[2 Pi pHG[u, g], {u, -1, 1}],
 Histogram[Map[-((-1 + g)^2 + 2 # (-1 + g) (1 + g^2) - 2 #^2 (g + g^3))/(1 + (-1 + 2 #) g)^2 &,
   Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
]
Clear[b, g];
```



Henyey-greenstein Scattering (Flatland)

Definition:

```
pH2[θ_, g_] := 1/(2 Pi) (1 - g^2)/(1 + g^2 - 2 g Cos[θ]);
```

Moments

```
Integrate[pH2[t, g] Cos[t], {t, -Pi, Pi}, Assumptions → g > -1 && g < 1 && g ≠ 0 && n ≥ 0]
```

g

```
Integrate[pH2[t, g] Cos[2 t], {t, -Pi, Pi},
 Assumptions → g > -1 && g < 1 && g ≠ 0 && n ≥ 0]
```

g^2

```
Integrate[pH2[t, g] Cos[7 t], {t, -Pi, Pi},
 Assumptions → g > -1 && g < 1 && g ≠ 0 && n ≥ 0]
```

g^7

Sampling:

```
g = -0.7;
Show[
 Histogram[Map[2 ArcTan[(1 - g)/(1 + g) Tan[Pi/2 (1 - 2 #)]] &,
   Table[RandomReal[], {i, 1, 100000}]], 50, "PDF"],
 Plot[ pH2[θ, g], {θ, -Pi, Pi}, PlotRange → All]
]
Clear[g];
```



Kagiwada-Kalaba (Ellipsoidal) Scattering

```
pEllipsoidal[u_, b_] := b (2 Pi Log[(1 + b) / (1 - b)] (1 - b u))^-1
```

```
pEllplot = Show[
  Plot[pEllipsoidal[Cos[t], .9], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .8], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .65], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pEllipsoidal[Cos[t], .95], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel → {{p[Cos[θ]],},
    {θ, "Ellipsoidal Scattering, b = 0.4, 0.65, 0.8, 0.9, 0.95"}}]
```



sampling

```
b = -0.8;
Show[Histogram[
  Map[(1 - (1 + b) ((1+b)/(1-b))^-#)/b &, Table[RandomReal[], {i, 1, 100000}]], 50, "PDF"],
 Plot[2 Pi pEllipsoidal[u, b], {u, -1, 1}]

]
Clear[b];
```



Binomial Scattering

```
pBinomial[u_, n_] := Pi^-1 ((n + 1) / 2^(n+2)) (1 + u)^n
```

```
pBinplot = Show[
  Plot[pBinomial[Cos[t], 1], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 2], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 3], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pBinomial[Cos[t], 5], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel → {{p[Cos[θ]],}, {θ, "Binomial Scattering, n = 1, 2, 3, 4, 5"}}]
```



Normalization condition

```
Integrate[2 Pi pBinomial[u, n], {u, -1, 1}, Assumptions → n ≥ 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi pBinomial[u, n] u, {u, -1, 1}, Assumptions → n ≥ 0]
```

$$\frac{n}{2 + n}$$

sampling

```
n = 25.8;
Show[
 Histogram[Map[-1 + (2^(1+n) #)^(1/(1+n)) &, Table[RandomReal[], {i, 1, 100000}]], 50, "PDF"],
 Plot[2 Pi pBinomial[u, n], {u, -1, 1}, PlotRange → All]

]
Clear[b];
```



Liu Scattering

```
pLiu[u_, e_, m_] := (e (2 m + 1) (1 + e u)^(2 m)) / (2 Pi ((1 + e)^(2 m+1) - (1 - e)^(2 m+1)))

Clear[m]

pLiuplot = Show[
   Plot[pLiu[Cos[t], 4, 2], {t, -Pi, Pi}, PlotRange → All],
   Plot[pLiu[Cos[t], 7, 2], {t, -Pi, Pi}, PlotRange → All],
   Frame → True,
   ImageSize → 400,
   FrameLabel →
    {{p[Cos[θ]],}, {θ, "Liu Scattering, (m = 2, ϵ = 4), (m = 2, ϵ = 7)"}}]
```



Normalization condition

```
Integrate[2 Pi pLiu[u, e, m], {u, -1, 1}, Assumptions → e > 0 && m > 0 && m ∈ Integers]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pLiu[u, e, m], {u, -1, 1},
 Assumptions → e > 0 && m > 0 && m ∈ Integers && e < 1]
```

$$\frac{(1+e)^{1+2m}(-1+e+2em)+(1-e)^{1+2m}(1+e+2em)}{2e\left(-(1-e)^{1+2m}+(1+e)^{1+2m}\right)(1+m)}$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pLiu[u, e, m] LegendreP[k, u] /. k → 0, {u, -1, 1},
 Assumptions → m > 0 && m ∈ Integers && e ∈ Reals && e ≠ 0 && Abs[e] < 1]
```

1

```
Integrate[2 Pi (2 k + 1) pLiu[u, e, m] LegendreP[k, u] /. k → 2, {u, -1, 1},
 Assumptions → m > 0 && m ∈ Integers && e ∈ Reals && e ≠ 0 && Abs[e] < 1]
```

$$\left(5\left((1+e)^{1+2m}(3+e(-3+2m(-3+2e(1+m))))+(1-e)^{2m}(-1+e)(3+e(3+2m(3+2e(1+m))))\right)\right)/\left(2e^2\left(-(1-e)^{1+2m}+(1+e)^{1+2m}\right)(1+m)(3+2m)\right)$$

sampling

```
m = 3.5;
ϵ = 0.9;
```

$$\text{Show}\left[\text{Histogram}\left[\text{Map}\left[\frac{-1+\left((-1+\#)(1-\epsilon)^{2m}(-1+\epsilon)+\#(1+\epsilon)^{1+2m}\right)^{\frac{1}{1+2m}}}{\epsilon}\ \&,\right.\right.\right.$$

```
    Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
 Plot[2 Pi pLiu[u, ϵ, m], {u, -1, 1}, PlotRange → All]

]
Clear[m, ϵ];
```



Gegenbauer Scattering

```
pGegenbauer[u_, g_, a_] := (1 + g^2 - 2 g u)^-(a+1) / (((1-g)^-2a - (1+g)^-2a) π / (a g))
```

$$\text{pGegenbauer}[u\_, g\_, a\_] := \frac{\left(1 + g^2 - 2\,g\,u\right)^{-(a+1)}}{\frac{\left((1-g)^{-2a} - (1+g)^{-2a}\right)\pi}{a\,g}}$$

```
Show[
 Plot[pGegenbauer[Cos[t], 0.5, 1], {t, -Pi, Pi}, PlotRange → All],
 Plot[pGegenbauer[Cos[t], 0.5, 3], {t, -Pi, Pi}, PlotRange → All],
 Plot[pGegenbauer[Cos[t], 0.5, 5], {t, -Pi, Pi}, PlotRange → All],

 Frame → True,
 FrameLabel →
  {{p[Cos[θ]],}, {θ, "Gegenbauer Scattering, G = 0.5, a = 1, 3, 5"}}]
```



Normalization condition

```
Integrate[2 Pi pGegenbauer[u, g, a], {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pGegenbauer[u, g, a], {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

$$\frac{(1+g)^{2a}\left(1-2\,a\,g+g^2\right)-(1-g)^{2a}\left(1+2\,a\,g+g^2\right)}{2\,(-1+a)\,g\left((1-g)^{2a}-(1+g)^{2a}\right)}$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pGegenbauer[u, g, a] LegendreP[k, u] /. k → 0,
 {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

1

```
FullSimplify[Integrate[2 Pi (2 k + 1) pGegenbauer[u, g, a] LegendreP[k, u] /. k → 3,
  {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]]
```

$$-\left(7\left(24\,a^2\,g^2\left(1+g^2\right)\left((1-g)^{2a}-(1+g)^{2a}\right)+3\left(5+3\,g^2+3\,g^4+5\,g^6\right)\left((1-g)^{2a}-(1+g)^{2a}\right)+8\,a^3\,g^3\left((1-g)^{2a}+(1+g)^{2a}\right)+2\,a\,g\left(15+14\,g^2+15\,g^4\right)\left((1-g)^{2a}+(1+g)^{2a}\right)\right)\right)\Big/\left(8\,(-3+a)\,(-2+a)\,(-1+a)\,g^3\left((1-g)^{2a}-(1+g)^{2a}\right)\right)$$

sampling

```
g = -0.8;
a = -1.2;
```

$$\texttt{Show}\left[\texttt{Histogram}\left[\texttt{Map}\left[\frac{1+g^2-\left(\#\,(1-g)^{-2a}-(-1+\#)\,(1+g)^{-2a}\right)^{-1/a}}{2\,g}\ \&,\right.\right.\right.$$

```
   Table[RandomReal[], {i, 1, 100 000}]], 100, "PDF"],
 Plot[2 Pi pGegenbauer[u, g, a], {u, -1, 1}, PlotRange → All]

]
Clear[g, a];
```



vMF (spherical Gaussian) Scattering

$$\texttt{pVMF[u\_, k\_] :=}\ \frac{k}{4\,\texttt{Pi}\ \texttt{Sinh[k]}}\ \texttt{Exp[k u]}$$

```
Show[
 Plot[pVMF[Cos[t], 5.8], {t, -Pi, Pi}, PlotRange → All],
 Plot[pVMF[Cos[t], 15], {t, -Pi, Pi}, PlotRange → All],
 Plot[pVMF[Cos[t], 30], {t, -Pi, Pi}, PlotRange → All],

 Frame → True,
 FrameLabel → {{p[Cos[θ]],}, {θ, "vMF, k = {5.8,15,30}"}}]
```



Normalization condition

```
Integrate[2 Pi pVMF[u, k], {u, -1, 1}, Assumptions → k > 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pVMF[u, k], {u, -1, 1}, Assumptions → k > 0]
```

$$-\frac{1}{k} + \mathrm{Coth}[k]$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 4,
 {u, -1, 1}, Assumptions → k > 0]
```

$$\frac{9 \left(105 + 45 k^2 + k^4 - 5 k \left(21 + 2 k^2\right) \mathrm{Coth}[k]\right)}{k^4}$$

sampling

```
k = 3;
Show[Histogram[
  Map[Log[E^-k (1 - #) + E^k #] / k &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
 Plot[2 Pi pVMF[u, k], {u, -1, 1}, PlotRange → All]

]
Clear[k];
```