Infinite 3D medium, Isotropic Point Source, Linearly-Anisotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
In[424]:= SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



α - single-scattering albedo
Σt - extinction coefficient
r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine
b - anisotropy parameter

Util

In[425]:= $$\texttt{SurfaceArea[d\_, r\_] := d}\,\frac{\texttt{Pi}^{d/2}}{\texttt{Gamma}\left[\frac{d}{2}+1\right]}\,\texttt{r}^{d-1}$$

Diffusion modes

In[426]:= $$\texttt{diffusionMode[v\_, d\_, r\_] :=}\ (2\,\pi)^{-d/2}\,r^{1-\frac{d}{2}}\,v^{-1-\frac{d}{2}}\,\texttt{BesselK}\left[\frac{1}{2}\,(-2+d),\ \frac{r}{v}\right]$$

Fluence: exact solution

In[779]:= $$\texttt{Alinearaniso[c0\_, g\_, v\_] :=}$$
$$\left(v\,\left(1-v^2\right)\right) \Big/ \left(c0\,\left(v^2-1+c0+3\,g\,(1-c0)\,\left(3-c0-3\,(1-c0)\,(1-g\,c0)\,/\,v^2\right)\right)\right);$$
$$\texttt{glinearaniso[c0\_, g\_, u\_] :=}\ 1\Big/\left(\left(\frac{\texttt{Pi}\,c0\,u}{2}\,\left(1+3\,g\,(1-c0)\,u^2\right)\right)^2+\left(1+3\,g\,c0\,(1-c0)\,u^2-\left(1+3\,g\,(1-c0)\,u^2\right)\,\frac{c0}{2}\,u\,\texttt{Log}\left[\frac{1+u}{1-u}\right]\right)^2\right);$$
$$\texttt{v0linearaniso[c\_, g\_] := ReplaceAll}\Big[\texttt{Abs[v]},$$
$$\texttt{FindRoot}\left[1+\frac{3\,g\,c\,(1-c)}{v^2}-\left(1+\frac{3\,g\,(1-c)}{v^2}\right)\,\frac{c}{2\,v}\,\texttt{Log}\left[\frac{1+v}{1-v}\right],\ \{v,\,1.1\}\right]\Big];$$

In[881]:= $$\texttt{inf3Disopointlinanisoscatter`}\phi\texttt{exact[r\_, }\Sigma\texttt{t\_, }\alpha\texttt{\_, b\_] :=}$$
$$\frac{\#\,\Sigma t}{2\,\texttt{Pi}\,r}\,\texttt{Alinearaniso}[\alpha,\,b/3,\,\#]\,\texttt{Exp}[-\#\,r\,\Sigma t]+\frac{\Sigma t}{4\,\texttt{Pi}\,r}\,\texttt{NIntegrate}\Big[$$
$$\frac{1}{u^2}\,\texttt{glinearaniso}[\alpha,\,b/3,\,u]\,\texttt{Exp}\left[-\Sigma t\,\frac{r}{u}\right],\ \{u,\,0,\,1\}\Big]\ \&[\texttt{v0linearaniso}[\alpha,\,b/3]]$$

Fluence: Classical Diffusion Approximation

In[910]:= $$\texttt{inf3Disopointlinanisoscatter`}\phi\texttt{Diffusion[r\_, }\Sigma\texttt{t\_, }\alpha\texttt{\_, b\_] :=}$$
$$\frac{e^{-r\,\sqrt{(1-\alpha)\,(3-b\,\alpha)}\,\Sigma t}\,(3-b\,\alpha)\,\Sigma t}{4\,\pi\,r}$$

Fluence: Grosjean Modified Diffusion Approximation

In[934]:= $$\texttt{inf3Disopointlinanisoscatter`}\phi\texttt{Grosjean[r\_, }\Sigma\texttt{t\_, }\alpha\texttt{\_, b\_] :=}$$
$$\frac{E^{-r\,\Sigma t}}{4\,\texttt{Pi}\,r^2}+\frac{\alpha}{1-\alpha}\,\frac{1}{\Sigma t}\,\texttt{diffusionMode}\left[\frac{1}{\sqrt{3}\,\sqrt{\frac{(-1+\alpha)\,(-3+b\,\alpha)}{6+b\,(-1+\alpha)^2-3\,\alpha}}\,\Sigma t},\ 3,\ r\right]$$

In[933]:= $$\texttt{FullSimplify[inf3Disopointlinanisoscatter`}\phi\texttt{Grosjean[r, }\Sigma\texttt{t, }\alpha\texttt{, b],}$$
$$\texttt{Assumptions}\rightarrow\Sigma t>0\ \&\&\ \alpha>0\ \&\&\ \alpha<1\ \&\&\ b>-1\ \&\&\ b<1]$$

Out[933]= $$\frac{e^{-r\,\Sigma t}-\frac{3\,e^{-\sqrt{3}\,r\,\sqrt{\frac{(-1+\alpha)\,(-3+b\,\alpha)}{6+b\,(-1+\alpha)^2-3\,\alpha}}\,\Sigma t}\,r\,\alpha\,(-3+b\,\alpha)\,\Sigma t}{6+b\,(-1+\alpha)^2-3\,\alpha}}{4\,\pi\,r^2}$$

Nth-collided fluence - Gaussian approximation

In[961]:=
```
inf3Disopointlinanisoscatter`twomomentGaussian[r_, m0_, m2_] :=
```
$$\frac{3\sqrt{\frac{3}{2}}\ e^{-\frac{3\,m0\,r^2}{2\,m2}}\ m0^{5/2}}{2\ m2^{3/2}\ \pi^{3/2}}$$

In[963]:=
```
inf3Disopointlinanisoscatter`ϕGaussian[r_, Σt_, α_, b_, n_] :=
 inf3Disopointlinanisoscatter`twomomentGaussian[
  r, α^n/Σt, (2 × 3^-n (b^(2+n) + 3^(2+n) (1 + n) - 3^(1+n) b (2 + n)) α^n)/((-3 + b)^2 Σt^3)]
```
$$r,\ \frac{\alpha^n}{\Sigma t},\ \frac{2\times 3^{-n}\left(b^{2+n}+3^{2+n}\,(1+n)-3^{1+n}\,b\,(2+n)\right)\alpha^n}{(-3+b)^2\,\Sigma t^3}$$

load MC data

In[852]:=
```
inf3Disopointlinanisoscatter`ppoints[xs_, dr_, maxx_, Σt_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]] / Σt}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[853]:=
```
inf3Disopointlinanisoscatter`ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[854]:=
```
inf3Disopointlinanisoscatter`fs =
  FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/data/
     inf3D_isotropicpoint_linanisoscatter*"];
```

In[855]:=
```
inf3Disopointlinanisoscatter`index[x_] := Module[{data, α, Σt, b},
   data = Import[x, "Table"];
   Σt = data[[1, 13]];
   α = data[[2, 3]];
   b = data[[1, 16]];
   {α, Σt, b, data}];
inf3Disopointlinanisoscatter`simulations =
  inf3Disopointlinanisoscatter`index /@ inf3Disopointlinanisoscatter`fs;
inf3Disopointlinanisoscatter`alphas =
 Union[#[[1]] & /@ inf3Disopointlinanisoscatter`simulations]
```

Out[857]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[858]:=
```
inf3Disopointlinanisoscatter`muts =
 Union[#[[2]] & /@ inf3Disopointlinanisoscatter`simulations]
```

Out[858]=
```
{1, 3}
```

In[859]:=
```
inf3Disopointlinanisoscatter`bs =
 Union[#[[3]] & /@ inf3Disopointlinanisoscatter`simulations]
```

Out[859]=
```
{-0.9, 0.7}
```

In[860]:=
```
inf3Disopointlinanisoscatter`numcollorders =
  inf3Disopointlinanisoscatter`simulations[[1]][[-1]][[2, 13]];
inf3Disopointlinanisoscatter`maxr =
 inf3Disopointlinanisoscatter`simulations[[1]][[-1]][[2, 5]];
inf3Disopointlinanisoscatter`dr =
 inf3Disopointlinanisoscatter`simulations[[1]][[-1]][[2, 7]];
inf3Disopointlinanisoscatter`numr =
  Floor[inf3Disopointlinanisoscatter`maxr / inf3Disopointlinanisoscatter`dr];
```

Compare Deterministic and MC

Fluence - Exact solution comparison to MC

In[882]:=
```
Manipulate[
 If[Length[inf3Disopointlinanisoscatter`simulations] > 0,
  Module[{data, maxr, dr, pointsϕ, plotpointsϕ, logplotϕ, plotϕ, exact1points},
   data = SelectFirst[inf3Disopointlinanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == b &][[-1]];
   maxr = data[[2, 5]];
   dr = data[[2, 7]];

   pointsϕ = data[[4]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disopointlinanisoscatter`ppoints[pointsϕ, dr, maxr, Σt];

   exact1points =
    Quiet[{#[[1]], 4 Pi #[[1]]^2 inf3Disopointlinanisoscatter`ϕexact[#[[1]],
          Σt, α, b]}] & /@ plotpointsϕ;

   plotϕ = Quiet[Show[
      ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      ListPlot[exact1points, PlotRange → All, Joined → True],
      Frame → True,
      FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      ListLogPlot[exact1points, PlotRange → All, Joined → True],
      Frame → True,
      FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
     ]];
   pp = Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800],
     PlotLabel -> "Exact solution (1a)\nInfinite 3D, isotropic point source,
         Linearly anisotropic scattering, fluence ϕ[r], α = " <>
       ToString[α] <> ", Σt = " <> ToString[Σt] <> ", b = " <> ToString[b]]
  ]
  ,
  Text["Uh oh!  Couldn't find MC data.
     Try to evaluate this entire notebook and ensure the data path is setup
      correctly."]
 ]
 , {{α, 0.8}, inf3Disopointlinanisoscatter`alphas},
 {{Σt, 1}, inf3Disopointlinanisoscatter`muts},
 {b, inf3Disopointlinanisoscatter`bs}]
```

Out[882]=

α 0.95

Σt 1 3

b −0.9 0.7



Fluence - Diffusion Approximations (Classical and Grosjean) Comparison to MC

In[935]:=
```
Clear[alpha, Σt];
Manipulate[
 If[Length[inf3Disopointlinanisoscatter`simulations] > 0,
  Module[{data, maxr, dr, pointsϕ, plotpointsϕ, logplotϕ, plotϕ, exact1points},
   data = SelectFirst[inf3Disopointlinanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == b &][[-1]];
   maxr = data[[2, 5]];
   dr = data[[2, 7]];

   pointsϕ = data[[4]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = inf3Disopointlinanisoscatter`ppoints[pointsϕ, dr, maxr, Σt];

   plotϕ = Quiet[Show[
      ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      Plot[4 Pi r^2 inf3Disopointlinanisoscatter`ϕGrosjean[r, Σt, α, b],
       {r, 0, maxr}, PlotRange → All],
      Plot[4 Pi r^2 inf3Disopointlinanisoscatter`ϕDiffusion[r, Σt, α, b],
       {r, 0, maxr}, PlotRange → All, PlotStyle → Dashed],
      Frame → True,
      FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
      LogPlot[4 Pi r^2 inf3Disopointlinanisoscatter`ϕGrosjean[r, Σt, α, b],
       {r, 0, maxr}, PlotRange → All],
      LogPlot[4 Pi r^2 inf3Disopointlinanisoscatter`ϕDiffusion[r, Σt, α, b],
       {r, 0, maxr}, PlotRange → All, PlotStyle → Dashed],
      Frame → True,
      FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
     ]];
   pp = Show[GraphicsGrid[{{plotϕ, logplotϕ}}], ImageSize → 800],
     PlotLabel -> "Classical (dashed) and Grosjean Modified (thin) Diffusion
         Approximation\nInfinite 3D medium, isotropic point source,
         linearly anisotropic scattering, fluence ϕ[r], α = " <>
       ToString[α] <> ", Σt = " <> ToString[Σt] <> ", b = " <> ToString[b]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.8}, inf3Disopointlinanisoscatter`alphas},
 {{Σt, 1}, inf3Disopointlinanisoscatter`muts},
 {b, inf3Disopointlinanisoscatter`bs}]
```

Out[936]=



N-th collided Fluence - Approximations

In[965]:=
```
Manipulate[
  If[Length[inf3Disopointlinanisoscatter`simulations] > 0,
   Module[{data, maxr, dr, pointsϕ,
     plotpointsϕ, logplotϕ, plotϕ, exact1points, numorders},
    data = SelectFirst[inf3Disopointlinanisoscatter`simulations,
       #[[1]] == α && #[[2]] == Σt && #[[3]] == b &][[-1]];
    maxr = data[[2, 5]];
    dr = data[[2, 7]];
    numorders = data[[2, 13]];

    pointsϕ = data[[9 + numorders + n + 1]];

    (* divide by Σt to convert collision density into fluence *)
    plotpointsϕ = inf3Disopointlinanisoscatter`ppoints[pointsϕ, dr, maxr, Σt];

    seriesclassical = α^n SeriesCoefficient[
       inf3Disopointlinanisoscatter`ϕDiffusion[r, Σt, C, b], {C, 0, n}];
    seriesG = α^n SeriesCoefficient[inf3Disopointlinanisoscatter`ϕGrosjean[
        r, Σt, C, b], {C, 0, n}];

    plotϕ = Quiet[Show[
       ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
       Plot[4 Pi r^2 inf3Disopointlinanisoscatter`ϕGaussian[r, Σt, α, b, n],
        {r, 0, maxr}, PlotRange → All],
       Plot[4 Pi r^2 seriesclassical, {r, 0, maxr}, PlotRange → All,
        PlotStyle → {Dashed, Blue}],
       Plot[4 Pi r^2 seriesG, {r, 0, maxr}, PlotRange → All,
```

```
      PlotStyle → {DotDashed, Gray}],
     Frame → True,
     FrameLabel -> {{4 Pi r^2 ϕ[r | n],}, {r,}}
    ]];
   logplotϕ = Quiet[Show[
     ListLogPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
     LogPlot[4 Pi r^2
       inf3Disopointlinanisoscatter`ϕGaussian[r, Σt, α, b, n], {r, 0, maxr}],
     LogPlot[4 Pi r^2 seriesclassical, {r, 0, maxr}, PlotRange → All,
      PlotStyle → {Dashed, Blue}],
     LogPlot[4 Pi r^2 seriesG, {r, 0, maxr}, PlotRange → All,
      PlotStyle → {DotDashed, Gray}],
     Frame → True,
     FrameLabel -> {{4 Pi r^2 ϕ[r | n],}, {r,}}
    ]];
   pp = Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 1000],
     PlotLabel -> "Approximate solutions, Gaussian (thin) Classical
         Diffusion (Dashed) Grosjean (Dot-Dashed)\nInfinite 3D
         medium, isotropic point source, linearly-anisotropic
         scattering, n-th scattered fluence ϕ[r|n], α = " <>
      ToString[α] <> ", Σt = " <> ToString[Σt] <> ", b = " <> ToString[b]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.8}, inf3Disopointlinanisoscatter`alphas},
 {{Σt, 1}, inf3Disopointlinanisoscatter`muts},
 {b, inf3Disopointlinanisoscatter`bs},
 {{n, 3}, Range[If[NumberQ[inf3Disopointlinanisoscatter`numcollorders],
    inf3Disopointlinanisoscatter`numcollorders, 1]]}]
```

Out[965]=


α 0.8
Σt 1 3
b −0.9 0.7
n 3
Approximate solutions, Gaussian (thin) Classical Diffusion (Dashed)
Infinite 3D medium, isotropic point source, linearly−anisotropic scattering, n−th scatte
0.15
0.10
0.05
0.00
0 5 10 15 20 25
r
4π r² ϕ(r 3)
0.100
0.010
0.001
10⁻⁴
10⁻⁵
10⁻⁶
0

Compare moments of ϕ

In[894]:=
```
Manipulate[
 If[Length[inf3Disopointlinanisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[inf3Disopointlinanisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt && #[[3]] == b &][[-1]];
   nummoments = data[[2, 15]];
   ϕmoments = N[{data[[6]]/Σt}];
   ks = Table[k, {k, 0, nummoments - 1}];
   analytic = {1/(1 - α) 1/Σt, 0, -6/((α - 1)^2 (α b - 3)) 1/Σt^3, 0,
     1/Σt^5 (24 (4 α - 9))/((α - 1)^3 (α b - 3)^2), 0, 1/Σt^7 144 (4 α (-55 α + 9 (α - 1)^2 b + 180) - 675)/(5 (α - 1)^4 (α b - 3)^3), 0};
   j = Join[{ks}, {analytic}, ϕmoments];

   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text["Uh oh!  Couldn't find MC data.
     Try to evaluate this entire notebook and ensure the data path is setup
      correctly."]
 ]
 , {{α, 0.8}, inf3Disopointlinanisoscatter`alphas},
 {{Σt, 1}, inf3Disopointlinanisoscatter`muts},
 {b, inf3Disopointlinanisoscatter`bs}]
```

Out[894]=

α 0.1

Σt 1 3

b −0.9 0.7

k	analytic	MC
0	0.37037	0.3704
1	0	0.129551
2	0.0887859	0.0887527
3	0	0.090292
4	0.122028	0.121767
5	0	0.204466
6	0.413501	0.410637
7	0	0.958387