Half Space, Albedo Problem, Isotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[105]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation

α - single-scattering albedo
Σt - extinction coefficient
z - depth in medium (positive inside the scattering half space)
u = cos θ - direction cosine

Caseology

Caseology quantities

In[106]:=
$$\text{CaseN0[c\_, v0\_]} := \frac{1}{2}\, c\, v0^3 \left(\frac{c}{v0^2 - 1} - \frac{1}{v0^2}\right)$$

In[107]:=
$$\text{Casev0[c\_?NumericQ]} := \text{If}\Big[c < 0.1,$$
$$1\Big/\left(1 - 2\, e^{-2/c}\left(1 + \frac{(512 - 384\, c + 72\, c^2 - 3\, c^3)\, e^{-6/c}}{3\, c^3} + \frac{(24 - 12\, c + c^2)\, e^{-4/c}}{c^2} + \frac{(4 - c)\, e^{-2/c}}{c}\right)\right),$$
$$\text{FindRoot}\Big[c\, v\, \text{ArcTanh}\Big[\frac{1}{v}\Big] - 1 == 0,\ \{v,\ 1 + 10^{-14},\ 10^{14}\},\ \text{Method} \to \text{"Brent"}\Big][[1]][[2]]$$
$$\Big]$$

In[108]:=
$$\text{CaseN[c\_, v\_]} := v\left(\text{Case}\lambda[v, c]^2 + \left(\frac{\pi\, c\, v}{2}\right)^2\right)$$

In[109]:=
```
Caseλ[v_, c_] := 1 - c v ArcTanh[v]
```

In[110]:=
$$\texttt{Case}\psi\texttt{0[u\_, v0\_, c\_, z\_] :=} \frac{c}{2}\frac{v0}{v0 - \texttt{Sign[z]}\, u}$$

H-function

H-function (Stibbs-Weir)

In[111]:=
$$\texttt{halfspaceAlbedoProblemIsotropic`H[}\alpha\texttt{\_, u\_] :=}$$
$$\texttt{Exp}\left[\frac{-u}{\texttt{Pi}}\texttt{NIntegrate}\left[\frac{\texttt{Log}[1-\alpha\, t\, \texttt{Cot}[t]]}{\texttt{Cos}[t]^2+u^2\,\texttt{Sin}[t]^2}, \left\{t, 0, \frac{\texttt{Pi}}{2}\right\}\right]\right]$$

In[112]:=
$$\texttt{halfspaceAlbedoProblemIsotropic`Hmoment[}\alpha\texttt{\_, j\_] :=}$$
$$\texttt{NIntegrate}\left[\texttt{halfspaceAlbedoProblemIsotropic`H}[\alpha, u]\, u^j, \{u, 0, 1\}\right]$$

Approximate H-function

In[113]:=
```
Clear[n, y];
```
$$\texttt{halfspaceAlbedoProblemIsotropic`H2[}\alpha\texttt{\_, u\_] :=}$$
$$\left(1-(1-y)\, u\left(n+\left(1-\frac{n}{2}-n\,u\right)\texttt{Log}\left[\frac{1+u}{u}\right]\right)\right)^{-1} \texttt{ /. } n\to\frac{1-y}{1+y}\texttt{ /. } y\to(1-\alpha)^{1/2}$$

In[115]:=
$$\texttt{halfspaceAlbedoProblemIsotropic`HmomentApprox[}\alpha\texttt{\_, j\_] :=}$$
$$\frac{1}{j+1}\frac{2}{1+y}\left(1+\frac{j}{2\,(j+2)}\frac{1-y}{1+y}\right)\texttt{ /. } y\to\sqrt{1-\alpha}$$

Albedo

Exact Solution

In[116]:=
$$\texttt{halfspaceAlbedoProblemIsotropic`albedoexact[}\alpha\texttt{\_, ui\_] :=}$$
$$1-\sqrt{1-\alpha}\;\texttt{halfspaceAlbedoProblemIsotropic`H}[\alpha, ui]$$

In[117]:=
```
halfspaceAlbedoProblemIsotropic`albedoexact[α_] :=
  NIntegrate[2 u halfspaceAlbedoProblemIsotropic`albedoexact[α, u], {u, 0, 1}]
```

Single-scattered Albedo (Exact Solution)

In[118]:=
$$\texttt{halfspaceAlbedoProblemIsotropic`albedoSingleScatterExact[}\alpha\texttt{\_, ui\_] :=}$$
$$\frac{1}{2}\alpha\left(1+ui\,\texttt{Log}\left[\frac{ui}{1+ui}\right]\right)$$

In[119]:=
$$\texttt{halfspaceAlbedoProblemIsotropic`albedoSingleScatterExact[}\alpha\texttt{\_] :=} \frac{2}{3}\left(\alpha-\alpha\,\texttt{Log}[2]\right)$$

Double-scattered Albedo (Exact Solution)

In[120]:=

$$\texttt{halfspaceAlbedoProblemIsotropic`albedoDoubleScatterExact}[\alpha\_,\ \texttt{ui}\_] := \texttt{If}\Big[\texttt{ui} == 1,$$
$$\frac{1}{48}\,\alpha^2\,\left(\pi^2 - 6\,\left(-1 + \texttt{Log}[2]^2 + \texttt{Log}[4]\right)\right),$$
$$-\frac{1}{8}\,\alpha^2\,\Big(-1 + \texttt{Log}\Big[e^{2\,\texttt{ui}^2\,\left(\texttt{Log}[1-\texttt{ui}]\,\texttt{Log}[\texttt{ui}] - \texttt{PolyLog}\left[2,\frac{\texttt{ui}}{-1+\texttt{ui}}\right]\right)}\Big] +$$
$$2\,\texttt{ui}\,\left(\texttt{ui}\,\texttt{Log}\left[1 + \frac{1}{\texttt{ui}}\right]^2 + \texttt{Log}\left[\frac{4\,\texttt{ui}}{1+\texttt{ui}}\right] - \texttt{ui}\,\texttt{Log}[1-\texttt{ui}]\,\texttt{Log}[1+\texttt{ui}]\right) +$$
$$2\,\texttt{ui}^2\,\left(\texttt{PolyLog}\left[2, -\frac{1}{\texttt{ui}}\right] + \texttt{PolyLog}\left[2, \frac{1+\texttt{ui}}{-1+\texttt{ui}}\right]\right)\Big)$$
$$\Big]$$

In[121]:=

$$\texttt{halfspaceAlbedoProblemIsotropic`albedoDoubleScatterExact}[\texttt{c}\_] := \frac{1}{24}\,\texttt{c}^2\,\left(4 + \pi^2 - 16\,\texttt{Log}[2]\right)$$

Approximate General Albedos

Classical Diffusion

In[122]:=

$$\texttt{halfspaceAlbedoProblemIsotropic`albedoClassicalDiffusion}[\texttt{c}\_,\ \texttt{ui}\_] := \frac{\texttt{c}}{\left(1 + \frac{2}{3}\,\left(3\,(1-\texttt{c})\right)^{1/2}\right)\,\left(1 + \left(3\,(1-\texttt{c})\right)^{1/2}\,\texttt{ui}\right)}$$

Rigorous Diffusion

In[123]:=

$$\texttt{halfspaceAlbedoProblemIsotropic`albedoRigorousDiffusion}[\texttt{c}\_,\ \texttt{ui}\_] := \frac{\texttt{c}}{\left(1 + \frac{2\,(1-\texttt{c})}{\#}\right)\,\left(1 + \#\,\texttt{ui}\right)}\,\&\left[\frac{1}{\texttt{Casev0}[\texttt{c}]}\right]$$

Grosjean Modified Diffusion 1

In[124]:=

$$\texttt{halfspaceAlbedoProblemIsotropic`albedoGrosjeanDiffusion1}[\texttt{c}\_,\ \texttt{ui}\_] :=$$
$$\frac{\texttt{c}}{2}\,\left(1 - \texttt{ui}\,\texttt{Log}\left[\frac{1+\texttt{ui}}{\texttt{ui}}\right]\right)\,\left(1 - \frac{\texttt{c}\,(13 - 5\,\texttt{c})}{13 - 5\,\texttt{c} + \frac{16}{3}\,(2-\texttt{c})\,\texttt{K}}\right) + \frac{\texttt{c}^2\,(13 - 5\,\texttt{c})}{(1 + \texttt{K}\,\texttt{ui})\,\left(13 - 5\,\texttt{c} + \frac{16}{3}\,(2-\texttt{c})\,\texttt{K}\right)} -$$
$$\frac{5}{4}\,\frac{\texttt{K}\,\texttt{c}^2\,(2-\texttt{c})\,\left(1 - 2\,\texttt{ui} + 2\,\texttt{ui}^2\,\texttt{Log}\left[\frac{1+\texttt{ui}}{\texttt{ui}}\right]\right)}{13 - 5\,\texttt{c} + \frac{16}{3}\,(2-\texttt{c})\,\texttt{K}}\ /.\ \texttt{K} \to \left(\frac{3\,(1-\texttt{c})}{2-\texttt{c}}\right)^{1/2}$$

Grosjean Modified Diffusion 2

In[125]:=

$$\texttt{halfspaceAlbedoProblemIsotropic`albedoGrosjeanDiffusion2}[\texttt{c}\_,\ \texttt{ui}\_] :=$$
$$\frac{\texttt{c}}{2}\,\left(1 - \texttt{ui}\,\texttt{Log}\left[\frac{1+\texttt{ui}}{\texttt{ui}}\right]\right)\,\left(1 - \frac{\texttt{c}}{1 + \frac{2\,(2-\texttt{c})\,\texttt{K}}{3}}\right) + \frac{\texttt{c}^2}{(1 + \texttt{K}\,\texttt{ui})\,\left(1 + \frac{2\,(2-\texttt{c})\,\texttt{K}}{3}\right)}\ /.\ \texttt{K} \to \left(\frac{3\,(1-\texttt{c})}{2-\texttt{c}}\right)^{1/2}$$

Approximate Normal Incidence Albedos

[Pomraning 1965]

```
In[126]:= halfspaceAlbedoProblemIsotropic`albedoNormalPomraning[α_] :=
            2
  ──────────────────── (Log[1 + #] - #) &[1 / Casev0[α]]
  (1 + #) Log[1 - #^2]
```

[Prahl 2002]

```
In[127]:= halfspaceAlbedoProblemIsotropic`albedoNormalPrahl[α_] :=
  (1 - #) (1 - 0.128 #)
  ───────────────────── &[Sqrt[1 - α]]
       1 + 1.83 #
```

Approximate WhiteSky Albedos

Integral of H2

```
In[128]:= halfspaceAlbedoProblemIsotropic`albedoH2[α_] :=
  NIntegrate[2 u (1 - Sqrt[1 - α] halfspaceAlbedoProblemIsotropic`H2[α, u]), {u, 0, 1}]
```

[Pomraning 1965]

```
In[129]:= halfspaceAlbedoProblemIsotropic`albedoWhiteskyPomraning[α_] :=
        -4
  ─────────────── (Log[1 + #] - #)^2 &[1 / Casev0[α]]
  #^2 Log[1 - #^2]
```

[van de Hulst 1980]

```
In[130]:= halfspaceAlbedoProblemIsotropic`albedoWhiteskyVanDeHulst[α_] :=
  (1 - #) (1 - 0.139 #)
  ───────────────────── &[Sqrt[1 - α]]
       1 + 1.17 #
```

BRDF

Exact Solution

```
In[131]:= halfspaceAlbedoProblemIsotropic`BRDF[α_, ui_, uo_] :=
        1
  ──────────────── α halfspaceAlbedoProblemIsotropic`H2[α, ui]
  4 Pi (ui + uo)
    halfspaceAlbedoProblemIsotropic`H2[α, uo]
```

```
In[132]:= halfspaceAlbedoProblemIsotropic`BRDFsinglescatter[α_, ui_, uo_] := α (1/(4 Pi)) (1/(ui + uo))
```

```
In[133]:= halfspaceAlbedoProblemIsotropic`BRDFdoublescatter[α_, ui_, uo_] :=
   α^2   (ui ArcCoth[1 + 2 ui] + uo ArcCoth[1 + 2 uo])
  ───── ───────────────────────────────────────────────
  4 Pi                     ui + uo
```

Internal Distribution

Fluence - Delta Illumination (“exact”)

NB: Uses the H2 approximation for efficiency

```
In[134]:= halfspaceAlbedoProblemIsotropic`ϕexact[z_, α_, ui_] :=
  halfspaceAlbedoProblemIsotropic`H2[α, ui]
  ─────────────────────────────────────────
                    2 Pi
    ((Caseψ0[ui, #, α, z] / (CaseN0[α, #] halfspaceAlbedoProblemIsotropic`H2[α,
           #])) Exp[-z / #] + NIntegrate[(Caseψ0[ui, v, α, 1] Exp[-z / v]) /
        (CaseN[α, v] halfspaceAlbedoProblemIsotropic`H2[α, v]),
       {v, 0, ui, 1}, Method → "PrincipalValue", PrecisionGoal → 5] + If[ui < 1,
       (Exp[-z / ui] / (CaseN[α, ui] halfspaceAlbedoProblemIsotropic`H2[α, ui]))
        Caseλ[ui, α], 0]) &[Casev0[α]]
```

Fluence - Delta Illumination (Rigorous Diffusion)

```
In[135]:= halfspaceAlbedoProblemIsotropic`ϕrigorousdiffusion[z_, α_, ui_] :=
  halfspaceAlbedoProblemIsotropic`H2[α, ui]
  ───────────────────────────────────────── ((Caseψ0[ui, #, α, z] / (CaseN0[α, #]
                    2 Pi
           halfspaceAlbedoProblemIsotropic`H2[α, #])) Exp[-z / #]) &[Casev0[α]]
```

Fluence - White-Sky Illumination (Rigorous Diffusion)

NB: Uses the H2 approximation for efficiency

```
In[136]:= halfspaceAlbedoProblemIsotropic`ϕrigorousdiffusion[z_, α_] :=
   1
  ── (((# √(1 - α)) / (CaseN0[α, #] halfspaceAlbedoProblemIsotropic`H2[α, #]))
  Pi
      Exp[-z / #]) &[Casev0[α]]
```

Fluence - White-Sky Illumination (“exact”)

NB: Uses the H2 approximation for efficiency

```
In[137]:= halfspaceAlbedoProblemIsotropic`ϕexact[z_, α_] :=
  halfspaceAlbedoProblemIsotropic`ϕrigorousdiffusion[z, α] + 1/Pi NIntegrate[
    (((v √(1 - α)) / (CaseN[α, v] halfspaceAlbedoProblemIsotropic`H2[α, v]))
      Exp[-z / v]), {v, 0, 1}]
```

Radiance - White-sky Illumination (“exact”)

```
In[138]:= halfspaceAlbedoProblemIsotropic`Lrigorousdiffusion[z_, α_, u_] :=
   1
  ── (((# √(1 - α) Caseψ0[-u, #, α, z]) / (CaseN0[α, #]
  Pi
           halfspaceAlbedoProblemIsotropic`H2[α, #])) Exp[-z / #]) &[Casev0[α]]
```

In[139]:=
```
halfspaceAlbedoProblemIsotropic`Lexact[z_, α_, u_] :=
  halfspaceAlbedoProblemIsotropic`Lrigorousdiffusion[z, α, u] +
   1/Pi (NIntegrate[(v Sqrt[1 - α] Caseψ0[-u, v, α, 1] Exp[-z / v]) /
        (CaseN[α, v] halfspaceAlbedoProblemIsotropic`H2[α, v]),
       {v, 0, -u, 1}, Method → "PrincipalValue", PrecisionGoal → 5] +
      If[u < 0, ((-u Sqrt[1 - α] Exp[-z / -u]) / (CaseN[α, -u]
            halfspaceAlbedoProblemIsotropic`H2[α, Abs[u]])) Caseλ[-u, α], 0])
```

Angular Moments

Delta-Illumination angular moments of exitant flux

In[140]:=
```
halfspaceAlbedoProblemIsotropic`angularMmoments[α_, ui_, H_, Hmoment_] := {
   1 - Sqrt[1 - α] H[ui],
   ui (-1 + H[ui] (Sqrt[1 - α] + (α Hmoment[1])/(2 ui))),
   ui^2 (1 + H[ui] (-Sqrt[1 - α] + 1/2 α (-Hmoment[1]/ui + Hmoment[2]/ui^2)))
  }
```

White-Sky Illumination angular moments of exitant flux

In[141]:=
```
halfspaceAlbedoProblemIsotropic`angularMmoments[c_, Hmoment_] := {
   1 - 2 Sqrt[1 - c] Hmoment[1],
   2/3 (-1 + 3 (1/2 c Hmoment[1]^2 + Sqrt[1 - c] Hmoment[2])),
   1/2 (1 - 4 Sqrt[1 - c] Hmoment[3])
  }
```

Fluence Moments

Delta-Illumination spatial fluence moments

In[142]:=
```
halfspaceAlbedoProblemIsotropic`x[c_, 0, ui_, Hmoment_] := 1;
halfspaceAlbedoProblemIsotropic`x[c_, n_, ui_, Hmoment_] :=
  n! (ui^n + c/(2 (1 - c)^(1/2)) Sum[Hmoment[j]/(n - j)! x[c, n - j, ui, Hmoment], {j, 1, n}])
```

In[144]:=
```
halfspaceAlbedoProblemIsotropic`spatialMoments[c_, ui_, H_, Hmoment_] :=
  H[ui]/(1 - c)^(1/2) Table[halfspaceAlbedoProblemIsotropic`x[c, n, ui, Hmoment], {n, 0, 4}]
```

White-Sky Illumination spatial fluence moments

In[145]:=
```
halfspaceAlbedoProblemIsotropic`spatialMoments[c_, Hmoment_] := {
```
$$\left\{\frac{2\,\text{Hmoment}[1]}{\sqrt{1-c}},\quad 2\left(\frac{c\,\text{Hmoment}[1]^2}{2\,(1-c)}+\frac{\text{Hmoment}[2]}{\sqrt{1-c}}\right),\quad 2\left(\frac{c^2\,\text{Hmoment}[1]^3}{2\,(1-c)^{3/2}}+\frac{2\,c\,\text{Hmoment}[1]\,\text{Hmoment}[2]}{1-c}+\frac{2\,\text{Hmoment}[3]}{\sqrt{1-c}}\right)\right\}$$

Load MC Data

Delta Incidence

In[146]:=
```
halfspaceAlbedoProblemIsotropic`deltafs = FileNames[
   "code/3D_medium/halfspace/albedoProblem/data/albedoproblem_delta*.txt"];
```

In[147]:=
```
halfspaceAlbedoProblemIsotropic`indexdelta[x_] := Module[{data, α, Σt, ui},
   data = Import[x, "Table"];
   Σt = data[[1, -1]];
   α = data[[2, 3]];
   ui = data[[1, -4]];
   {α, Σt, ui, data}];
halfspaceAlbedoProblemIsotropic`simulationsdelta =
  halfspaceAlbedoProblemIsotropic`indexdelta /@
   halfspaceAlbedoProblemIsotropic`deltafs;
halfspaceAlbedoProblemIsotropic`alphas =
 Union[#[[1]] & /@ halfspaceAlbedoProblemIsotropic`simulationsdelta]
```

Out[149]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[150]:=
```
halfspaceAlbedoProblemIsotropic`muts =
 Union[#[[2]] & /@ halfspaceAlbedoProblemIsotropic`simulationsdelta]
```

Out[150]=
```
{1}
```

In[151]:=
```
halfspaceAlbedoProblemIsotropic`uis =
 Union[#[[3]] & /@ halfspaceAlbedoProblemIsotropic`simulationsdelta]
```

Out[151]=
```
{0.1, 0.25, 0.5, 1}
```

In[152]:=
```
halfspaceAlbedoProblemIsotropic`numcollorders =
  halfspaceAlbedoProblemIsotropic`simulationsdelta[[1]][[4]][[2, 13]];
halfspaceAlbedoProblemIsotropic`maxz =
 halfspaceAlbedoProblemIsotropic`simulationsdelta[[1]][[4]][[2, 7]];
halfspaceAlbedoProblemIsotropic`dz =
 halfspaceAlbedoProblemIsotropic`simulationsdelta[[1]][[4]][[2, 9]];
halfspaceAlbedoProblemIsotropic`numz = Floor[
   halfspaceAlbedoProblemIsotropic`maxz / halfspaceAlbedoProblemIsotropic`dz];
```

WhiteSky Illumination

In[155]:=
```
halfspaceAlbedoProblemIsotropic`whiteskyfs = FileNames[
   "code/3D_medium/halfspace/albedoProblem/data/albedoproblem_whitesky*.txt"];
```

In[156]:=
```
halfspaceAlbedoProblemIsotropic`indexwhitesky[x_] := Module[{data, α, Σt},
   data = Import[x, "Table"];
   Σt = data[[1, -1]];
   α = data[[2, 3]];
   {α, Σt, data}];
halfspaceAlbedoProblemIsotropic`simulationswhitesky =
  halfspaceAlbedoProblemIsotropic`indexwhitesky /@
   halfspaceAlbedoProblemIsotropic`whiteskyfs;
```

Util

In[158]:=
```
halfspaceAlbedoProblemIsotropic`plotpoints[data_, du_] :=
  Table[{du i - 0.5 du, data[[i]]}, {i, 1, Length[data]}];
```

In[159]:=
```
halfspaceAlbedoProblemIsotropic`plotpoints[data_, d_, min_] :=
  Table[{min + d i - 0.5 d, data[[i]]}, {i, 1, Length[data]}];
```

Albedo Benchmarks

Normal Incidence

In[160]:=
```
MCMLBenchmarkData =
  {{0.9990009990009991`, 0.912484`}, {0.09090909090909091`, 0.0147879`},
   {0.16666666666666666`, 0.0286725`}, {0.3333333333333333`, 0.0653595`},
   {0.5`, 0.11524`}, {0.6666666666666666`, 0.189042`},
   {0.9090909090909091`, 0.432273`}, {0.9950248756218907`, 0.816965`},
   {0.8333333333333334`, 0.319804`}, {0.998003992015968`, 0.879038`},
   {0.9995002498750625`, 0.937142`}, {0.9998000399920016`, 0.959626`},
   {0.9999000099990001`, 0.971259`}, {0.9523809523809523`, 0.543346`},
   {0.9803921568627451`, 0.673321`}, {0.9900990099009901`, 0.75327`}};
```

In[161]:=
```
halfspaceAlbedoProblemIsotropic`normalRs = {#[[1]], #[[4]][[3, 3]]} & /@
  Select[halfspaceAlbedoProblemIsotropic`simulationsdelta, #[[3]] == 1 &]
```

Out[161]=
```
{{0.01, 0.0015482}, {0.1, 0.0164138}, {0.3, 0.0572445},
 {0.5, 0.115279}, {0.7, 0.208883}, {0.8, 0.28562}, {0.95, 0.535438},
 {0.999, 0.913071}, {0.99, 0.753413}, {0.9, 0.414801}}
```

In[162]:=
```
Quiet[Show[
  Show[
   ListLogLinearPlot[halfspaceAlbedoProblemIsotropic`normalRs,
    PlotStyle → {PointSize[0.01], Black}],
   ListLogLinearPlot[MCMLBenchmarkData, PlotStyle → {PointSize[0.006], Red}],
   LogLinearPlot[Quiet[halfspaceAlbedoProblemIsotropic`albedoexact[c, 1]],
    {c, 0.001, .9999}, PlotRange → All]
  ], Frame → True, FrameLabel → {{R[α, 1],}, {"Single scattering albedo: α",
     "Total Reflectance/Albedo R(α): isotropically-scattering half
       space, normal incidence (ui=1), indexed matched boundary"}}
 ]]
```

Out[162]=



Albedo - Delta Illumination - General incidence

In[163]:=
```
Manipulate[
 Module[{Rs},
  Rs = {#[[1]], #[[4]][[3, 3]]} & /@
    Select[halfspaceAlbedoProblemIsotropic`simulationsdelta, #[[3]] == ui &];
  Quiet[Show[
    Show[
     ListLogLinearPlot[Rs, PlotStyle → {PointSize[0.01], Black}],
     LogLinearPlot[Quiet[halfspaceAlbedoProblemIsotropic`albedoexact[c, ui]],
      {c, 0.001, .9999}, PlotRange → All]
    ], Frame → True, FrameLabel → {{R[α, ui],},
      {"Single scattering albedo: α", "Total Reflectance/Albedo R(α,uᵢ):
          isotropically-scattering half space\nincidence (uᵢ=" <>
        ToString[ui] <> "), indexed-matched boundary"}}
   ]]
 ]
 , {ui, halfspaceAlbedoProblemIsotropic`uis}]
```

Out[163]=



Single & Double-Scattered Albedo - Delta Illumination - General incidence

In[164]:=
```
Manipulate[
 Module[{RsSingle, RsDouble},
  RsSingle = {#[[1]], #[[4]][[5, 2]]} & /@
    Select[halfspaceAlbedoProblemIsotropic`simulationsdelta, #[[3]] == ui &];
  RsDouble = {#[[1]], #[[4]][[5, 3]]} & /@
    Select[halfspaceAlbedoProblemIsotropic`simulationsdelta, #[[3]] == ui &];
  Quiet[Show[
    Show[
     ListLogLinearPlot[RsSingle, PlotStyle → {PointSize[0.01], Black}],
     ListLogLinearPlot[RsDouble, PlotStyle → {PointSize[0.01], Black}],
     LogLinearPlot[
      Quiet[halfspaceAlbedoProblemIsotropic`albedoSingleScatterExact[c, ui]],
      {c, 0.001, .9999}, PlotRange → All, PlotStyle → Dashed],
     LogLinearPlot[Quiet[
       halfspaceAlbedoProblemIsotropic`albedoDoubleScatterExact[c, ui]],
      {c, 0.001, .9999}, PlotRange → All]
    ], Frame → True, FrameLabel → {{Rn[α, ui],}, {"Single scattering albedo: α",
      "Single (Dashed) and Double-Scattered Reflectance/Albedo Rn(α,ui):
         \nisotropically-scattering half space\nincidence (ui=" <>
       ToString[ui] <> "), indexed-matched boundary"}}
   ]]
 ]
 , {ui, halfspaceAlbedoProblemIsotropic`uis}]
```

Out[164]=



Approximate General Albedos

In[165]:=
```
Manipulate[
 Module[{Rs},
  Rs = {#[[1]], #[[4]][[3, 3]]} & /@
    Select[halfspaceAlbedoProblemIsotropic`simulationsdelta, #[[3]] == ui &];
  GraphicsRow[{Quiet[Show[
      Show[
       ListPlot[Rs, PlotStyle → {PointSize[0.01], Black}],
       Plot[halfspaceAlbedoProblemIsotropic`albedoClassicalDiffusion[c, ui],
        {c, 0.001, .9999},
        PlotRange → All, PlotStyle → Dashed],
       Plot[halfspaceAlbedoProblemIsotropic`albedoRigorousDiffusion[c, ui],
        {c, 0.001, .9999},
        PlotRange → All, PlotStyle → Red]
      ], Frame → True,
      FrameLabel → {{R[α, ui],}, {"Single scattering albedo: α",
         "Total Reflectance/Albedo R(α,uᵢ): isotropically-scattering
            half space\nincidence (uᵢ=" <> ToString[ui] <>
          "), indexed-matched boundary\nMC(dots), Classical Diffusion
            (dashed), Rigorous Diffusion (thin)"}},
      ImageSize → 500
     ]],
    Quiet[Show[
      Show[
       Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, ui] -
         halfspaceAlbedoProblemIsotropic`albedoClassicalDiffusion[
          c, ui], {c, 0.001, .9999},
        PlotRange → All, PlotStyle → Dashed],
       Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, ui] -
         halfspaceAlbedoProblemIsotropic`albedoRigorousDiffusion[
          c, ui], {c, 0.001, .9999},
        PlotRange → All, PlotStyle → Red],
       Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, ui] -
         halfspaceAlbedoProblemIsotropic`albedoGrosjeanDiffusion1[
          c, ui], {c, 0.001, .9999},
        PlotRange → All, PlotStyle → DotDashed]
      ], Frame → True,
      FrameLabel → {{"R_exact - R_approx",}, {"Single scattering albedo: α",
         "Total Reflectance/Albedo R(α,uᵢ): isotropically-scattering
            half space\nincidence (uᵢ=" <> ToString[ui] <>
          "), indexed-matched boundary\nClassical Diffusion (dashed),
            Rigorous Diffusion (thin), Grosjean1 (DotDashed)"}}
     ]]
   }]
 ]
 , {ui, halfspaceAlbedoProblemIsotropic`uis}]
```

ui 0.1 0.25 0.5 1

Out[165]=



Normal incidence - Approximate Albedos

In[166]:=
```
Quiet[Show[
  Show[
   Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, 1] -
     halfspaceAlbedoProblemIsotropic`albedoClassicalDiffusion[c, 1], {c,
     0.001, .9999},
    PlotRange → All, PlotStyle → Dashed],
   Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, 1] -
     halfspaceAlbedoProblemIsotropic`albedoRigorousDiffusion[c, 1], {c,
     0.001, .9999},
    PlotRange → All, PlotStyle → Red],
   Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, 1] -
     halfspaceAlbedoProblemIsotropic`albedoGrosjeanDiffusion1[c, 1], {c,
     0.001, .9999},
    PlotRange → All, PlotStyle → DotDashed],
   Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, 1] -
     halfspaceAlbedoProblemIsotropic`albedoNormalPomraning[c], {c,
     0.001, .9999},
    PlotRange → All, PlotStyle → {Thick, Dashed}],
   Plot[halfspaceAlbedoProblemIsotropic`albedoexact[c, 1] -
     halfspaceAlbedoProblemIsotropic`albedoNormalPrahl[c], {c, 0.001, .9999},
    PlotRange → All, PlotStyle → Opacity[0], Filling → Axis]
  ], Frame → True, ImageSize → 500,
  FrameLabel → {{"R_exact - R_approx",},
    {"Single scattering albedo: α", "Total Reflectance/Albedo
      R(α,1): isotropically-scattering half space\nNormal
      incidence (u_i=1), indexed-matched boundary\nClassical
      Diffusion (dashed), Rigorous Diffusion (thin), Grosjean1
      (DotDashed)\nPomraning (Thick-Dashed), Prahl (Filled)"}}
]]
```

Out[166]=


Total Reflectance/Albedo R(α,1): isotropically-scattering half space
Normal incidence (u_i=1), indexed-matched boundary
Classical Diffusion (dashed), Rigorous Diffusion (thin), Grosjean1 (DotDashed)
Pomraning (Thick-Dashed), Prahl (Filled)
$R_{exact} - R_{approx}$
0.000
−0.005
−0.010
0.0
0.2
0.4
0.6
0.8
1.0
Single scattering albedo: α

Albedo - WhiteSky Illumination - Integral using H2 approximation

In[167]:=

```
Module[{Rs},
 Rs = {#[[1]], #[[3]][[3, 3]]} & /@
   halfspaceAlbedoProblemIsotropic`simulationswhitesky;
 Quiet[Show[
   Show[
    ListLogLinearPlot[Rs, PlotStyle → {PointSize[0.01], Black}],
    LogLinearPlot[Quiet[halfspaceAlbedoProblemIsotropic`albedoH2[c]],
     {c, 0.001, .9999}, PlotRange → All]
   ], Frame → True, FrameLabel → {{R[α],}, {"Single scattering albedo: α",
     "Total Reflectance/Albedo R(α): isotropically-scattering half
       space\nwhite-sky illumination, indexed-matched boundary"}}
  ]]
]
```

Out[167]=



In[168]:=
```
Module[{RsSingle, RsDouble},
 RsSingle = {#[[1]], #[[3]][[5, 2]]} & /@
   halfspaceAlbedoProblemIsotropic`simulationswhitesky;
 RsDouble = {#[[1]], #[[3]][[5, 3]]} & /@
   halfspaceAlbedoProblemIsotropic`simulationswhitesky;
 Quiet[Show[
   Show[
    ListLogLinearPlot[RsSingle, PlotStyle → {PointSize[0.01], Black}],
    ListLogLinearPlot[RsDouble, PlotStyle → {PointSize[0.01], Black}],
    LogLinearPlot[
     Quiet[halfspaceAlbedoProblemIsotropic`albedoSingleScatterExact[c]],
     {c, 0.001, .9999}, PlotRange → All, PlotStyle → Dashed],
    LogLinearPlot[Quiet[
      halfspaceAlbedoProblemIsotropic`albedoDoubleScatterExact[c]],
     {c, 0.001, .9999}, PlotRange → All]
   ], Frame → True, FrameLabel → {{Rn[α],}, {"Single scattering albedo: α",
      "Single (Dashed) and Double-Scattered Reflectance/Albedo Rn(α):
        \nIsotropically-Scattering Half Space\nWhite-Sky
        Illumination, indexed-matched boundary"}}
  ]
 ]
]
```

Out[168]=



Approximate White-sky Albedos

In[169]:=
```
Quiet[Show[
  Show[
   Plot[halfspaceAlbedoProblemIsotropic`albedoH2[c] -
     halfspaceAlbedoProblemIsotropic`albedoWhiteskyPomraning[c], {c,
     0.001, .9999},
    PlotRange → All, PlotStyle → Dashed],
   Plot[halfspaceAlbedoProblemIsotropic`albedoH2[c] -
     halfspaceAlbedoProblemIsotropic`albedoWhiteskyVanDeHulst[c], {c,
     0.001, .9999},
    PlotRange → All, PlotStyle → Red]
  ], Frame → True, ImageSize → 500,
  FrameLabel → {{"Rexact - Rapprox",}, {"Single scattering albedo: α",
     "Total Reflectance/Albedo R(α): isotropically-scattering
       half space\nWhite-sky illumination, indexed-matched
       boundary\nPomraning (dashed), VanDeHulst (thin)"}}
 ]]
```

Out[169]=



BRDF Benchmarks

In[170]:=
```
Manipulate[
 Module[{emerging, imdata, sim, du, emergingS, emergingD, stride},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationsdelta,
    #[[3]] == ui && #[[1]] == α && #[[2]] == 1 &];
  imdata = sim[[4]];
  du = imdata[[2, 5]];
  stride = 3;
  emerging = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[7]], du][[
    1 ;; -1 ;; stride]];
  emergingS = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[9]], du][[
    1 ;; -1 ;; stride]];
  emergingD = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[11]], du][[
    1 ;; -1 ;; stride]];
  Show[
   Show[
    Plot[2 Pi u halfspaceAlbedoProblemIsotropic`BRDF[α, ui, u],
     {u, 0, 1}, PlotRange → All],
    Plot[2 Pi u halfspaceAlbedoProblemIsotropic`BRDFsinglescatter[α, ui, u],
     {u, 0, 1}, PlotStyle → Dashed],
    Plot[2 Pi u halfspaceAlbedoProblemIsotropic`BRDFdoublescatter[α, ui, u],
     {u, 0, 1}, PlotStyle → Thick],
    ListPlot[emerging, PlotStyle → {PointSize[0.004], Black}],
    ListPlot[emergingS, PlotStyle → {PointSize[0.004], Black}],
    ListPlot[emergingD, PlotStyle → {PointSize[0.004], Black}]
   ], Frame → True, ImageSize → 500,
   FrameLabel → {{2 π uₒ fᵣ[uᵢ, uₒ],}, {uₒ, "Emerging Distribution fᵣ(uᵢ,uₒ):
         isotropically-scattering half space\nincidence (uᵢ=" <>
       ToString[ui] <> "), indexed-matched boundary\nTotal (thin),
         Single-Scatter (dashed), Double-scatter (thick)"}}
  ]
 ]
 ,
 {{α, 0.8}, halfspaceAlbedoProblemIsotropic`alphas},
 {{ui, 0.25}, halfspaceAlbedoProblemIsotropic`uis}
]
```

Out[170]=

α 0.8

ui 0.1 0.25 0.5 1



Internal Distribution Benchmarks

Delta Illumination - Fluence - Exact Solution

In[171]:=
```
Manipulate[
 Module[{imdata, sim, dz, maxz, fluence3, fluence, stride, numzs},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationsdelta,
    #[[3]] == ui && #[[1]] == α && #[[2]] == Σt &];
  imdata = sim[[4]];
  maxz = imdata[[2, 7]];
  dz = imdata[[2, 9]];
  stride = 3;
  numzs = Length[imdata[[13]]];
  fluence = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[13]], dz][[
    1 ;; Floor[numzs / 5]]];
  fluence3 = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[13]], dz][[
    1 ;; -1 ;; stride]];

  Quiet[
   GraphicsRow[{
     Show[
      Show[
       Plot[2 Pi halfspaceAlbedoProblemIsotropic`ϕexact[z, α, ui],
        {z, 0, maxz/5}, PlotRange → All],
       ListPlot[fluence, PlotStyle → {PointSize[0.004], Black}]
      ], Frame → True, ImageSize → 500,
      FrameLabel → {{ϕ[z],}, {z, "Fluence ϕ(z): isotropically-scattering
            half space\nincidence (u_i=" <> ToString[ui] <>
          "), indexed-matched boundary, α = " <> ToString[α]}}
     ],
     Show[
      Show[
       LogPlot[2 Pi halfspaceAlbedoProblemIsotropic`ϕexact[z, α, ui],
        {z, 0, maxz}, PlotRange → All],
       ListLogPlot[fluence3, PlotStyle → {PointSize[0.004], Black}]
      ], Frame → True, ImageSize → 500,
      FrameLabel → {{ϕ[z],}, {z, "Fluence ϕ(z): isotropically-scattering
            half space\nincidence (u_i=" <> ToString[ui] <>
          "), indexed-matched boundary, α = " <> ToString[α]}}
     ]
    }]
  ]
 ]
 ,
 {{α, 0.99}, halfspaceAlbedoProblemIsotropic`alphas},
 {{ui, 0.5}, halfspaceAlbedoProblemIsotropic`uis},
 {Σt, halfspaceAlbedoProblemIsotropic`muts}
]
```

Out[171]=


α 0.99
ui 0.1 0.25 0.5 1
Σt 1
Fluence φ(z): isotropically-scattering half space
incidence (ui=0.5), indexed-matched boundary, α = 0.99
φ(z)
3.5
3.0
2.5
2.0
1.5
0 1 2 3 4 5
z

White-Sky Illumination - Fluence - Exact Solution

```
In[172]:= Manipulate[
  Module[{imdata, sim, dz, maxz, fluence3, fluence, stride, numzs},
   sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationswhitesky,
     #[[1]] == α && #[[2]] == Σt &];
   imdata = sim[[3]];
   maxz = imdata[[2, 7]];
   dz = imdata[[2, 9]];
   stride = 3;
   numzs = Length[imdata[[13]]];
   fluence = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[13]], dz][[
     1 ;; Floor[numzs / 5]]];
   fluence3 = halfspaceAlbedoProblemIsotropic`plotpoints[imdata[[13]], dz][[
     1 ;; -1 ;; stride]];

   Quiet[
    GraphicsRow[{
      Show[
       Show[
        Plot[2 Pi halfspaceAlbedoProblemIsotropic`ϕexact[z, α],
         {z, 0, maxz/5}, PlotRange → All],
        ListPlot[fluence, PlotStyle → {PointSize[0.004], Black}]
       ], Frame → True, ImageSize → 500,
       FrameLabel → {{ϕ[z],}, {z, "Fluence ϕ(z): isotropically-scattering
             half space white-Sky illumination\n
             indexed-matched boundary, α = " <> ToString[α]}}
      ],
      Show[
       Show[
        LogPlot[2 Pi halfspaceAlbedoProblemIsotropic`ϕexact[z, α],
         {z, 0, maxz}, PlotRange → All],
        ListLogPlot[fluence3, PlotStyle → {PointSize[0.004], Black}]
       ], Frame → True, ImageSize → 500,
       FrameLabel → {{ϕ[z],}, {z, "Fluence ϕ(z): isotropically-scattering
             half space white-Sky illumination\n
             indexed-matched boundary, α = " <> ToString[α]}}
      ]
     }]
   ]
  ]
  ,
  {{α, 0.3}, halfspaceAlbedoProblemIsotropic`alphas},
  {Σt, halfspaceAlbedoProblemIsotropic`muts}
 ]
```

Out[172]=


α 0.3
Σt 1
Fluence ϕ(z): isotropically–scattering half space white–Sky illumination
indexed–matched boundary, α = 0.3
ϕ(z)
z

White-Sky Illumination - Radiance - Exact solution

In[176]:=

```
Manipulate[
 Module[{imdata, sim, dz, maxz, Lz, numzs, z, du},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationswhitesky,
    #[[1]] == α && #[[2]] == Σt &];
  imdata = sim[[3]];
  maxz = imdata[[2, 7]];
  dz = imdata[[2, 9]];
  du = imdata[[2, 5]];
  numzs = Length[imdata[[13]]];

  z = 0.5 dz + (n - 1) dz;

  pointsLz = halfspaceAlbedoProblemIsotropic`plotpoints[
    imdata[[-halfspaceAlbedoProblemIsotropic`numz + n - 2]], du, -1];
  Show[
   Show[
    Plot[Pi halfspaceAlbedoProblemIsotropic`Lexact[z, α, u], {u, -1, 1}],
    Plot[Pi halfspaceAlbedoProblemIsotropic`Lrigorousdiffusion[z, α, u],
     {u, -1, 1}, PlotStyle → Dashed],
    ListPlot[pointsLz[[1 ;; -2 ;; 3]], PlotStyle → Black]
   ], Frame → True, ImageSize → 500, FrameLabel → {{L[z, u],},
     {u, "Radiance L(z,u): isotropically-scattering half space white-Sky
         illumination\n indexed-matched boundary, exact(thin),
         rigorous diffusion (dashed), α = " <> ToString[α]}}
  ]
 ]
 ,
 {{α, 0.7}, halfspaceAlbedoProblemIsotropic`alphas},
 {Σt, halfspaceAlbedoProblemIsotropic`muts},
 {{n, 3}, Range[If[NumberQ[halfspaceAlbedoProblemIsotropic`numz],
    halfspaceAlbedoProblemIsotropic`numz, 1]]}
]
```

Out[176]=


α 0.7
Σt 1
n 3
Radiance L(z,u): isotropically-scattering half space white-Sky illumination
indexed-matched boundary, exact(thin), rigorous diffusion (dashed), α = 0.7
L(0.25, u)
u
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
-1.0
-0.5
0.0
0.5
1.0

Angular Moments of Exitant Radiance - Delta Illumination

In[177]:=
```
Manipulate[
 Module[{imdata, sim, dz, maxz, fluence3, fluence, stride, numzs},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationsdelta,
    #[[3]] == ui && #[[1]] == α && #[[2]] == Σt &];
  imdata = sim[[4]];
  TableForm[{
    Join[{"MC:"}, imdata[[17]][[1 ;; 3]]],
    Join[{"Analytic:"},
     halfspaceAlbedoProblemIsotropic`angularMmoments[α, ui,
      halfspaceAlbedoProblemIsotropic`H[α, #] &,
      halfspaceAlbedoProblemIsotropic`HmomentApprox[α, #] &
     ]]
   }]
 ]
 ,
 {{α, 0.99}, halfspaceAlbedoProblemIsotropic`alphas},
 {{ui, 0.5}, halfspaceAlbedoProblemIsotropic`uis},
 {Σt, halfspaceAlbedoProblemIsotropic`muts}
]
```

Out[177]=

α 0.99

ui 0.1 0.25 0.5 1

Σt 1

MC:	0.815295	0.532314	0.394424
Analytic:	0.81514	0.537737	0.399148

Angular Moments of Exitant Radiance - White-Sky

In[173]:=
```
Manipulate[
 Module[{imdata, sim, dz, maxz, fluence3, fluence, stride, numzs},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationswhitesky,
    #[[1]] == α && #[[2]] == Σt &];
  imdata = sim[[3]];
  TableForm[{
    Join[{"MC:"}, imdata[[17]][[1 ;; 3]]],
    Join[{"Analytic:"}, halfspaceAlbedoProblemIsotropic`angularMmoments[
      α, halfspaceAlbedoProblemIsotropic`HmomentApprox[α, #] &]]
   }]
 ]
 ,
 {{α, 0.99}, halfspaceAlbedoProblemIsotropic`alphas},
 {Σt, halfspaceAlbedoProblemIsotropic`muts}
]
```

Out[173]=

α 0.99

Σt 1

MC:	0.795052	0.522534	0.388567
Analytic:	0.793388	0.535875	0.386777

Spatial Moments of Fluence - Delta Illumination

In[174]:=

```
Manipulate[
 Module[{imdata, sim, dz, maxz, fluence3, fluence, stride, numzs},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationsdelta,
    #[[3]] == ui && #[[1]] == α && #[[2]] == Σt &];
  imdata = sim[[4]];
  TableForm[{
    Join[{"MC:"}, imdata[[15]][[1 ;; 5]]],
    Join[{"Analytic:"}, halfspaceAlbedoProblemIsotropic`spatialMoments[α, ui,
      halfspaceAlbedoProblemIsotropic`H[α, #] &,
      halfspaceAlbedoProblemIsotropic`HmomentApprox[α, #] &
     ]]
   }]
 ]
 ,
 {{α, 0.99}, halfspaceAlbedoProblemIsotropic`alphas},
 {{ui, 0.5}, halfspaceAlbedoProblemIsotropic`uis},
 {Σt, halfspaceAlbedoProblemIsotropic`muts}
]
```

Out[174]=

α 0.99

ui 0.1 0.25 0.5 1

Σt 1

MC:	18.4767	102.65	1173.22	20 120.4	457 940
Analytic:	18.486	103.774	1204.17	21 047.7	490 762.

Spatial Moments of Fluence - White-Sky

In[175]:=
```
Manipulate[
 Module[{imdata, sim, dz, maxz, fluence3, fluence, stride, numzs},
  sim = SelectFirst[halfspaceAlbedoProblemIsotropic`simulationswhitesky,
    #[[1]] == α && #[[2]] == Σt &];
  imdata = sim[[3]];
  TableForm[{
    Join[{"MC:"}, imdata[[15]][[1 ;; 3]]],
    Join[{"Analytic:"}, halfspaceAlbedoProblemIsotropic`spatialMoments[
      α, halfspaceAlbedoProblemIsotropic`HmomentApprox[α, #] &]]
   }]
 ]
 ,
 {{α, 0.99}, halfspaceAlbedoProblemIsotropic`alphas},
 {Σt, halfspaceAlbedoProblemIsotropic`muts}
]
```

Out[175]=

α 0.99

Σt 1

MC:	20.4797	118.167	1365.25
Analytic:	20.6612	120.254	1401.84