Infinite Rod, Isotropic Point Source, Isotropic Scattering

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
In[1]:= SetDirectory[Import["~/.hitchhikerpath"]]
```

Exponential random flight

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - position coordinate in rod (source at x = 0)

Analytic solutions

'Radiance'

In[2]:= $\text{LR}[x\_, \alpha\_, \Sigma t\_] := \dfrac{e^{-\text{Abs}[x]\sqrt{1-\alpha}\,\Sigma t}\left(1+\sqrt{1-\alpha}\,\text{Sign}[x]\right)}{4\sqrt{1-\alpha}}$

In[3]:= $\text{LL}[x\_, \alpha\_, \Sigma t\_] := \dfrac{e^{-\text{Abs}[x]\sqrt{1-\alpha}\,\Sigma t}\left(1-\sqrt{1-\alpha}\,\text{Sign}[x]\right)}{4\sqrt{1-\alpha}}$

Fluence

In[4]:= $\phi[x\_, \alpha\_, \Sigma t\_] := \dfrac{1}{2\sqrt{1-\alpha}}\,E^{-\Sigma t\sqrt{1-\alpha}\,\text{Abs}[x]}$

n-th collided fluence

In[5]:= $\phi[x\_, n\_, \alpha\_, \Sigma t\_] := \frac{2^{-n-\frac{1}{2}} \alpha^n (\text{Abs}[x]\, \Sigma t)^{n+\frac{1}{2}} \text{BesselK}\left[n + \frac{1}{2}, \Sigma t\, \text{Abs}[x]\right]}{\sqrt{\text{Pi}}\ \text{Gamma}[n+1]}$

moments

In[6]:= $\phi m[\alpha\_, \Sigma t\_, k\_] := \frac{1}{2} \left(1 + (-1)^k\right) \Sigma t^{-1-k} (1-\alpha)^{-1-\frac{k}{2}} k\,!$

In[7]:= $\phi m[\alpha\_, \Sigma t\_, k\_, n\_] := \alpha^n \left(\frac{1}{2} (-1)^n \left(1 + (-1)^k\right) \Sigma t^{-1-k} \text{Binomial}\left[-1 - \frac{k}{2}, n\right] k\,!\right)$

load MC data

In[8]:=
```
ppoints[xs_, dx_, maxx_, Σt_] :=
  Table[{-maxx + dx (i - 1) + 0.5 dx, (1 / Σt) xs[[i]]}, {i, 1, Length[xs]}][[2 ;; -2]]
```

In[9]:=
```
fs = FileNames[
    "code/rod/infiniterod/Isotropicpointsource/data/infrod_isotropicpoint
      _isotropicscatter_exp*"];
```

In[10]:=
```
index[x_] := Module[{data, α, Σt},
    data = Import[x, "Table"];
    Σt = data[[1, 12]];
    α = data[[2, 3]];
    {α, Σt, data}];
simulations = index /@ fs;
```

In[12]:=
```
alphas = Union[#[[1]] & /@ simulations]
```

Out[12]= {0.1, 0.3, 0.5, 0.7, 0.9}

In[13]:=
```
muts = Union[#[[2]] & /@ simulations]
```

Out[13]= {1, 3}

Compare Deterministic and MC

Internal distributions

In[14]:=
```
Clear[α, Σt];
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  maxx = data[[2, 5]];
  dx = data[[2, 7]];
  numcollorders = data[[2, 11]];
  nummoments = data[[2, 13]];

  densmom = data[[8]];

  pointsCL = data[[4]];
  (* divide by Σt to convert collision density into L *)
  plotpointsCL = ppoints[pointsCL, dx, maxx, Σt];
  pointsCR = data[[6]];
  plotpointsCR = ppoints[pointsCR, dx, maxx, Σt];
  (* divide by Σt to convert collision density into fluence *)
  plotpointsϕ = ppoints[pointsCL + pointsCR, dx, maxx, Σt];

  plotϕ = Show[
    ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[ϕ[x, α, Σt], {x, -maxx, maxx}, PlotRange → All]
    , Frame → True,
    FrameLabel -> {{ϕ[x],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, fluence
          ϕ[x], α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]}}
   ];

  plotLL = Show[
    ListPlot[plotpointsCL, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LL[x, α, Σt], {x, -maxx, maxx}, PlotRange → All],
    Plot[ϕ[x, α, Σt], {x, -maxx, maxx}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L-[x],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, L-[x], α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];

  plotLR = Show[
    ListPlot[plotpointsCR, PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[LR[x, α, Σt], {x, -maxx, maxx}, PlotRange → All],
    Plot[ϕ[x, α, Σt], {x, -maxx, maxx}, PlotRange → All, PlotStyle → Dashed]
    , Frame → True,
    FrameLabel -> {{L+[x],},
      {x, "Infinite Rod, isotropic point, isotropic scattering, L+[x], α = " <>
        ToString[α] <> ", Σt = " <> ToString[Σt]}}, PlotRange → All
   ];
  Show[GraphicsGrid[{{plotϕ}, {plotLL}, {plotLR}}], ImageSize → 500]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}]
```

α 0.1 0.3 0.5 0.7 0.9

Σt 1 3

Out[15]=









Compare moments of ϕ

Divide these results, which are collision density moments, by Σt to produce radiance/fluence moments:

In[16]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  ϕmoments = {data[[8]]} / Σt;
  ks = {Table[k, {k, 0, nummoments - 1}]};
  analytic = Table[ϕm[α, Σt, k], {k, ks}];
  j = Join[ks, analytic, ϕmoments];
  TableForm[
   Join[{{"k", "analytic", "MC"}}, Transpose[j]]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts}]
```

Out[16]=

α: 0.1 | 0.3 | 0.5 | 0.7 | 0.9

Σt: 1 | 3

k	analytic	MC
0	3.33333	3.333
1	0.	-0.000342417
2	7.40741	7.40627
3	0.	-0.061048
4	98.7654	98.7463
5	0.	-3.08009
6	3292.18	3296.57
7	0.	60.4107
8	204 847.	206 417
9	0.	75 498

nth-collided moments of ϕ

In[17]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  ϕmoments = N[{data[[10 + n]]}/Σt];
  ks = {Table[k, {k, 0, nummoments - 1}]};
  analytic = Table[Quiet[N[ϕm[α, Σt, k, n]]], {k, ks}];
  j = Join[ks, analytic, ϕmoments];
  TableForm[
   Join[{{"k", "analytic", "MC"}}, Transpose[j]]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

α 0.1 0.3 0.5 0.7 0.9

Σt 1 3

n 2

Out[17]=

k	analytic	MC
0	0.49	0.489989
1	0.	-0.0000489696
2	2.94	2.93856
3	0.	-0.0148411
4	70.56	70.3666
5	0.	-1.21298
6	3528.	3508.95
7	0.	60.286
8	296 352.	296 243.
9	0.	128 332.

N-th order Radiance/Angular flux

In[18]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];
  nthL = data[[10 + numcollorders + 1 ;; 10 + 2 numcollorders]];
  nthR = data[[10 + 2 numcollorders + 2 ;; -1]];

  Clear[c];
  LnR = FullSimplify[SeriesCoefficient[LR[x, c, Σt], {c, 0, n}] α^n];
  Show[
   ListPlot[ppoints[nthR[[n + 1]], dx, maxx, Σt],
    PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[LnR, {x, -maxx, maxx}, PlotRange → All]
   , Frame → True,
   FrameLabel -> {{L₊["x|2"],}, 
     {x, "Infinite Rod, isotropic point, isotropic scattering, L₊[x|" <> 
       ToString[n] <> "], α = " <> ToString[α] <> ", Σₜ = " <> ToString[Σt]}},
   PlotRange → All, ImageSize → 400
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and 
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[18]=



N-th order Fluence / scalar flux

In[19]:=
```
Manipulate[
 If[Length[simulations] > 0,
  data = SelectFirst[simulations, #[[1]] == α && #[[2]] == Σt &][[3]];

  nthL = data[[10 + numcollorders + 1 ;; 10 + 2 numcollorders]];
  nthR = data[[10 + 2 numcollorders + 2 ;; -1]];

  Show[
   ListPlot[ppoints[nthR[[n + 1]] + nthL[[n + 1]], dx, maxx, Σt],
    PlotRange → All, PlotStyle → PointSize[.01]],
   Plot[ϕ[x, n, α, Σt], {x, -maxx, maxx}, PlotRange → All]
   , Frame → True,
   FrameLabel -> {{ϕ["x|7"],},
     {x, "Infinite Rod, isotropic point, isotropic scattering, ϕ[x|" <>
       ToString[n] <> "], α = " <> ToString[α] <>
       ", Σt = " <> ToString[Σt]}}, PlotRange → All
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {α, alphas}, {Σt, muts},
 {n, Range[If[NumberQ[numcollorders], numcollorders, 1]]}]
```

Out[19]=